
让函数成为提升效率的利器




——写给Excel函数初学者的序言

简单地说，函数是数学领域中的一种对应关系，也是很多人记忆中躺在课本里的各式各样冰冷的符号和表达式。在Excel中，函数其实就是预先定义好的，可以执行计算、分析等信息处理任务的特殊公式。通俗一点说，就是规定一个单元格如何随着另外N（N≥1）个单元格的变化而变化。对经常使用Excel的人来说，函数是用来提高工作效率的最重要的工具之一。另外一个现实是，很多人常用Excel，也听说过Excel函数，但对于函数的学习心生畏惧，总因觉得函数太难而却步。

广义地说，我们的生活中就充满了各种各样的函数——一件事情的结果往往取决于其他的一件或者多件事情。比如，甲问乙：你周末准备干啥？乙回答说：看情况吧，有时间的话我想去郊外转转，否则只能加班了。乙的回答，实际上就可以看成一个函数——去不去郊外，取决于是否有时间。如果用函数表达式来说就是：if（有时间，去郊外转转，加班）。很显然，前提是“有时间”是否成立，“去郊外转转”是成立的结果，“加班”是不成立的结果。当然，在这里，你不需要知道if函数究竟有什么作用，甚至不需要知道“if”的中文意思是“如果”，在本书中自然会有详细而深入的介绍。

在Excel中，我们用函数来简化我们的工作，节省我们的时间。另一方面，我们不需要知道函数内部是怎么样“工作”的，对大多数人来说，常用函数只需要知道3个元素：这个函数是干什么的，需要的是什么参数，输出的是什么结果。就好比我们只需要知道水在零度以下会变成冰一样，把这个当成常识就OK了，至于为什么，分子原子等是怎样变化的，我们不必理会。

几乎可以肯定，世上没有多少人能背下所有Excel函数的用途、拼写、参数等，很多学用Excel时间不长的朋友，对这一点的认识恰恰存在误区——Excel那么多函数，怎么背得下来？其实，并不是所有的函数都需要记下来，我们只要记住最常用的那些就行了。本书正是基于这一点编写的，只讲最常用的函数，而且把这些函数的用法彻底说清楚，以帮助读者在工作中能够真正做到灵活使用。

本书在内容的编排上，充分考虑到Excel函数初学者的学习需求和实际使用需求，所以不涉及那些使用频度低，或者过于偏行业应用的函数，以常用的、通用的函数为主。看看本书的内容梗概就知道了。

消除初学者学习函数的心理障碍。

清晰透彻地讲解学习函数前需要掌握的基础知识。

把经常用来进行逻辑判断、统计和处理数值、查找数据、处理文本、计算时间和日期的函数说明白。

介绍了用函数增强Excel功能、公式的简化与调试、宏表函数等相关知识。

可以说，这样的内容安排，不仅完全可以让读者掌握Excel常用函数在工作中的应用，而且可以写出运算更快速、结构更简单的函数公式——让读者进一步提高用Excel处理各种工作的效率。

读者在阅读本书的过程中可以发现，本书对每一个常见函数的用法，包括函数的基本用法、每一个参数变化用法、函数的运行效率、函数的相互替代可能，以及函数的扩展用法都进行了深入的讲解，力求让读者对这些常用函数拥有全方位的认识，进而提升函数的使用效率——在遇到问题的时候，能够在第一时间本能地想到要使用什么函数、如何使用，也就是让这些函数知识成为读者工作中的“常识”。

本书在策划之初，就得到了Excel精英培训网（www.excelpx.com，以下简称EP）各大版主的热情支持，也吸引了EP各大函数高手与微软MVP的踊跃参与。他们是小妖、小佩、展翅、法老等，当然还有EP的校长——兰色幻想。另外，每一个EP人的学习与工作需求都是我们尽心尽职地写作本书的动力，每一个EP人在论坛上提出的各种问题都是我们写作灵感的来源，在此对广大的EP人说一声“谢谢”。

在写作过程中，我们竭尽所能把最好的内容呈现给读者，但也难免有疏漏与不妥之处，敬请不吝指正。本书的源文件请登录www.ptpress.com.cn下载。如果读者在学习的过程中有任何问题，欢迎发E-mail至jiahongfei@ptpress.com.cn，或者登录http://www.excelpx.com和我们交流，也可以登录EP的新浪官方微博http://weibo.com/u/2191998127给我们留言。

编者

2012年1月





第1章 消除学习函数的心理障碍








1.1 远观函数，跨出误区




Excel中的函数很难学。

之前未接触过函数的新手朋友在学习之前，大都会有这样的认知。其实这是一个大大的误会。Excel中一些常用的函数是非常简单的，而使用这些常用函数足以应付我们日常的工作和学习。

现实中很多朋友不敢踏进Excel的函数世界，并不是被函数本身难倒了，而是被认识上的一些误区挡在了门外。





1.1.1 误区一：英文差就学不好函数




有这样的误区并不奇怪。对我们来说，Excel的诞生地是地球背面那个说英语的国家，那些工程师在编写函数程序时，当然不可能使用汉字或者拼音。所以，所有Excel自带的函数名要么是完整的英文单词，如if、sum等；要么是字母和英文单词的组合，如vlookup、midb等。

此外，每一个函数的参数名也都是英文的，如图1-1所示。




图1-1

如此一来，对英文不太熟悉的朋友就望而却步了……然而事实又是如何呢？我们不妨先来看这样几个真实的、对Excel函数名的看法的例子。

北部湾，某公司财务总监

“要我说呀，Excel里的函数是个很有用的工具，函数名称只是工具的代号而已，英文好，能记住工具的代号，仅仅是方便使用，和使用工具本身没有关系。就好像中国人不论认不认识“筷子”这两个字，都会用筷子夹菜吃饭；而一个老外，就算他认识‘筷子’这两个字，但不会使用，餐桌上没有刀叉勺他就得受罪，这种‘认识’与直接的‘使用’又有什么关系呢，所以说，会英语和Excel函数使用水平的高低没太大关系。”

王波，某高校行政管理

“函数和英语靠不上多大关系的，函数名不就是个名字吗？就像每个人都有自己的名字，函数也是。叫错别人的名字虽然不太礼貌，但叫错函数的名字，也不会怎么样啊，更不会影响函数的使用。虽然中文的名字感觉会更亲切一些，但是Microsoft Excel帮助文档那么详细，英文、中文真的影响不大，代号而已。”

孙晶，某公司测量工程师

“比如SUM函数，有人不知道它是由S-U-M这样3个字母组成，也不知道念成‘桑母’，但是想求和的时候，能够找到它并得心应手地用它。再比如，有人认识SUM单词，并且对它的拼写、读法了如指掌，但是当要求和的时候，只会敲计算器，而不会函数SUM(A1:A10)，这种‘认识’有什么用？所以我的结论是：英文水平和学习Excel没有关系，有点关系也仅仅只有26个字母而已，我英文差，念不了矮、比、希、滴（英文），但我念成啊、拨、次、的（拼音）又能怎么样？”

……

他们都是Excel精英培训网的资深会员，通过学习，从菜鸟变成函数高手，有的通过了网站的各种水平测试，有的则直接参与了这类测试的出题。他们的共同点是：在纯中文的环境中练就了高超的Excel函数水平。

可以看到，大部分函数名都是一些简单的单词，例如上面列举到的if和sum，即使是英文真的很没底子的朋友，相信也不会对这样的单词完全陌生。如果真的遇到一些比较少用的单词如substitute之类的，也可以借助翻译软件了解其含义。另外，如果使用Excel的2007版或者2010版，在单元格里输入函数名时，就会出现如图1-2所示的提示，然后只要轻按键盘上的【Tab】键，一长串字母就自动“跳”进单元格里了。




图1-2

至于函数的那些参数名，又不像函数名一样出现在“正式场合”，完全可以给个代号称呼而不必理会其英文大名。例如，把IF函数里的3个参数依序分别称为“第1个参数”、“第2个参数”、“第3个参数”，或者根据其含义分别称为“条件”、“是则结果”、“否则结果”。

另外，从图1-1中我们可以看到，中文版的Excel还是中文占了主导地位，对函数功能以及参数的说明都是汉字，可见Excel函数到了中国，仅仅是留了张英文的“皮”，其骨子里早已彻彻底底地被汉化了。

事实证明，有英文基础确实可以对函数名的记忆起到一定的帮助作用，但绝不是学好函数的必要条件。





1.1.2 误区二：注重排版，忽略制表和计算难度




Excel不是排版软件，这是绝大多数人的共识，可是在具体的使用过程中，Excel却经常被当作排版软件来用。如此一来，给制表和计算徒增了许多麻烦，甚至原本有些用简单函数就能解决的问题，因为排版原因而不得不使用最复杂的公式，让新手朋友从此把函数和自己划分到两个毫不相干的世界。

先看看如图1-3所示的表格。




图1-3

乍一看上去，这个表做得似乎十分清楚，两种产品在两个城市的库存情况都一目了然，但是这个表的制做过程，却是费时又费力。

斜线表头用绘图线条和文本框，经过仔细调整才成型。

型号种类不定，制表过程中需要不断调整A列中城市的合并单元格。

所有的小计总计都是用计算器算一遍核一遍，再仔细填入单元格，不仅工作量大，还极容易出错。

但是工作还没有就此结束，这时如果任何一项数据，如城市、产品型号、库存数量等有调整，整个表都要重做。更要命的是，如果老板突然想看各型号的库存量，或者想知道哪种型号、哪种产品库存量最大时，恐怕又要大大地辛苦一番了。

在这个表的基础上，要用函数公式完成所有的小计总计，以及各型号的库存从大到小排列，不仅复杂，而且难度非常大，完全不适合刚接触函数的朋友。

可以将表格结构调整一下，变成如图1-4所示的样子。




图1-4

所有数据不依顺序，只如流水账般记录，制表过程就变得非常简单，再有新增数据时，只要填到表后空白处即可。而老板会比较感兴趣的各地、各产品、各型号库存量，用几个简单的函数就可以统计出来了。

具体用哪个函数，这里先留个悬念，等看完本书所有章节后再回过头来看这一节，相信那时候的您一定会有一个非常明确的答案了，因为解决这几个问题要用到的函数，都是很基础的，根本没有难度。





1.1.3 误区三：会背的函数越多越好




这绝对是一个误区。Excel一共有400多个函数，迄今为止，还没有听说过有人号称对这400多个函数都熟背熟用。

函数不用背

细心的朋友可能早已发现，在工作表里编写函数的过程中，用什么函数、这个函数怎么拼写、参数有几个、各参数怎么用，等等，还没等我们来得及搜索完自己的大脑，种种提示早已经从屏幕的某个角落冒出来，传统的“背”函数已经变得毫无意义。当然，明天要参加Excel函数闭卷的笔试者不在此列。

如果没有注意到这些提示，那也没关系，本章1.3.3小节中“输入公式时的自动‘提示’部分”将详细列出每个提示。

熟悉几个常用函数就完全可以应付日常工作

在日常工作中，我们面对的问题绝大部分都可以被“求和计数”、“判断”、“查找”等函数所囊括，熟悉了这几个函数，再掌握一点基本的文本与日期函数，不敢说“打遍天下无敌手”，至少也可以“横扫一大片”了。

例如人事专员每月要统计当月新入职了多少人，就是先判断所有员工的入职日期是否是当月，再计数；再如销售员想知道自己销售的各种产品，与去年同期相比是多了还是少了，就是先分别查找各产品在去年同月和本月的销量，再判断大小……而以上这些，最常用的IF、COUNT、VLOOKUP、SUM等函数就可以解决了。

另外，不同函数之间也有一定的相互替代性，比如要计算平均值，一时想不起来用哪个函数，完全可以用SUM（求和）除以COUNT（计数），如图1-5所示。




图1-5

如果不知道用日期函数DATE可以直接将分别列在3个单元格里的年、月、日合到一个单元格，则可以用“&”符号连接，如图1-6所示。




图1-6

可见，Excel函数公式是否用得好，通常是用“熟练”或者“灵活”来判断，而与“死记硬背”相去甚远。





1.1.4 误区四：不使用辅助单元格




所谓辅助单元格就是在表外设置一些单元格进行过渡计算，对最终结果的计算起到帮助作用。有些新手朋友在解决问题的过程中拒绝使用辅助单元格，和一些难度较大的函数“纠缠”不清，最终以“函数太难”而退却。其实，加几个辅助单元格会让难度下降很多。

如图1-7所示，是个隔两行求和的例子。




图1-7

图1-7中所示的做法是在B2:B8 （即表示从B2单元格到B8单元格的区域）使用辅助单元格，最终的结果引用辅助单元格区域：

=SUM(B2:B8)

辅助单元格最大的缺点就是“不好看”，特别是需要打印的时候，还要将其隐藏或单独设置打印区域，所以到处遭人嫌弃。当然，我们可以直接在A9单元格里输入公式：

{=SUM((A2:A8)*IF(MOD(ROW(3:9),3),,1))}

不过，这是个数组公式，需要同时按下【Ctrl】、【Shift】、【Enter】这一组合键完成，虽然结果一步到位，但是这个公式复杂、难理解、难扩展、运行速度慢，这就给新手朋友徒增了许多困难。所以，在现实工作中，完全没有必要将辅助单元格拒于千里之外，以致最终将函数拒于千里之外。





1.2 走近函数，了解优点




曾经有人睁着一双很迷茫的眼睛问我：Excel的计算功能，能比函数计算器还强？我惊愕之余只能想，问这问题的人太不了解Excel了。

现实中很多朋友也许不会这么直接地问，但是从他们的办公桌上不离不弃的计算器就可看出，他们还远远没有体会到Excel函数与公式的优点。事实上，掌握了Excel一些最基础的函数后，就完全可以把计算器扔进箱底了。





1.2.1 优点一：入门简单，操作易学




Excel函数的入门非常容易，只要会四则运算，会英文输入法和至少一种汉字的输入法，就可以开始学习了。

1．什么是公式、什么是函数、什么是函数公式

公式

Excel里的公式，并不是一个多么高深的概念，我们逗邻家还未上学的孩子时问的“1加1等于几？”就是一个公式，这个公式被老师们写在黑板上的格式是：

1+1=

只不过到了Excel的单元格里，需要调整一下，把等号放在最前面：

=1+1

按下【Enter】键后我们就会看到结果：2。

换言之：由等号引出的，都叫公式。

在这个公式里，除了等号外，另由两个参数“1”和一个运算符号“+”组成。

公式里的参数

和函数的参数一样，公式里的参数可以是数字，可以是文本，可以是逻辑值，可以是单元格区域的引用，还可以是一个函数……具体都会在后面“2．函数的组成部分”里详细介绍，这里就不再赘言了。

公式里的运算符号

公式里运算符号的使用和数学中的四则运算没太大差别，只是在写法上略有不同，详见表1-1。

表1-1




需要注意的是，公式里的符号都是半角状态（即英文输入的状态）下的，原因，当然还是和Excel的诞生地有关，在那个不说汉语的国度里，怎么会冒出全角符号来呢？虽然有少部分全角符号在公式输入时会自动转成半角状态，但是我们在学习和使用函数时，最好养成“半角输入”的习惯。

函数

Excel里的函数，就是个相对复杂的概念了：“通过使用一些称为参数的特定数值来按特定的顺序或结构执行计算”（摘自Microsoft Office软件的帮助文档）。不过像这样抽象的概念性名词解释，恐怕不是那么容易让人理解并接受。不要着急，只要您看完“2．函数的组成部分”就会对函数有个初步的了解。

函数公式

Excel里的函数公式，就是最前面加了等号的函数，或者说是把函数作为参数的公式。例如：

SUM(A:A)

这是一个函数，直接这样写在单元里无法得出A列各数字之和，也即无法实现函数的功能，只有在前面加了等号，变成函数公式后，才会执行函数功能。

=SUM(A:A)

公式包括了函数公式和无函数公式，我们也会把函数公式简称为公式。

2．函数的组成部分

函数由3个部分组成：函数名、小括号、函数参数，不同参数间以半角逗号分隔。

函数名

函数名就是函数的一个代号，就像每个人的名字一样，用以区分不同函数。编写函数的人为了方便记忆，就用了含义相关的一些英文单词或组合来命名函数。这虽然给英文基础不太好的朋友学习函数带来一定困难，但绝不会成为函数学习过程中一道不可逾越的墙。

函数名的输入虽然不限制大小写，但是输入完成后，都会自动转成大写，如果没有自动转换，那就说明这个函数名的输入有误。

小括号

这其实是每个函数划分“势力范围”的一种标志，所以一个函数的小括号必须至少是一对，当然了，如果弄个十对八对上去（比如“=SUM((((((((A1:A10))))))))”），也不影响计算结果，只是看着会比较奇怪。

这里用的小括号和四则运算里的括号并不冲突，我们把一个函数作为某公式的一个参数来看待，可以将小括号套在公式外面，如：

=(SUM(A1:A10)+6)*2

“SUM(A1:A10)”就被当作是公式的一个参数，计算这个参数加6之和乘以2的结果。

函数中的参数也可以用到一些四则运算，如：

=SUM((A1+6)*2,B1)

SUM有两个参数，其中第1个参数就是一个四则运算，计算A1单元格的值加6之和乘以2。

参数及符号

函数的参数就是函数学习的重点了，基本上可以分为如表1-2所示的几类。

表1-2




参数之间以半角逗号分隔，某参数如果引用的是单元格矩阵区域，则以半角冒号连接这一区域的左上角和右下角

1


 ，详见第二章的“2.2单元格的引用方式”。

1 确切地说应该是以半角冒号连接某区域的任意一条对角线的两个单元格，但是在输入以后，会自动转换成左上角和右下角。

函数中符号的应用十分广泛，可以这么说，只要键盘上有的符号，函数里都用得着。当然了，对于刚接触Excel不久的朋友来说，没有必要一下子把所有的符号都背下来，因为有一些符号在函数的高级应用中才会用到。

常用符号的介绍请参阅1.3.3小节中的“3．函数公式里一些常见的符号及作用”。

3．新手学习函数的办法

学习使用帮助文档

打开帮助窗口的快捷键是功能键【F1】。

Microsoft Excel帮助文档应该说是学习Excel最好的老师，也是最权威、最完整的教材，一般情况，当遇到问题都可以从帮助文档中搜索到答案（尽管不一定看得明白）。

帮助文档的缺点就是用词比较晦涩。图1-8所示是SUM函数的帮助，以至于很多新手朋友不愿意看帮助，或者看了帮助也没获得帮助，进而选择通过其他途径学习。




图1-8

通过网络学习

在网络如此发达的今天，通过网络学习已不是什么新鲜事，网上有铺天盖地的资料，包括微软官方的易宝典、专业人士博客、专业技术论坛等，只要轻点鼠标搜索，只有想不到的，没有找不到的。

买一本书学习

市面上的书有两类，一类是“字典型”的，其中的内容并不是十分系统，但是对于没有时间系统学习的朋友，这类书非常有用，一旦遇到问题，就可以从这本“字典”中查找答案。

另一类是属于“系统型”的，非常合适需要系统学习的朋友。

本书属于“系统型”，适合想把Excel函数真正学明白的人阅读。

与人交流、分享

知识只有在交流和分享的过程中，才会体现出其最大的价值。关于这一点，每个经常在专业技术论坛里“泡坛子”的朋友都有深刻的体会。

Excel技能就是在不断地解决现实问题的过程中提升。当您的Excel技能达到一定程度时，会发觉自己能遇到的问题越来越少，这时候进步的速度也逐渐变慢，只有不断寻找新的问题并解决之，水平才会进一步提升。而与人交流、分享正是寻找新问题的途径，那种把自己学的一点点技能藏着掖着的人，永远也不可能成为高手。





1.2.2 优点二：统计迅速




电脑制表肯定比手工制表要快得多，量尺寸画格子这类杂事统统可以免掉。但是Excel的迅速，绝不仅仅体现在这方面，而是有无数让人意外的惊喜。

比如某公司有一千多名员工，现在需要进行工资计算，传统的做法是一个一个依次计算，假如计算一个人的工资需要3分钟，那么要完成当月工资报表，就需要不吃不喝不休息连续作战50个小时才能完成。而用Excel计算的效率呢？只要先计算好1个人的工资，其他人的工资通过一个“下拉”动作（即“向下填充”，详见本书第33页中图2-12所示）就可以完成了。哪怕我们计算这一个人的工资蜗牛般地用了半个小时，剩下的一个“下拉”动作也只需几秒钟就能完成。





1.2.3 优点三：杜绝错误




除了迅速以外，Excel还有个计算器所不能替代的优点，就是大大减少了错误，甚至可以毫不夸张地说成杜绝错误。

我们来看计算器计算的过程：输入数据到计算器，得出结果，结果填写入表，在这3个过程中，任何一个环节出现一点小问题，错误就出来了。而Excel呢，只有输入数据一个环节有可能产生错误，这就大大减少了错误的产生。另外，如果能很好地利用Excel的查找引用功能，将数据输入的过程变成数据引用的过程，这样一来，不就完全杜绝了错误了么？

当然，编写公式的过程必须十分小心，不然会出现要么不错、要错全错的悲惨局面。





1.2.4 优点四：修改联动




联动是Excel中一项让人叫绝的优点。如图1-9所示，我们在C3单元格里输入这样的函数公式，以计算B1:B10单元格区域的和：

=SUM(B1:B10)

结果当然是地球人都知道的55，可这时假设我们突然发现，B6单元格的数据输入是错误的，应该是16。那么按照传统的做法，就需要修改两处：一是把B6单元格里的数字改成16，二是把C3单元格里的数字改成65。而当我们将B6单元格里的数字改好以后，这个让人叫绝的联动功能就出现了，C3单元格里的数字自动变成了65，如图1-10所示。




图1-9




图1-10

想想，如果没有这个联动，一个错误只改两处还好，换成那种一个错误要改个十处八处的，一不小心错改一处或漏改两处，那结果就可想而知了。





1.3 输入函数，亲密接触




有了前面那么多铺垫，该我们亲自动手输入一个函数公式，感受一下它的迅速、准确和联动了。





1.3.1 查找函数




有很多新手朋友面对大堆的数据会很茫然，不知道该用哪个函数。确实如此，当一些常用函数还未装入我们的大脑时，我们要从哪里入手呢？

Excel的设计者似乎早就预料到了这个问题，所以给我们提供了这样一个功能，我们单击一下如图1-11所示的“fx”这个位置。




图1-11

之后就会出现如图1-12所示的一个对话框（这是“插入函数”的一种快捷方法，后面还会介绍其他打开这个对话框的方法）。




图1-12

这里有一句话，不知是否引起了您的注意：“请输入一条简短的说明来描述您想做什么，然后单击‘转到’”。假如我们在这里输入“求和”二字，再单击“转到”按钮，看看这个对话框会出现什么样的变化，如图1-13所示。

电脑会非常聪明地把与求和相关的所有函数列出来(一共8个)，通过这个提示，我们就可以很方便地寻找到想要用的函数，而不需要从所有400多个函数中去“捞”。

顺便提一下，单击这个对话框左下角的“有关该函数的帮助”链接，就可以直接打开所选取函数的帮助文档，免去了搜索之苦。

不过，Excel毕竟是来自异乡，“中文水平”实在一般，所以在“搜索函数”时经常需要用不同的同义词多加测试才行。我们可以测试一下，在“搜索函数”框中分别输入“如果”和“条件”，看看会得到什么不同的结果。




图1-13

当然也可以从本书的目录中找到想用的函数进行详细的学习。





1.3.2 练习输入无参数函数公式




之前我们了解到，函数是由函数名、一对括号、参数构成的，但是有些函数很奇特，只有函数名和括号，没有参数，这类函数我们称之为“无参数函数”。比如：显示当前日期、时间的函数，随机数函数等。

1．通过“插入函数”对话框输入函数公式

接下来让我们一起输入一个显示电脑系统当前日期的函数公式。

虽然说函数公式的输入方法是多种多样的，而且Excel也不是十分在意输入的过程，只要结果正确统统接受。但是对于刚接触函数不久的朋友来说，还是用“插入函数”对话框来输入函数公式比较稳妥，这样不仅不易出错，而且得到的提示也更多。

关于插入函数，之前我们已经了解，可以单击编辑栏左边的“fx”按钮进入，另外也可以通过执行菜单命令的方式进入。

2003版：“插入”→“函数”命令

2007版或2010版：“公式”→“函数库”→“插入函数”命令

进入这个对话框后，我们在“或选择类别”下拉列表框中选取“日期与时间”选项，再选择函数“TODAY”，如图1-14所示。




图1-14

不知道您注意到没有，当我们选取了“TODAY”这个函数后，下面就出现了对应这个函数功能的简单说明，提示我们这个函数的基本功能。

当这一切都选取好后，单击“确定”按钮，又有一个新的对话框出现了，如图1-15所示。




图1-15

因为是个无参数函数，所以这里不需要做任何设置，也不需要输入任何字符，直接单击“确定”按钮，单元格里就会出现电脑系统当前的日期，而编辑栏里，则出现了这样的函数公式：

=TODAY()

显示当前日期的函数公式就完成了，如图1-16所示。




图1-16

如果我们把这个工作簿保存下来，到第二天再打开时就会发现，编辑栏并没有变化，而输入了TODAY函数单元格中的日期变了。

2．直接在单元格里写公式

Excel公式输入是不论过程只问结果的，也即无论输入的过程是通过对话框输入，还是直接在单元格里写，或者在别处写好后粘贴入单元格，只要公式是正确的，就会得出一个正确的结果。就像上面这个计算系统当前日期的函数公式，我们一样可以把“=TODAY()”通过粘贴的方式贴入单元格，哪怕输入的是小写字母都没有问题。





1.3.3 练习输入简单参数函数公式




毕竟，无参数函数只占非常少的一部分，绝大多数的函数还是有参数的，一个函数也正是有了千变万化的参数，才会发挥种种妙用，那么有参数的函数公式应当如何输入呢？

1．快捷输入参数

让我们一起输入一个求和的函数公式，表格结构如图1-17所示，B1:B10区域是数据源，在C3单元格输入结果。




图1-17

先打开“插入函数”对话框，因为求和函数SUM用处太大了，所以它被列在的“常用函数”中的前列，很容易找到，如图1-12所示。选取“SUM”后单击“确定”按钮，就可以看到这样一个对话框，如图1-18所示。




图1-18

当我们把鼠标点到Number1旁边的框里时，下面就会出现关于这个参数的说明，通过这一提示，可以帮助我们了解这个参数具体的使用方法。

输入参数的过程，当然可以在Number1旁边的框里输入单元格地址，如：A1、B1或者A1:A10之类的，但是这种逐字符的输入方式未免有点让人出汗，最好有个简化的办法。

事实上，Excel确实提供了这种简化功能，就在这个对话框打开的状态下，随意点一个单元格，例如A5单元格，这个单元格的地址“A5”就会自动出现在Number1旁边的框里。输入单元格区域也可以用同样的方法。按照要求，求和区域应是B1:B10这个区域，那就用鼠标直接选取，如图1-19所示。




图1-19

单击“确定”按钮，这个求和的函数公式就输入完成了。

眼神好的朋友也许又发现了一处新的提示，如图1-20所示；图中所示的每个输入参数的框框后面有被选取的单元格区域里的内容，再下面更有计算结果，此处的提示也许在这里用处还不大，但当我们输入公式有误时，就可以在未单击“确定”按钮的情况下，直接从这个对话框里看出错误的出处。




图1-20

当我们对函数熟悉到一定程度以后，就可以直接在单元格里写函数公式，而不用再通过“插入函数”对话框了。直接在单元格里写函数公式时，单击单元格区域即可输入单元格地址的功能也一样能用。我们先在单元格中输入：

=SUM(

然后用鼠标选取单元格区域，就可以看到被选取的地址自动列到左括号后面，如图1-21所示

2


 。

2 事实上，不一定非要输入“=SUM(”，只要输入等号后，再用鼠标选取单元格区域，这个区域的地址就自动进入单元格里的等号后面。




图1-21

这个方法在引用本工作表的单元格时，可能其优点还不是十分明显，如果我们要做一些跨工作表、甚至跨工作簿的引用时，这一个小小的动作就可以帮我们自动完成工作簿名、工作表名和单元格名的输入，除了方便、快捷外，还保证无错，可以说是函数公式输入中的又一亮点。

但是这个功能也有偶尔失效的时候。假设现在已有两个工作簿被打开，我们在其中一个工作簿的任意单元格内设置SUM公式，光标已经处于SUM函数的括号中，但在点选另一个表的B1:B10单元格区域后，并没有形成如图1-21所示的效果。

这种情况是由于Excel文档打开方式不正确所导致的。在正常情况下，多个Excel文档都会在同一个Excel程序进程中打开。如果在打开一个工作簿后，再次从“开始”菜单中打开Excel主程序，在新的Excel程序窗口内打开另一个工作簿，就会使两个工作簿文件运行于不同的Excel程序进程中，而不同的Excel进程之间无法互相通信，致使引用失败。

可以通过同时按下【Ctrl】、【Alt】、【Del】3个键打开任务管理器，在任务管理器的“进程”选项卡中，能看到系统中运行着两个Excel程序进程，如图1-22所示。




图1-22

解决的方法也很简单，关闭其中一个工作簿，通过在另一个工作簿的Excel窗口中执行“文件”→“打开”命令再次打开这个工作簿，此时，两个Excel工作簿文件都会运行于同一个Excel程序进程中。

跨工作簿成功引用数据，保存并关闭这两个工作簿后，当再次打开带有公式的工作簿时，将弹出“外部数据源链接”对话框，如图1-23所示。如果单击“更新”按钮，公式将重新读取带数据源的工作簿里的数据。




图1-23

另外，函数输入过程中有个小框，不知您是不是会觉得输入的时候特别“碍事”，如图1-24所示。




图1-24

这个小框叫“参数工具提示”，十分有用，可以告诉我们这个函数有哪些参数。如果整个函数公式已输入完毕，点这个提示里的某一参数时，就会在单元格里自动选取这个参数的完整内容，如图1-25所示。




图1-25

这个“参数工具提示”还有个用处就是单击函数名，可以直接跳转到“关于该函数的帮助”。

2．养成良好的公式输入习惯

当我们对函数熟悉到一定程度时，输入公式就不再需要借助“插入函数”对话框，而是直接在单元格里写。这时候如果能养成良好的输入习惯，会在很大程度上降低公式出错的概率。例如VLOOKUP函数有4个参数，我们在输入时按如下顺序。

第1步：输入“＝vlookup(,,,)”。

第2步：在每个逗号之间输入参数。

乍一看，这样输入似乎复杂化了，可是当我们输入的公式是由好几个函数套用时，这样的输入习惯可以帮助我们减少出错。

当然了，等成为函数高手后，常用函数早已了然于胸，就没必要这么复杂了。

3．函数公式里一些常见的符号及作用

函数入门学习的常用符号除了表1-1所示的内容外，还包括一些，如表1-3所示。

表1-3




4．输入公式时的“自动提示”

函数名提示

在单元格里输入等号和一个字母后，逐步出现函数名提示，如图1-2所示（仅限于2007版或2010版）。

使用指定函数提示

插入函数对话框后，根据要求搜索函数，如图1-13所示。

函数功能提示

在函数对话框中显示被选取函数的主要功能，如图1-14所示。

参数功能提示

函数对话框中显示各参数的使用方法，如图1-18所示。

计算过程提示

函数对话框中显示各参数计算结果和最终结果，如图1-20所示。

参数工具提示

在输入函数过程中有“参数工具提示”框时提示参数，如图1-24所示。

5．实例操作

两眼正盯着本书的朋友，放松一下，按下面的流程自己动手测试一下吧。

（1）打开一个新的工作簿，在Sheet1工作表的A1:A10输入1～10这10个数字。

（2）将这个工作簿保存到桌面，命名为data。

（3）再打开一个新的工作簿，选取A1后，插入函数，再选取SUM。

（4）用鼠标选取data工作簿Sheet1工作表的A1:A10。

（5）按下【Enter】键。

（6）关闭data工作簿。

这时，A1里的公式是什么样的？

如果用的是Windows XP 操作系统，公式是这样的：

=SUM('C:\Documents and Settings\[用户名]\桌面\[data.xls]Sheet1'!$A$1:$A$10)

如果用的是Windows 7操作系统，公式则是这样的：

=SUM('C:\Users\[用户名]\Desktop\[data.xls]Sheet1'!$A$1:$A$10)

对比一下，您是不是也做到了？





第2章 学好Excel函数必须先掌握这些知识








2.1 Excel中的数字格式




在Excel中，数据保存在单元格内，书写公式时，我们会通过书写单元格地址的方式达到引用该单元格内数据的目的。为了数据能够更方便地存储和引用，不同类型的数据将会采用迥异的数字格式进行保存。

修改单元格的数字格式可以通过以下方式完成：选择目标单元格，执行“格式”→“单元格”菜单命令，在弹出的“单元格格式”窗口的“数字”选项卡内列出了12种格式分类，点选所需格式，并单击“确定”按钮。

如图2-1所示， 将B列单元格修改为A列中描述的单元格格式，并全部输入数字“10000”后显示效果完全不同。对于数字而言，由于格式的不同，在单元格中会有不同的表现形式，在使用公式计算时也会带来不同的结果，这样的问题经常给Excel初学者带来很大的困扰。在开始介绍本节下面的内容前，我们首先需要了解一点原则：对于一个已经具有数字格式的数字而言，修改数字格式的操作无法对其值的大小造成任何改变，只能更改其表现形式。我们将在本节分别介绍Excel中的12种数字格式，并分析、解决可能会在公式引用中出现的问题。




图2-1





2.1.1 数值类的数字格式




数值类的数字格式，是指依然被Excel承认是“数”的一类格式，具备“大小”的属性，包含除“文本”与“自定义”之外的其他10类格式，它们具备以下特点。

可以自由地在数值类的各类格式中转换，但不能仅通过格式修改转换成文本类的数字，例如可以将“常规”格式的“10000”更改为“数值”格式，变成“10000.00”，但是其本身值的大小并无改变。

可以在不经过特殊处理的情况下，直接使用统计类函数进行运算，并且这种运算在逻辑上是有意义的。如图2-2所示，此例中使用了SUM函数，表示将A2到B2单元格中的数字加在一起，当数据同为数值类的数字时，对它进行加、减、乘、除等运算是有意义的，而左右相连得到“1000010000.00”是无意义的。




图2-2

除“会计专用”格式采用“分散对齐”外，其他数值类的数字格式默认对齐方式均为“右对齐”。

常被Excel做出一些“自作聪明”的修改，例如去掉无意义的0、四舍五入等。除这些共同的特点外，数值类的各种数字格式也具备各自的特点，下面我们将分别做简单介绍。

常规格式

这是在Excel单元格中输入数字的默认格式，可以表示正或负的整数或小数，但不能显示无意义的0值，例如输入“001”和“1.00”都将会显示为“1”；如果单元格的宽度不足以显示整个数字，则“常规”格式将对含有小数点的数字进行四舍五入处理，并对较大的数字使用科学记数法。

数值格式

可以设置小数点后保留的位数、是否使用千分位分隔符（“，”）、是否用红颜色显示负数。

货币格式

可以设置小数点后保留的位数、是否用红颜色显示负数、使用何种货币符号，默认货币符号位于数值的左侧并紧贴数值。

会计专用格式

可以设置小数点后保留的位数、使用何种货币符号，默认货币符号位于单元格的左侧与数值分离。

日期格式

在Windows系统中安装Microsoft Excel将默认使用1900日期系统。在1900日期系统中，1900年1月1日将被规定为日期的开始。如图2-3所示，在单元格内输入数字“1”后，再改为日期格式，其值将显示为“1900-1-1”；输入数字“2”后，再将单元格改为日期格式，其值将显示为“1900-1-2”。而输入一个日期，例如“1927-5-18”后，将单元格更改为“常规”格式，其值将显示为10000，表示“1927-5-18”这个日期是自“1900-1-1”开始的第10000天。




图2-3

如果我们将日期“1927-5-18”与常规格式的“10000”进行比较，如图2-4所示，会发现两者相等，说明修改单元格格式后，虽然日期“1927-5-18”的表现值变为10000，但大小并没有变化，所以本质上，日期也是数值类的数字格式。




图2-4

注意：将两个数值直接用等号相连，表示对两个数值进行比较，如果二者相等则返回“TRUE”，反之则返回“FALSE”。

时间格式

如上文所述，在Excel中，数字1表示1天，那么1小时则为1/24，如果将常规格式的“0.5”更改为时间格式，将显示为“12:00:00”，表示12个小时即0.5天。

百分比格式

将单元格内的数字乘以100，并在数字后加百分号（“%”），可以修改小数点后保留的位数。例如将常规格式“0.8”更改为百分比格式，将显示为“80.00%”。

分数格式

以分数的形式显示数值，可以选择分母的数值，Excel将会自动对可约分的分数进行约分。

科学计数格式

以科学计数法的形式显示数值，例如将常规格式的“10000”更改为科学计数格式，将显示为“1.00E+04”，表示1乘以10的4次方。

特殊格式

可以将一个数值显示为邮政编码（只保留整数部分，对小数部分进行“四舍五入”，不足6位的数字用“0”在数值左侧补足到6位）、中文小写数字及中文大写数字。虽然数值在显示为邮政编码的时候采用了“四舍五入”的处理方式，但仅是改变了数字的显示方式，真实值并没有改变。如图2-5所示，将常规格式的“1.5”更改为邮政编码格式，将显示为“000002”，用这个值与“1.5”比较，结果显示两者相等。




图2-5





2.1.2 文本数字格式




文本数字格式是指用数字格式中的“文本”格式，它是用来表示“编码”的数字格式，它不被Excel承认为“数”，也不具备“大小”的属性。与数值类的数字格式相比，它具有以下截然不同的特性。

对于同一个数字而言，文本格式与数值类的格式并不相等。如图2-6所示，将A2单元格更改为文本格式，B2单元格不作更改，使用默认的常规格式，在A2单元格与B2单元格内分别输入数字“1”，比较大小后发现不相等。由于文本类型的“1”与数值类型的“1”并不相等，而更改单元格的格式并不能改变单元格内值的大小，所以只修改单元格数字格式无法将文本类型的数字更改为数值类型的数字。例如将图2-6中A2单元格更改为日期格式，虽然单元格的数字格式已经变为日期格式，但文本类型的“1”并没有变成日期格式的“1900-1-1”，其真实值依然是文本类型的“1”。




图2-6

由于不再被Excel当作数值对待，不能直接使用SUM、PRODUCT等函数进行计算，如图2-7所示，使用SUM函数对A2到A4区域求和，其结果为0。另外，由于文本类型的数字通常用来表示代码，对它进行加、减、乘、除的运算也没有意义，例如区号“010”与电话号码“67132816”，需要将它们连接在一起，构成“01067132816”。




图2-7

在文本格式的单元格中输入数据，默认的对齐方式为“左对齐”。

以文本形式录入的数据，无论是否符合数值的书写规范，Excel都不会对其进行任何修改。例如将单元格设置为“文本格式”后，输入“001”，会显示为“001”。

注意：在单元格内直接输入身份证号码等长数字时，由于Excel的计算精度只有15位，会将18位身份证号码的后3位自动修改为零，并使用科学计数法的格式显示数字，所以必须使用文本格式记录。除了先将单元格设置为“文本”格式，再输入身份证号码的方法外，不设置单元格格式，而在输入数字时在数字前加“’”（半角格式单引号）的方式也能达到同样的效果。





2.1.3 不同类型间数字的转化




由于不同格式的这些特性，导致我们在处理这类数据时会遇到很多问题，例如不能使用SUM函数对文本类数字求和、格式不匹配时无法用查找类函数查找到正确结果等，也不能通过修改单元格格式的方式，实现数值类数据与文本格式数据间的转化。在函数使用中遇到此类问题，可以通过以下方法解决。

1．数值类的数字转化为文本格式的数字

如上文所述，数值类的数字是依然被Excel承认为“数”的数字，这类数字进行加、减、乘、除等运算，在逻辑上是有意义的，而左右连接在一起是无意义的。但如果使用连字符（“&”）进行左右连接处理，数值类的数字将被强制转化为文本格式的数字。如图2-8所示，将数值类的数字连接空字符“”””，其数字格式变为“文本”格式。




图2-8

注意：“&”被我们称作连字符，它用于将两个字符串左右连接在一起，本例将数值与“”””连接在一起，由于“”””中不包含任何字符，所以，既达到了连接的目的，又实现了不改变数字而转化格式的效果。

2．文本类的数字转化为数值类的数字

文本格式的数字是指代表编码的一类数字，在逻辑上，这样的数字进行加、减、乘、除等运算是无意义的，并且使用SUM等函数无法运算，但如果使用加、减、乘、除等运算符进行强制运算，文本格式的数字将被转化为数值类数字。如图2-9所示，将文本格式的数字乘以1后，转化为数值类的数字，而由于任何值乘以1的结果都等于其本身，导致数字仅改变了数字格式而大小并无改变。




图2-9

本例中C列数字与E列数字在转换为数值类数字后，对齐方式均为“左对齐”，查看单元格格式后，发现单元格数字格式依然为文本格式，这是由于它们继承了A列数字的格式所致，但实际上C列与E列的数字已经变成了数值类的数字，这一点可以在C列数据能使用SUM函数进行求和运算中得到佐证；也可以将C3单元格数字格式修改为“日期”格式，此时C3单元格将显示“1927-5-18”，再次说明C列单元格已经被转化为数值类的数字。





2.1.4 “自定义”数字格式




单元格的数字格式中，最特殊的数字格式非“自定义”莫属，前面章节中列出的所有单元格数字格式达到的显示效果都可以通过在“自定义”数字格式中书写代码的方式实现。例如，可以先将单元格数字格式设置为“常规”，再点选“自定义”数字格式，在“单元格格式”窗口中，“自定义”数字格式的类型为“G/通用格式”，表示单元格数字格式中的“常规”格式等同于“自定义”数字格式中的“G/通用格式”，如图2-10所示；将单元格数字格式设置为“日期”中的“*2001-3-14”，再点选“自定义”数字格式，在“单元格格式”窗口中“自定义”数字格式的类型为“yyyy-m-d”；将单元格数字格式设置为“文本”，再点选“自定义”数字格式，在“单元格格式”窗口中“自定义”数字格式的类型为“@”。

可以认为，除“自定义”数字格式本身外，其他11种数字格式都是由“自定义”数字格式中的代码编写得到。由于这11种格式非常常用，所以将它们分类并分别设置按钮，方便我们点选使用。相对前面11种数字格式，“自定义”数字格式在技术上更加靠近底层，使它在实际使用中拥有更大的操作自由度。

例如，将A2到A4单元格数字格式设置为“数值”，并保留两位小数后，全部输入数字“10000”，显示为“10000.00”，如果我们希望在数字后面增加单位名称“元”，可以在“单元格格式”窗口中选择“自定义”数字格式，在类型“0.00_”后增加单位名称“元”后，变为“0.00"元"”，此时单元格内数值将显示为“10000.00元”，如图2-11所示。而由于修改单元格数字格式无法改变值本身的大小，也不能将数值类的数字变为文本格式数字，所以依然可以使用SUM函数对这3个数字进行求和运算。




图2-10




图2-11

“自定义”数字格式极高的操作自由度，使它成为日常工作中处理单元格格式问题的利器，而“自定义”数字格式中的类型代码能与TEXT函数完美地结合，用于解决很多数据显示样式问题。关于“自定义”数字格式的更多编码规则，详见本书第6章“用函数处理文本”中与TEXT函数有关的小节。





2.2 单元格的引用方式




将数据存储在Excel里，好处显而易见，可以非常方便地对数据进行处理，省略了很多需要手工运算的工作。在使用函数处理数据时，函数通过引用单元格地址的方式引用单元格内的数据。Excel默认使用“A1”引用样式，这是一种以字母为列号、数字为行号的引用样式，我们可以把Excel的列号看作“经度”、行号看作“纬度”，通过“经纬度”在Excel表内确定一个唯一的点，即单元格。例如单元格“C5”表示C列与第5行交点的单元格。

在描述单元格区域时，可以将该区域的起点与终点单元格使用“:”（半角冒号）连接起来，表示以起点单元格为左上角、终点单元格为右下角的矩形范围，如果起点单元格位于终点单元格的右侧或下方，Excel将做自动修正。例如，单元格区域“A1:C2”表示以A1单元格为左上角、C2单元格为右下角的矩形范围，包含“A1”、“A2”、“B1”、“B2”、“C1”、“C2”6个单元格。公式“=SUM(A1:C2)”表示将这6个单元格求和。而如果输入“=SUM(A2:C1)”，公式将被自动修正为“=SUM(A1:C2)”。

我们也经常会使用针对整行或整列的引用，例如引用“1:1”表示第1行整行、“C:C”表示C列整列、“1:10”表示第1行到第10行这10行的整行数据。

如果说函数大大提高了我们的计算效率，那么公式的可复制性功不可没。在Excel中进行数据计算时，我们总是希望找到一个方法只书写一次公式，再使用拖曳的方式把公式应用到其他同类运算中，这样就大大减少了重复劳动，这也是函数优于手工运算的根本原因之一。如图2-12所示，在H3单元格输入公式“=SUM(B3:G3)”，表示求B3到G3单元格数字之和，按下“Enter”键后点选H3单元格，并将鼠标指针悬停于H3单元格的右下角，此时鼠标指针将变为黑色十字形状，按下鼠标左键并向下拖曳至H5单元格后再放开左键，公式将被填充到H4与H5单元格内，并得到关于第4行与第5行的求和结果。




图2-12

采用拖曳填充方式，当单元格填充至H4单元格时，原公式“=SUM(B3:G3)”自动变为“=SUM(B4:G4)”从而得到第4行数据之和，这正是我们希望达到的效果，而有这样的效果也是由于使用了相对引用。不同的引用方式，会在函数公式的复制过程中产生不同的效果。本节将会介绍单元格的3种引用方式及常见应用。

注意：相对于“A1”引用样式，Excel还支持另外一种不常见的引用样式，即R1C1引用样式，在这种引用样式中，行号和列号均为数字，引用中R代表行号、C代表列号，例如R4C2表示第4行与第2列交点的单元格。





2.2.1 单元格相对引用




如果说“北京市东城区夕照寺街14号”这个地址是绝对引用，则“我家隔壁”就是相对引用。“北京市东城区夕照寺街14号”指固定的一个地方，而且是唯一的；而“我家隔壁”会随着“我家”的搬迁而变化，总是描述“隔壁”这个相对位置。

在Excel中，相对引用是针对单元格相对位置的引用，如果在公式中使用相对引用，公式一旦被拖曳复制，相对引用的地址将根据当前所在单元格的位置进行自动调整。如上例图2-12所示，H3单元格内公式“=SUM(B3:G3)”中，由于使用了相对引用“B3:G3”，表示将公式所在单元格左侧的6个单元格累加，在公式被拖曳复制到H4单元格时，被自动调整为“=SUM(B4:G4)”依然表示将公式所在单元格左侧的6个单元格累加。所以，当我们希望公式拖曳复制后单元格内的引用随位置的变化而变化时，使用相对引用。





2.2.2 单元格绝对引用




绝对引用是针对单元格绝对地址的引用，这种引用不会因单元格的拖曳复制而改变，它的写法是在单元格的行列号前加“$”符号，例如“$A$1”表示对A1单元格的绝对引用。如图2-13所示，原始表记载了不同型号产品的出库记录，我们希望在F4:F6区域内分别统计这3类产品的出现次数，即出库单数。这里使用了COUNTIF函数，本例中“F4”单元格公式为“=COUNTIF($C$3:$C$11,E4)”，表示在$C$3:$C$11区域内统计“E4”单元格内值的数量。

在这个示例中，可以看到绝对引用的重要作用，如果不使用绝对引用，在F4单元格内输入公式“=COUNTIF(C3:C11,E4)”并拖曳填充至F6单元格，F6单元格内的公式将变为“=COUNTIF(C5:C13,E6)”，表示在C5:C13区域内统计E6单元格内型号的数量。我们并不希望公式这样变化，因为无论统计哪种型号的出现次数，都应该在C3:C11区域内统计，所以必须使用绝对引用“$C$3:$C$11”将这个区域“锁定”，表示拖曳复制单元格后公式中对于这个区域的引用不能改变。




图2-13

在本书的后续章节中，还会经常涉及一种由一个相对引用和一个绝对引用共同构成的可伸缩区域，例如“$A$1:A1”。这类引用，会随着公式所在单元格的变化扩大或缩小引用区域。如图2-14所示，在D2单元格内公式“=SUM($B$2:B2)-SUM($C$2:C2)”中，$B$2:B2区域表示从第2行到公式当前所在行的B列数据区域。例如当公式被拖曳复制到D5单元格时，引用区域由D2单元格中的“$B$2:B2”变为“$B$2:B5”，引用的区域扩大了。




图2-14

注意：将公式中的某个引用从相对引用转变为绝对引用，可以在公式中选择该单元格引用后，按下【F4】键。如图2-15所示，选中公式中的“A1”后，按下【F4】键，“A1”转变为“$A$1”，再次按下【F4】键，转变为“A$1”，第三次按下【F4】键，会转变为“$A1”，第四次按下【F4】键，会变回为“A1”。




图2-15





2.2.3 单元格混合引用




除了相对引用（例如“A1”）、绝对引用（例如“$A$1”）外，混合引用（例如“A$1”、“$A1”）也是一类经常被使用的引用方式。通过前面的学习，我们了解到，单元格的引用方式完全为公式的拖曳复制服务，如果公式不被拖曳复制，就不会涉及到引用的变化。

混合引用分为“A$1”、“$A1”两种样式，分别表示“列变行不变”与“行变列不变”。可以简单地把“$”符号看作一个锁定符号，对于引用“A$1”而言，无论我们向哪个方向拖曳复制公式，只可能带来列号的变化，而行号不会变化；而对于引用“$A1”而言，无论我们向哪个方向拖曳复制公式，只可能带来行号的变化，而列号不会变化。

由于混合引用对于行、列号分别描述是否锁定，使其适用于行、列号分别可能变化的情况，体现在公示引用中，可用于处理公式既要水平拖曳复制又要垂直拖曳复制的情况。如图2-16所示，B1:J1区域与A2:A10区域中分别记录了1～9的顺序序列，我们希望在B2单元格内输入公式后，向右、向下拖曳复制公式，通过对这两个顺序序列相乘的方式，将B2:J10区域填充为“九九乘法表”的结果集。

在B2单元格中输入公式“=$A2*B$1”，向右拖曳复制至J2单元格，此时B2:J2区域为选中状态，再将这个区域向下拖曳复制至表格第10行，完成计算。

对于公式“=$A2*B$1”而言，“$A2”是对垂直1～9顺序序列的引用，而这个垂直顺序序列只存在于A列中，不能让引用变为其他列，所以在“A”前加“$”；“B$1”是对水平1～9顺序序列的引用，而这个水平顺序序列只存在于第1行，不能让引用变为其他行，所以在“1”前加“$”。




图2-16





2.2.4 跨工作表与跨工作簿引用




Excel中的运算不仅支持同工作表内的数据运算，还可以引用其他工作表乃至其他工作簿的数据进行运算。当公式中引用其他工作表的数据时，引用格式为“工作表名称!单元格或区域”，例如“Sheet1!A1”表示引用名为“Sheet1”的工作表中的A1单元格，我们称这种对于其他工作表的引用为“跨工作表引用”；当公式中引用其他工作簿的数据时，引用格式为“[工作簿名称]工作表名称!单元格或区域”，例如“[Book1.xls]Sheet1!A1”表示引用“Book1”工作簿中 “Sheet1”工作表的A1单元格，我们称这种对于其他工作簿的引用为“跨工作簿引用”。

了解跨工作表与跨工作簿引用的使用方式和原理，对排查Excel中一些由引用引起的公式错误具有重要意义，特别是对在后续章节中学习使用INDIRECT函数的引用尤为重要。

1．跨工作表引用

可以通过手工输入或鼠标点选的方式使用跨工作表引用，例如在Sheet1工作表的A1单元格输入数字1，B1单元格输入数字2，在Sheet2工作表中对这两个单元格求和。在Sheet2工作表的A1单元格输入“=SUM()”后将光标移动到括号内，点选Sheet1工作表标签，此时Sheet1工作表被激活，使用鼠标点选A1:B1区域后按下【Enter】键。此时公式自动生成为“=SUM(Sheet1!A1:B1)”，计算结果为3。

将Sheet1工作表重命名为“Sheet”，观察Sheet2工作表中A1单元格的公式，该公式自动转变为“=SUM(Sheet!A1:B1)”，可见当工作表名称变化时，公式中针对这张工作表的跨工作表引用将会做自动修改。

再次将Sheet工作表重命名为“Sheet 1”（在“Sheet”与“1”间增加一个空格），此时Sheet2工作表中A1单元格的公式变成“=SUM('Sheet 1'!A1:B1)”，与上个情况相比，不仅引用中的工作表名称被自动修正了，而且在表名“Sheet 1”外增加了一对单引号。可见，当工作表名称含有一些特定字符的时候，如果对这张工作表进行跨表引用，需要在工作表名外增加单引号。

当工作表名称为数字或内含有“空格”、“～”、“！”、“@”、“#”、“$”、“%”、“^”、“&”、“(”、“)”、“+”、“=”、“,”、“;”、“{”、“}”等字符时，因为这些符号在公式中出现会表示其他的意义，需要在工作表名外增加单引号。例如“^”表示某个数字的N次方、“&”表示连字符，所以需要使用单引号圈定表示表名的文本。如图2-17所示，工作表名称决定了在跨表引用中是否需要使用单引号。




图2-17

使用鼠标点选的方式引用其他工作表内的数据，当需要在工作表名称上使用单引号时，Excel会自动添加单引号，无需使用单引号时不会添加；如果使用手工输入的方式引用其他工作表的数据，当需要在工作表名称上使用单引号时，必须在输入工作表名时添加单引号，否则该引用会出错，但如果在不需要使用单引号的表名外添加了单引号，Excel会自动去掉这对“多余”的单引号。所以在手工输入跨工作表引用的工作表名时，如果不能确定是否使用单引号，可以一律添加单引号，降低出错的可能性。

2．多表引用

多表引用是指同时对多张工作表数据的引用，常见于多张工作表格式一致，并将这些工作表上的数据进行汇总的工作中。

如图2-18所示，同一工作簿内的4张工作表，顺次为“一月”、“二月”、“三月”及“一季度”。在一季度工作表中计算3个月中3个城市的销量总和，可以使用公式“=SUM(一月:三月!B2:B4)”，其中“一月:三月”表示从“一月”表到“三月”表之间，包含“一月”表与“三月”表在内的所有工作表。




图2-18

3．跨工作簿引用

相对跨工作表引用，跨工作簿引用显得更为复杂，数据源工作簿是否处于打开状态、通过何种方式打开，都会对引用产生影响。

首先在D盘根目录下新建两个工作簿文件，分别命名为“表1”和“表2”，在表1中Sheet1工作表的A1单元格内输入1，B1单元格内输入2，再打开表2，在表2中Sheet1工作表的A1单元格内输入“=SUM()”后，将光标移动到括号内，点选表1窗口，使用鼠标点选A1:B1区域后按下回车键。此时公式被填充为“=SUM([表1.xls] Sheet1!$A$1:$B$1)”，并得到结果3，如图2-19所示。




图2-19

在进行上一步操作时，可能会遇到以下问题：光标已经处于SUM函数的括号中，但在点选表1并选取A1:B1单元格后，并没有形成如图2-18所示的跨工作表单元格引用。这种情况是由于Excel文档打开方式不正确所导致的。在正常情况下，多个Excel文档都会在同一个Excel程序进程中打开。如果在打开表1后，再次从“开始”菜单中打开Excel主程序，在新的Excel程序窗口内打开表2，就会使两个工作簿文件运行于不同的Excel程序进程中，而不同的Excel进程之间无法互相通信，致使引用失败。可以通过同时按下【Ctrl】、【Alt】、【Del】 3键打开任务管理器，在任务管理器的“进程”选项卡中能看到系统中运行着两个Excel程序进程，如图2-20所示。解决的方法也很简单，关闭表2，在表1的Excel窗口中执行“文件”→“打开”命令，在“打开”窗口中选择表2并打开，此时，两个Excel工作簿文件都将运行于同一个Excel程序进程中。




图2-20

在表2中引用表1数据成功后，保存并关闭表1，此时表2中的跨工作簿引用被自动修改为含有路径信息的引用样式“=SUM('D:\[表1.xls]Sheet1'!$A$1:$B$1)”。保存并关闭表2。再次打开表1，将表1中Sheet1表内B1单元格数据修改为5后，保存并关闭该表。再次打开表2，将弹出“外部数据源链接”对话框，如图2-21所示。如果单击【更新】按钮，公式将重新读取表1数据并得到新的结果6。




图2-21

注意：在本节中，使用了SUM函数构建跨工作簿的引用，并且能在数据源工作簿关闭的情况下正确更新数据，但Excel中只有少数函数支持不打开数据源工作簿引用数据，绝大多数函数都需要将数据源工作簿打开才能正确引用。所以对于绝大多数函数而言，如果数据源工作簿没有打开，而又单击了“更新”按钮，会导致公式出错。





2.3 使用名称




在公式中灵活使用名称可以给函数应用带来莫大的好处，名称不仅可以增加公式的可读性，使公式层次更加清晰、简洁，而且可以突破一些功能上的限制，使函数愈发强大，并将函数与图表、数据透视表等功能更加完美地结合在一起，发挥更大的作用。

以下是Excel不同版本中定义名称的方法。

2003版：执行“插入”→“名称”→“定义”命令

2007版和2010版：执行“公式”→“定义名称”命令





2.3.1 为常量定义名称并在公式中使用




工作中，我们经常会使用一些相对来说变动得不频繁，而又代表一些特殊含义的数值，这类数值也经常会在不同的计算中被调用，例如本例中提到的养老保险个人缴费比例（8%）。

在图2-22所示的数据中，C列数据为月薪数值，在D列计算养老保险扣款金额。D2单元格公式应为“=C2*0.08”，但这样的公式可读性不强，特别是在引用很多单元格公式中，无法很直观地看到0.08所表示的意义。为了解决这个问题，可以将0.08定义为名称“养老保险比例”，并在公式中直接引用这个名称。




图2-22

创建名称的方法非常简单，执行“插入”→“名称”→“定义”菜单命令后，弹出如图2-23所示的“定义名称”窗口，在“在当前工作簿中的名称”处输入名称“养老保险比例”，并在“引用位置”处输入“=0.08”，然后单击“添加”按钮，再单击【确定】按钮完成创建。自此，“养老保险比例”这几个字在本工作簿中就被赋予了全新的含义，它代表了0.08这个常量数值，并可以直接在公式中被引用完成计算。在上个示例（见图2-22）工作表中，由于使用了这种定义名称的方式，所以可以在D2单元格输入公式“=C2*养老保险比例”得到养老保险比例的扣款金额。




图2-23

在公式中使用名称的另一个好处是使公式更加便于修改。在上个示例中，养老保险比例可能会在很多其他的公式中使用，一旦养老保险个人缴费比例上调至9%，则意味着会有很多公式需要修改。如果在创建这些公式时都采用了引用名称的方式，那么只需要将名称“养老保险比例”对应的值由0.08更改为0.09即可。

将名称“养老保险比例”由0.08修改为0.09的方法也同样简单。

2003版：执行“插入”→“名称”→“定义”命令，在“定义名称”对话框中点选“养老保险比例”，并将“引用位置”内的“=0.08”修改为“=0.09”，单击“添加”按钮后，单击“确定”按钮。

2007版和2010版：执行“公式”→“名称管理器”命令，在“名称管理器”对话框中点选“养老保险比例”，单击“编辑”按钮，并将“引用位置”内的“=0.08”修改为“=0.09”后，单击“确定”按钮，关闭名称管理器。

注意：本例在计算养老保险比例时，由于目前掌握的函数比较少，所以没有考虑工资低于最低缴费基数的情况。这类问题的处理方法详见“3.2 IF函数”。





2.3.2 为单元格或区域定义名称并在公式中使用




为单元格或区域定义名称不仅能增加公式的可读性，还为迅速找到这个单元格或区域提供了极大的便利。

在图2-24所示的工作表中，将单元格区域“$A$2:$A$5”定义为名称“月薪”，在后续计算中涉及到将这个区域求和时，只需要使用公式“=SUM(月薪)”即可。在定义名称时，引用位置中的“=Sheet2!$A$2:$A$5”，不仅可以手工输入，也可以通过使用鼠标点选的方式完成输入。在本例中，引用A2:A5区域时使用了绝对引用，是因为一旦使用相对引用,随着当前活动单元格的变化，名称引用的位置也会发生变化，关于在名称中使用相对引用的问题在后续章节（2.3.6 小节“体验宏表函数”）中将做详细介绍。




图2-24

在本例中，A2:A5区域有了它的新名字“月薪”，而有了名字的另一个好处就是在工作表中“随叫随到”。

在图2-25所示的“名称框”右侧下拉列表中，点选名称“月薪”，“A2:A5”区域将被选择。如果工作表中存在多个经常查看的区域，可以将这些常用区域定义名称，方便查找。




图2-25

名称框除了可以“呼叫”名称外，也可以完成定义名称的工作，例如将B2:B5单元格定义名称为“测试”，可以先选中B2:B5区域，并在名称框中输入“测试”后按下【Enter】键。





2.3.3 为函数定义名称




除了可以为常量、单元格或区域定义名称外，也可以为公式定义名称，这样做可以大大减少公式的嵌套层次，使公式的条理更加清晰，也减少了公式的重复输入。例如公式“=IF(SUM($A$1:$A$5,$A$7:$A$11,$B$4:$B$9,$C$7:$C$11)>10000,SUM($A$1:$A$5,$A$7:$A$11,$B$4:$B$9,$C$7:$C$11),SUM($A$1:$A$5,$A$7:$A$11,$B$4:$B$9,$C$7:$C$11)*2)”表示如果“SUM($A$1:$A$5,$A$7:$A$11,$B$4:$B$9,$C$7:$C$11) ”的结果大于10000，则显示“SUM($A$1:$A$5,$A$7:$A$11,$B$4:$B$9,$C$7:$C$11)”的求和结果，否则显示“SUM($A$1:$A$5,$A$7:$A$11,$B$4:$B$9,$C$7:$C$11)*2”的结果，这里可以定义名称“求和结果”，其引用位置为“=SUM($A$1:$A$5,$A$7:$A$11,$B$4:$B$9,$C$7:$C$11)”，从而公式简化为“=IF(求和结果>10000,求和结果,求和结果*2)”。

注意：本例中使用IF函数进行判断，IF函数用法详见54页。





2.3.4 定义名称的限制




在定义名称时，有些名字不被Excel承认，会造成无法定义名称的情况出现，所以在定义名称前，需要检查一下名字是否符合Excel对合法名称的要求。

不能以数字或除下划线外的特殊符号开头。

如果要在名称中使用特殊符号，只能使用“_”（下划线）和“.”（小数点）。

不能以现有的单元格地址命名，例如名称“A2”为非法，但“AAAAA2”合法。

不能以符合“R1C1”引用样式的字符串命名。例如在Excel 2003中共有65536行，所以名称“R65536”为非法名称，而“R65537”为合法名称。





2.3.5 工作簿级名称与工作表级名称




在定义名称时，可以根据需要定义不同作用范围的名称。由于工作簿由多个工作表构成，明确名称的作用范围将使名称的创建更加合理。例如在我们工作的公司部门中，现任经理只有一个，所以直接称呼“经理”我们都知道是谁，但如果在公司其他部门提到这位经理，则需要在“经理”前增加前缀“某某部”，构成“某某部经理”，表示“经理”这个称呼的“作用范围”只有本部门；而无论在公司内的哪个部门，提到“董事长”，我们都知道所指的是谁，可以看出“董事长”这个称呼的“作用范围”是全公司。

1．工作簿级名称

前面章节中定义的名称都是工作薄名称，作用范围是整个工作簿，这类名称能在名称所在工作薄的任意工作表中被直接使用。在图2-26的示例中，在Sheet1工作表中为A2:A5单元格区域定义名称“月薪”，在Sheet1工作表中可以直接使用公式“=SUM(月薪)”对该区域求和，而在Sheet2工作表中同样可以使用公式“=SUM(月薪)”对其求和，无论在哪张工作表中，名称“月薪”都是指Sheet1表的A2:A5单元格区域。




图2-26

值得一提的是，在定义工作薄名称时，如果在引用位置前加“！”，将产生非常奇妙的效果：在本工作簿的任何一张工作表中使用这样的名称时，名称引用的区域都将是公式所在工作表本身的单元格区域，而不是固定的某一张工作表的单元格区域。

如图2-27所示，在Sheet1工作表中定义名称“月薪”，设置引用位置为“=!$A$2:$A$5”。作为一个工作簿名称，“月薪”可以在本工作簿的任何一张工作表中被引用。在引用时，将表示当前工作表的A2:A5单元格区域。




图2-27

如图2-28所表述的情况，在Sheet1工作表中定义的，名称“月薪”由于引用位置被设置为“=!$A$2:$A$5”，在Sheet1工作表中使用，将表示引用Sheet1的A2:A5单元格区域；在Sheet2工作表中使用，将表示引用Sheet2的A2:A5单元格区域。




图2-28

2．工作表级名称

相对于工作簿名称，工作表名称的作用范围仅限于名称所在工作表，如果在其他工作表中调用这个名称，需要在引用时加上名称所在工作表的表名称，而且在“定义名称”对话框中不会显示其他工作表的工作表级名称。

定义工作表级名称的方法是在创建名称时在名称前加工作表名，如图2-29所示，在Sheet1工作表中为单元格区域A2:A5定义工作表级名称“Sheet1!月薪”，单击“添加”按钮后，该名称会添加到名称列表中。与工作簿级名称不同，名称列表中显示该名称时，在名称右侧增加了工作表名“Sheet1”，表示这个工作表级名称属于Sheet1工作表。




图2-29

这个属于Sheet1工作表的名称在Sheet1工作表中使用时，与工作簿名称完全相同，但在其他工作表中引用时，必须在名称前加“Sheet1!”，否则将返回“#NAME?”错误。图2-30所描述的就是在Sheet2中使用工作表级名称“Sheet1!月薪”时的情况。




图2-30





2.3.6 体验宏表函数




在前面所有为单元格区域定义名称的例子中，无一例外地将引用位置中的单元格引用写成绝对引用，这样做会使选区固定，如果在公式中引用了这个名称，拖曳复制该公式不会造成引用区域的变化，类似于在公式中使用绝对引用。而有时候我们希望名称引用的区域能随着公式的拖曳复制而变化，同样也可以用相对引用来定义名称。

新建工作表，点选Sheet1工作表的B1单元格，执行“插入”→“名称”→“定义”菜单命令，在弹出的“定义名称”对话框中，在“当前工作簿中的名称”处输入“测试”，在“引用位置”内输入相对引用“=A1”，单击“添加”按钮后，单击“确定”按钮，这样就定义了一个相对引用的名称。点选B2单元格，并再次打开“定义名称”对话框，在名称列表内选择名称“测试”，此时“引用位置”已经变为“=Sheet1!A2”；点选B10单元格，查看该名称会发现其“引用位置”变为“=Sheet1!A10”。

可见，对于B1单元格而言，名称“测试”所引用的是A1单元格；对于B2单元格而言，名称“测试”所引用的是A2单元格；而对于B10单元格而言，名称“测试”所引用的是A10单元格。这种情况类似于在B1单元格直接输入引用“=A1”后，向下拖曳复制所产生的效果。

用相对引用定义名称常见于宏表函数的使用中。宏表函数是一类特殊的函数，它兼具“宏”与“函数”特征，功能强大，但不能在工作表中直接使用，必须在被定义为名称后使用。本节将通过对宏表函数EVALUATE的使用，简单认识宏表函数及定义相对引用名称的好处，关于宏表函数的更多内容详见第10章。

EVALUATE能将一个文字表示的公式或表达式求值，并返回结果。在图2-31所示的Sheet1工作表中，A2:A5单元格区域中保存着一些公式，希望在B2:B5单元格区域计算它们的结果。

点选B2单元格，执行“插入”→“名称”→“定义”菜单命令，在弹出的“定义名称”对话框中，在“当前工作簿中的名称”内输入“结果”，在“引用位置”内输入公式“=EVALUATE(A2)”，此处必须使用相对引用“A2”，不能使用绝对引用“$A$2”，单击“添加”按钮后，单击“确定”按钮，完成名称的创建。在B2单元格中输入公式“=结果”，按下【Enter】键后，将公式拖曳填充至B5单元格。




图2-31

在B2:B5单元格区域中，显示的值分别为同行左侧表达式计算的结果。但公式均为“=结果”，因为在B2单元格中，名称“结果”的引用位置为“=EVALUATE(Sheet1!A2)”，表示将A2单元格内的表达式计算得到结果；而对于B3单元格而言，名称“结果”的引用位置为“=EVALUATE(Sheet1!A3)”，表示将A3单元格内的表达式计算得到结果。虽然公式相同，均为“=结果”，但在不同的单元格内，“结果”所引用的单元格不同，通过使用相对引用定义名称的方法，达到了类似于在单元格内创建相对引用公式并拖曳复制的效果。





2.4 加载分析工具库




为了处理一些比较复杂的数据分析问题，Excel提供了一组非常强大的分析工具，并且将这些工具整合打包为分析工具库，方便我们使用。分析工具库中也同样加入了很多工作表函数，如果未加载分析工具库，将无法使用这些函数。分析工具库作为一个插件，在使用时需要手工加载。

加载分析工具库需要在“加载宏”对话框中勾选“分析工具库”复选项，不同版本Excel中“加载宏”对话框的位置略有不同。

2003版：执行“工具”→“加载宏”命令。

2007版和2010版：程序左上角圆形“Office按钮”→“Excel选项”→“加载项”命令，在“管理”下拉列表中选择“Excel加载项”，单击“转到”按钮。

在弹出的“加载宏”对话框中勾选“分析工具库-VBA函数”复选项后，单击“确定”按钮，如图2-32所示。




图2-32

分析工具库中的函数在未加载分析工具库时无法使用，如果在未加载分析工具库的工作表中出现这类函数，将返回“#NAME？”错误。如图2-33所示，使用LCM函数计算A2:A4单元格区域中3个数字的最小公倍数，在未加载分析工具库时，LCM函数无法使用。需要注意，如果由于未加载分析数据库导致公式出现“#NAME?”错误，在加载分析工具库后，该公式不会自动更新为正确结果，需要强制更新。强制更新可以通过点选公式所在单元格后，按下【F2】键，再按下“回车键”完成。




图2-33

注意：在部分精简版Excel中，分析工具库将无法加载，可将其卸载后安装完全版Excel。





第3章 使用函数进行逻辑判断




逻辑函数是函数中的一大分类，虽然函数的数量不多，但在日常工作中却不可或缺。例如在员工退休年龄的计算中，男性退休年龄为60岁，女性退休年龄为55岁，这就需要通过一次对性别的简单判断来计算不同的退休年龄。如果使用逻辑函数来描述这个判断过程，方式将为：“如果性别为男，则55岁，否则50岁”。

逻辑函数能非常直观地将某些判断转化为不同的结果，由于是基于判断的，所以逻辑函数引用了布尔代数中的一些基础理论来计算这些判断。本章内容将会介绍3个基本逻辑运算函数：与（AND）、或（OR）、非（NOT），以及如何使用IF函数将这3个函数反馈的值转化为我们预想的结果。





3.1 TRUE和FALSE




布尔值只包含TRUE与FALSE两个值，一般意义上这两个值用来描述一次条件判断的正、反两个结果，即“真”与“假”，而一次条件判断只可能有这两种结果。例如比较两值大小的公式“X>Y”，当数值X大于Y时，该公式结果为TRUE，即为真；当X等于或小于Y时，该公式结果为FALSE，即为假。

布尔值可以通过输入一个条件判断公式得到，需要使用比较运算符（=、<>、>、<、>=、<=），也可以使用部分逻辑类函数或信息类函数直接得到。当然，在Excel中直接输入“TRUE”或“FALSE”，也会被当作布尔值处理，如图3-1所示。为了区别文本与数值，布尔值在单元格中默认的对齐方式为“居中对齐”。在配合IF函数处理条件判断的公式中，大多数情况下布尔值由判断得出。




图3-1

在四则运算中，可以将布尔值当作数值来运算，而TRUE和FALSE可以分别当作1和0来进行运算；如果将数值当作布尔值来处理，则0将被视为FALSE，非0有理数均被视为TRUE。如图3-2所示的结果中，A2、A3单元格使用乘1的方式将布尔值当作数值来运算；A4：A8单元格区域中使用AND函数将不同的数值当作布尔值来运算。

注意：AND用来表示多个布尔值之间“且”的运算，即逻辑与。当AND函数的参数中所有布尔值为TRUE的时候，公式结果为TRUE；当AND函数的参数中存在FALSE时，公式结果为FALSE。本例中AND函数只有一个参数，是为了验证数值与布尔值的对应关系。




图3-2

布尔值在四则运算中可以当作数值1和0来运算，是因为1和0在数学上是两个比较特殊的数字，其中任何数字乘以1，结果均为这个数字本身，任何数字乘以0，结果都为0，这符合布尔运算定律，导致我们可以直接使用这种对应关系来完成一些基于判断的简单运算。如图3-3所示，在C列根据同行B列单元格内的性别计算退休年龄，在C2单元格输入公式“=55+(B2=”男")*5”后下拉填充。

在C2单元格中，对应性别为男时，判断公式“B2="男"”的结果为TRUE，在四则运算中被视为1，乘以5以后公式“(B2="男")*5”结果依然为5，加上55得到男性退休年龄60；在C3单元格中，对应性别为女时，判断公式“B3="男"”的结果为FALSE，在四则运算中被视为0，乘以5以后“(B3="男")*5”结果为0，加上55后得到女性退休年龄55。




图3-3





3.2 AND函数、OR函数与NOT函数




在公式中使用条件判断时，我们经常会涉及到根据一个或多个判断的再判断，例如判断一条记录是否为60岁以上的男性，则需要进行两个条件的判断：年龄是否大于60、性别是否为男性。只有当某条记录的两个条件同时满足时，才是符合条件的记录，表示两个条件是“且”的关系（年龄大于60且性别为男），即“与运算”。

本节中将会涉及布尔运算中的“与运算”、“或运算”、“非运算”，这3种运算方式分别对应Excel中的AND、OR、NOT 3个函数，分别描述一个或多个条件判断之间“且”、“或”、“非”的3种关系。





3.2.1 AND函数




AND函数表示将多个条件判断关联在一起，最终得到一个布尔值，只有当所有条件判断的结果均为TRUE时，AND才会返回结果TRUE，否则将返回FALSE。AND函数用来表示布尔运算中的与运算，描述多个参数间“且”的关系。

函数语法：

AND( 条件判断 1,条件判断 2, ...)

其中“条件判断1, 条件判断2, ...”表示待检测的1到30个条件判断，可以为数组、布尔值，或可计算出布尔值的公式。

在图3-4所示的表格中，对年龄大于60岁的男性员工给予1000元奖金。在D2单元格输入公式“=AND(A2="男",B2>60)*1000”后向下拖曳填充完成运算。




图3-4

在公式“AND(A2="男",B2>60)”中，AND函数的参数由两次条件判断组成，当这两次条件判断的结果均为TRUE时，公式“AND(A2="男",B2>60)”的结果为TRUE，否则为FALSE。例如第2行数据的情况——性别为男并且年龄大于60岁——同时满足了两个条件，公式“AND(A2="男",B2>60)”的结果为TRUE，又由于TRUE与1000相乘时被视为数值1运算，导致最终乘积为1000；而在第3行的数据中，不能同时满足年龄和性别的要求，公式“AND(A3="男",B3>60)”的结果为FALSE，由于FALSE与1000相乘时被视为数值0运算，导致最终乘积为0。

注意：在AND函数的函数语法介绍中，提到了关于使用AND函数进行与运算时只支持30个条件判断的问题，AND函数在Excel 2003中最多支持30个条件判断的再判断，而2007版中支持255个。这个数量限制是指AND函数参数个数的限制。例如在Excel2003中，公式“AND（条件判断1,条件判断2, ... ,条件判断29,条件判断30）”中，最多只能有30个条件判断，如果这30个条件判断分别存在于A1:A30单元格区域内，公式“AND(A1:A30)”也同样表示对30个条件判断的再判断，但“A1:A30”只会被当作一个参数。如果遇到参数超过数量限制的情况，可以在书写公式时使用双括号，例如：“AND（（条件判断1,条件判断2, ... ,条件判断49,条件判断50））”。





3.2.2 OR函数




OR函数同样用作将多个条件判断关联在一起，最终得到一个布尔值，在这些条件判断中，任何一个或多个条件判断的结果为TRUE时，OR函数就会返回结果TRUE，只有当全部条件判断的结果均为FALSE时，才会返回结果FALSE。OR函数用来表示布尔运算中的或运算，描述多个参数间“或”的关系。

函数语法：

OR( 条件判断 1, 条件判断 2, ...)

其中“条件判断1, 条件判断2, ...”表示待检测的1到30个条件判断，可以为数组、布尔值，或可计算出布尔值的公式。

在图3-5所示的表格中，对年龄大于60岁的员工或女性员工给予1000元奖金。在C2单元格输入公式“=OR(A2="女",B2>60)*1000”后，向下拖曳填充完成运算。

在这个问题中，两个条件判断（年龄大于60、性别为女）只要有任何一个满足条件，都需要给予1000元奖金，所以使用OR函数进行或运算。





3.2.3 NOT函数




NOT函数是将条件判断“取反”，例如当“条件判断1”的结果为TRUE时，“NOT （条件判断1）”的结果为FALSE；如果“条件判断1”的结果为FLASE，则“NOT（条件判断1）”的结果为TRUE。

函数语法：

NOT(条件判断)

参数可以为数组、布尔值或可计算出布尔值的公式。

由于NOT函数在实际使用中具备较强的可替代性，所以使用频率不如AND函数和OR函数高，例如图3-6中将小于60的A列数值在“是否过低”中标记为TRUE，公式1与公式2中的结果完全相同，而两相比较，公式2的公式更加简洁。

3.3 IF函数




IF函数是逻辑函数中的核心函数，它大大提升了条件判断后所得到布尔值的可用性，并且非常直观地描述了当条件判断的结果为TRUE或FALSE时，分别采取何种数据处理方式或返回什么样的结果。





3.3.1 IF函数基本语法




IF函数在参数表述上非常贴近语言习惯，其函数结构为“如果怎么样，就怎么样，否则怎么样”，这3个“怎么样”就是IF函数的3个参数。其中第2参数与第3参数可以省略，例如在至尊宝的台词“如果上天再给我一次机会,我会对那个女孩说我爱你”中，只提到了满足条件时的处理方式，并没有说明对“否则”如何处理，这是省略第三参数的写法，如果判断条件没有被满足，将返回FALSE。

函数语法：

IF(条件判断,满足条件如何处理,不满足条件如何处理)

条件判断：必不可少的参数，可以是值或表达式。

满足条件如何处理：当第一参数条件判断的结果为TRUE时返回的结果，可以是值或表达式。

不满足条件如何处理：当第一参数条件判断的结果为FALSE时返回的结果，可以是值或表达式。

在图3-7所示的表格中，根据实际费用和预算费用判断使用情况，如果实际费用大于预算费用，则显示“超出预算”，否则显示“正常”。在C2单元格输入公式“=IF(A2>B2,"超出预算","正常")”后下拉填充。

公式“=IF(A2>B2,"超出预算","正常")”表达的意思为如果A2大于B2，则显示“超出预算”，否则显示“正常”。在本例应用中，IF函数的第一参数为一个条件判断，其结果可能为TRUE或FALSE。当结果为TRUE时，公式所在单元格显示第二参数的值；当结果为FALSE时，公式所在单元格显示第三参数的值。




图3-7





3.3.2 IF函数实例




IF函数依据第一参数条件判断的结果来决定返回第二参数或第三参数，但对于条件判断而言，其结果只能为TRUE或FALSE，这造成单一的IF函数最多只能描述两种情况。如上例中，仅有“超出预算”与“正常”两种可能性，可以用一个IF函数来完成。在实际工作中，绝大多数情况是根据两种以上的可能性处理数据。在本节中，我们将分别探讨IF函数在多条件处理时的一些常见使用方式。

1．使用IF函数处理多重条件的判断

如图3-8所示，在C列根据转入金额与币种，计算折合人民币的数额，在C2单元格输入公式“=IF(B2="美元",A2*6.5,IF(B2="欧元",A2*9.3,A2))”后下拉填充。

在本例中，由于币种有3种，导致出现3种不同的可能性，需要使用两个IF函数嵌套在一起对3种情况分别描述。




图3-8

以C2单元格内公式“=IF(B2="美元",A2*6.5,IF(B2="欧元",A2*9.3,A2))”为例，外围的IF首先判断B2单元格的文字是否等于美元，如果等于美元，C2单元格将显示A2*6.5的计算结果；如果不等于美元，C2单元格将显示“IF(B2="欧元",A2*9.3,A2)”的计算结果，也就是说，当B2单元格中的文字不是美元时，公式“IF(B2="欧元",A2*9.3,A2)”才会被运算。公式“IF(B2="欧元",A2*9.3,A2)”首先判断B2单元格的值是否为欧元，如果为欧元，则C2单元格将显示A2*9.3的计算结果；而如果B2单元格的值不等于欧元，则说明B2单元格的值既不是美元也不是欧元，所以必然是人民币，将返回A2单元格的值。

需要注意，在多个IF函数嵌套使用时，IF函数并不遵循“先算括号里面，再算括号外面”的原则，而是自左向右运算公式。在本例中公式“IF(B2="美元",A2*6.5,IF(B2="欧元",A2*9.3,A2))”首先比较B2单元格的文字是否为美元，如果是美元，则返回“A2*6.5”的结果，在这种情况下第三参数的公式“IF(B2="欧元",A2*9.3,A2)”将不被运算；只有在B2单元格内的值不是美元时，才会运算公式“IF(B2="欧元",A2*9.3,A2)”，在这种情况下公式“A2*6.5”将不会被运算。

在上个示例中，由于存在3种情况（金额乘以美元汇率、金额乘以欧元汇率、金额不乘以汇率），所以将2个IF函数嵌套在一起使用，而两个IF将构成1层嵌套。如果存在4种情况，例如在币种内增加英镑，则需要使用3个IF函数嵌套处理，C2单元格公式应为“=IF(B2="美元",A2*6.5,IF(B2="欧元",A2*9.3,IF(B2="英镑",A2*10.6,A2)))”。可见，当使用IF函数处理多种可能性的问题时，如果存在N种可能性，将需要N-1个IF函数嵌套在一起。

通过这种函数嵌套，IF函数看似在处理多条件问题时无所不能，但实际上，嵌套的层数是有限制的。在Excel 2003中，IF函数最多嵌套7层，而在Excel 2007及Excel2010中，IF函数最多嵌套63层。为了使IF函数能处理超过最大数量限制的条件判断，我们可以采用一些特殊的公式书写方法创建公式，例如定义名称方法、分散方法和函数并行方法。

定义名称方法

以Excel 2003中使用IF函数描述10种可能为例，图3-9列出了10个不同币种对人民币的汇价，在F列根据汇价与数额计算折合人民币的金额。在本例中，币种达到10个，显然已经超出了Excel 2003中7层嵌套所允许的最大数量。于是，将其中5个币种的处理定义为名称。

点选F2单元格后创建名称，名称命名为“后五种情况”，引用位置为公式“=IF(D2="瑞士法郎",E2*7.37,IF(D2="瑞典克朗",E2*1.02,IF(D2="丹麦克朗",E2*1.23,IF(D2="挪威克朗",E2*1.17,E2*0.81))))”。这个公式能处理瑞士法郎、瑞典克朗、丹麦克朗、挪威克朗、澳门元这5个币种的折合人民币计算。

在F2单元格输入公式“=IF(D2="美元",E2*6.47,IF(D2="欧元",E2*9.18,IF(D2="英镑",E2*10.52,IF(D2="加拿大元",E2*6.64,IF(D2="港元",E2*0.83,后五种情况)))))”。这个公式中不仅描述了当币种为美元、欧元、英镑、加拿大元、港元时的运算，还阐明当币种不是这5个币种之一时，将采用名称“后五种情况”所定义的公式来计算。

通过这种定义名称的方式，可以将一部分针对某些可能性所做的运算交给定义名称功能来处理，变相增加了IF函数的嵌套层次数量。

均衡分散方法

分散方法与定义名称方法类似，都是将过多的条件判断分组来处理。

在F2单元格输入公式“=IF(OR(D2="美元",D2="欧元",D2="英镑",D2="加拿大元",D2="港元"),IF(D2="美元",E2*6.47,IF(D2="欧元",E2*9.18,IF(D2="英镑",E2*10.52,IF(D2="加拿大元",E2*6.64,E2*0.83)))),IF(D2="瑞士法郎",E2*7.37,IF(D2="瑞典克朗",E2*1.02,IF(D2="丹麦克朗",E2*1.23,IF(D2="挪威克朗",E2*1.17,E2*0.81)))))”。

这个公式使用了OR函数进行判断，将10个币种分为两部分，并分别作为最外侧IF函数的第二参数及第三参数进行运算，而第二参数与第三参数都由一组IF函数公式构成，每个参数负责处理5个币种。如果币种为美元、欧元、英镑、加拿大元、港元则按照最外侧IF函数第二参数中的IF函数公式处理，否则按照最外侧IF函数第三参数中的IF函数公式处理。

公式并行方法

除以上两种方法外，改变IF的书写方式也可以带来相同的效果。例如在图3-9所示数据中，在F2单元格内直接输入嵌套型的IF函数，将会受到嵌套层次数量的限制，这时就可以使用公式并行方法不使用嵌套而完成运算。

在F2单元格内输入公式“=IF(D2="美元",E2*6.47,0)+IF(D2=”欧元",E2*9.18,0)+IF(D2=”英镑",E2*10.52,0)+IF(D2=”加拿大元",E2*6.64,0)+IF(D2=”港元",E2*0.83,0)+IF(D2=”瑞士法郎",E2*7.37,0)+IF(D2=”瑞典克朗",E2*1.12,0)+IF(D2=”丹麦克朗",E2*1.23,0)+IF(D2=”挪威克朗",E2*1.17,0)+IF(D2=”澳门元",E2*0.81,0)”。这个公式使用了10个IF，并把这10个IF加在一起，每个IF只描述一个币种的处理方式，如果不是该币种则返回0，最后将这些IF加在一起得到折合人民币的金额。这样的写法将会造成10个IF中有且仅有一个IF函数中的条件判断结果为TRUE，从而得到金额，而另外9个关于其他币种的IF函数仅会等于0，9个0值与1个折合人民币金额相加，不会改变金额的数量。

需要注意，无论使用何种方式，回避IF函数最大嵌套数量限制，都仅是权宜之计，并不是主流的函数用法。上述3种方法都会造成公式书写过于繁复，不仅不利于公式的阅读，也会给公示排错造成困难。由于IF函数的特性，导致其高能却低效的特点，所以在使用IF函数时，如果遇到判断条件过多的情况，要积极寻找新的运算方式，例如后续章节中将会涉及到使用其他函数代替IF函数的问题。

注意：本节提到了对于IF函数的最大嵌套层次限制，Excel 2003为7层，Excel2007及Excel2010为63层。嵌套是指函数的某一参数由另外一个函数公式构成的情况，单独书写一个IF函数的公式并没有形成嵌套。以公式“=IF(B2="美元",A2*6.5,IF(B2="欧元",A2*9.3,IF(B2="英镑",A2*10.6,A2)))”为例，3个IF函数构成2层嵌套，描述4种可能。

2．分段比较的IF

IF函数的第一参数在绝大多数情况下都需要由一个条件判断公式构成，前面几个例子中都使用了“=”创建条件判断公式。IF函数同样也可以使用“>”、“<”、“>=”、“<=”等比较运算符对数字进行条件判断，用于处理与数据范围相关的运算。

如图3-10所示，B列数据中记录了销售业绩，在C列根据销售业绩为业务员评级。如果销售业绩大于或等于1000000评定为“优”；销售业绩介于1000000到800000之间（含800000）评定为“良”；销售业绩介于800000到500000之间（含500000）评定为“中”；销售业绩不足500000评定为“差”。

在C2单元格输入公式“=IF(B2>=1000000,"优",IF(B2>=800000,"良",IF(B2>=500000,"中","差")))”后下拉填充。这是一个根据数据所在范围返回结果的典型例子。在本例中，第二个IF函数在确定“良”的范围时，仅使用了“B2>=800000”的判断，这足以表示“1000000到800000之间（含800000）”的数据范围。因为在第一个IF函数的判断中，明确表述当数据大于或等于1000000时，将返回“优”，只有当数据小于1000000时，才运算第三参数的公式（“IF(B2>=800000,"良",IF(B2>=500000,"中","差"))”）。

这类公式在书写时，建议使用相同意义的比较运算符，例如本例中全部使用“>=”进行比较，反映出一种类似于逐层剥离的效果。如果将函数修改为“=IF(B2>=1000000,"优",IF(B2<500000,"差",IF(B2>=800000,"良","中")))”，其结果依然正确，但在逻辑关系上显得不够清晰，层次划分过多时较容易出现错误，不如逐层剥离的方式一目了然。




图3-10

3．IF函数中的AND、OR与乘、加

在IF函数第一参数的条件判断比较复杂的情况中，也经常会看到AND函数与OR函数的身影。在图3-11所示的数据中，根据性别及年龄判断员工是否可以退休。其中男性满60岁可以退休，女性满55岁可以退休。在看似复杂的条件判断中，实际上只有“是否可以退休”为“是”或“否”两种可能出现的情况，所以只需要使用一个IF函数进行处理，在C2单元格输入公式“=IF(OR(AND(A2>=60,B2="男"),AND(A2>=55,B2="女")),"是","否")”后下拉填充。

在这个公式中，IF函数的第一参数由4个条件判断根据其逻辑关系共同组成。大于或等于60岁与性别为男必须同时满足，表示满60岁的男性，使用AND函数相连，描述且的关系；大于或等于55岁与性别为女必须同时满足，表示满55岁的女性，使用AND函数相连，描述且的关系。而这两个AND函数之间又是或的关系，表示满60岁的男性或满55岁的女性都将在C列返回“是”，所以使用OR函数相连。

AND函数和OR函数分别表示“与运算”和“或运算”，布尔代数中布尔值在进行运算时，具备表3-1所示定律。

表3-1




表3-1中的布尔运算定律同样适用于Excel中的与运算和或运算，当我们对布尔值进行四则运算时， TRUE将被视为1，FALSE将被视为0，则有如表3-2所示的关系。

表3-2




根据以上计算方法，可以使用乘法、加法替换图3-11所示数据中的公式，公式“=IF(OR(AND(A2>=60,B2="男"),AND(A2>=55,B2="女")),"是","否")”可被替换为“=IF((A2>=60)*(B2="男")+(A2>=55)*(B2=”女"),"是","否")”。在这个公式中，两次条件判断相乘的公式 “(A2>=60)*(B2="男")”表示两次判断之间为且的关系，仅当两次判断的结果均为TRUE时，乘积才为1，即TRUE；而公式“(A2>=60)*(B2="男")+(A2>=55)*(B2=”女")”将两个乘积加在一起，表示这两个乘积中有任何一个不为0，相加的结果就不为0，即仅当所有条件判断的结果全部为FALSE时，才会返回结果FALSE，表示或的关系。

在处理多个条件判断时，使用加法替代OR函数有时也会发生判断的结果大于1的情况，例如图3-12所示情况，对年龄大于60岁员工或女性员工给予1000元奖金。在C2单元格输入公式“=IF((A2="女")+(B2>60),1000,0)”后下拉拖曳。

在表格第三行数据中，同时满足了性别为女性和年龄大于60这两个条件，此时表示条件判断的公式“(A3="女")”和“(B3>60)”相加的结果为2。在介绍TRUE与FALSE的章节里，我们提到过，在四则运算中，TRUE与FALSE将被视为1和0；而0被视为FALSE，其他有理数被视为TRUE，所以即使相加的结果为2，依然被视为TRUE，即布尔或运算中“1+1=1”的运算。




图3-12

注意：在非数组公式的创建中，我们可以使用乘法代替AND函数，使用加法代替OR函数；同样也可以使用AND函数代替表示逻辑判断的乘法，使用OR函数代替表示逻辑判断的加法。从一定意义上讲，在表示逻辑判断时，AND与乘法通用、OR函数与加法通用，但在数组公式中则不然，如果在数组公式中使用AND函数替换表示逻辑判断的乘法，或使用OR函数替换表示逻辑判断的加法，将会造成判断出错。





3.3.3 IF函数扩展应用




1．使用IF函数隐藏不需要计算的公式

在工作中，绝大多数数据都是逐渐增加的，反映在表格上，则意味着数据区域的行数增加。在数据不断增加的情况下，如果希望新数据也能被公式运算，我们经常会在公式下拉填充时多填充若干行，为即将增加的数据预留位置。但如果把不经过任何处理的公式拖曳到没有引用数据的行中时，将会返回一些我们不希望看到的值，包括0值、错误值等。

如图3-13所示，C2单元格公式为“=IF(B2="美元",A2*6.5,IF(B2="欧元",A2*9.3,IF(B2="英镑",A2*10.6,A2)))”，当前数据只有4行，为了达到输入新数据后就能运算折合人民币的效果，将公式下拉拖曳至C8，为新数据预留3行。但由于A6：B8区域不存在数据，导致公式返回0值，显得表格不够整洁。

这类情况不仅出现在IF函数的使用中，任何函数公式都可能遇到此类问题，解决的方法是增加一个IF函数的判断，用来识别引用区域是否有数据。如果有数据，则进行运算，否则返回空值。




图3-13

将C2单元格公式修改为“=IF(AND(A2<>"",B2<>""),IF(B2="美元",A2*6.5,IF(B2="欧元",A2*9.3,IF(B2="英镑",A2*10.6,A2))),"")”，下拉填充至C8单元格。其中“IF(B2="美元",A2*6.5,IF(B2="欧元",A2*9.3,IF(B2="英镑",A2*10.6,A2)))”是原有的公式，足以处理3个币种的折合人民币运算，在此公式外增加一层IF函数，则是为了判断同行的A列与B列单元格中是否有数据，如果都有数据，则进行运算，否则返回空文本“""”。

使用这种处理方式，相当于在原有问题基础上人为地增加一个判断条件，即无数据显示为空单元格，有数据才计算，避免了多余的0值和错误值的出现。

2．使用IF函数屏蔽错误值

通过前面的学习，我们了解到，IF函数是根据判断做出反馈的函数，所以第一参数必须是布尔值或可以得到布尔值的表达式。在前面的例子中，我们通过使用比较运算符、逻辑函数等方式得到布尔值，而实际上Excel中还有另一些函数也可以得到布尔值，例如信息类函数中的IS系列函数（以IS开头的函数）。

信息类函数中的IS系列函数主要包含以下函数。

ISBLANK：检测是否引用了空单元格。

ISERR：检测一个值是否为#N/A错误以外的错误值（#VALUE!、#REF!、#DIV/0!、#NUM!、#NAME?、#NULL）。

ISERROR：检测一个值是否为错误值（#N/A、#VALUE!、#REF!、#DIV/0!、#NUM!、#NAME?、#NULL!）。

ISLOGICAL：检测一个值是否为布尔值。

ISNA：检测一个值是否为#N/A错误。

ISNONTEXT：检测一个值是否不是文本。

ISNUMBER：检测一个值是否为数值。

ISREF：检测一个值是否为引用。

ISTEXT：检测一个值是否为文本。

此类函数在语义上都表示判断是否，其结果也将返回布尔值TRUE与FALSE。本节将以ISERR为例，与IF函数结合使用屏蔽公式计算中的错误值。

在图3-14所示的数据中，由于总价和数量中存在“无数据”的情况，导致直接使用总价除以数量的方式计算价格出现错误。在这种情况下，可以使用ISERR函数与IF函数结合，屏蔽“#VALUE!”错误。




图3-14

可将公式修改为“=IF(ISERR(B2/C2),"",B2/C2)”，这个公式表示如果“B2/C2”的运算结果返回了错误，则显示为空文本，否则显示“B2/C2”的结果。屏蔽错误值的方式有很多种，本例中使用检测运算结果是否为错误值的方式回避错误，当然也可以通过检测总价和数量是否有数值的方式达到同样的目的。





3.3.4 IF函数的常见组合




就与其他函数组合而言，IF函数可以算作当之无愧的万金油函数，它可以与任何类型的函数组合，帮助其他函数公式判断什么时候执行什么样的运算、屏蔽公式错误，甚至帮助其他函数生成引用区域。

无论在Excel里，还是生活中，我们每天都在不停地使用“如果……就……”的条件判断方式，所以后续章节会在各类函数中提及IF函数在与其他函数组合中作出的贡献，由于组合方式过多，本节暂不一一列举，请在后续学习中留意。





3.3.5 在公式中使用双引号




对于公式与函数的初学者而言，经常会困扰于何时在函数参数中使用双引号，一旦使用不当，将会带来公式错误，而能找到的答案也仅是“数值不加文本加”。本节将重点探讨公式与函数中双引号的使用，寻找规律，避免反复修改公式的尴尬。

寻找何时在公式与函数中使用双引号的规律，必须先了解双引号的最根本意义，在Excel中使用双引号，是为了标记字符串的可运算属性。当字符串在当前公式中可运算，也需要被运算时，不使用双引号，字符串将在被运算后显示其结果；当字符串在当前公式中不可被运算，也不需要被运算时，不使用双引号，将会显示其本身内容。字符串是否可运算，取决于我们的意愿。

如图3-15所示，A2:A9区域中的值由B2:B9区域同行公式得到。其中，“=a1”中“a1”如果不使用双引号，这是一个符合地址格式的字符串，表示这是一次引用，它也将被视为一次引用来运算，结果得到A1单元格中的内容“结果”；而A3单元格使用了双引号，字符串“a1”将不被运算而直接得到“a1”，即字符串本身的内容。所以公式内的引用，不能使用双引号。例如前文示例图3-8中的公式“IF(B2="美元",A2*6.5,IF(B2="欧元",A2*9.3,A2))”，在“B2="美元"”这次比较中，并不是用“B2”这两个字符与“美元”比较，而是要用B2单元格中的内容与“美元”比较，所以没有使用双引号。




图3-15

A4单元格公式与A5单元格公式相比，公式“=001”由于没有使用双引号，被视为可运算，运算后去掉了没有意义的0；而A5单元格公式由于使用了双引号，将不被运算而直接得到其本身内容“001”。就纯数字字符串使用双引号而言，符合“数值不加文本加”的规律。一般意义上讲，具有“大”、“小”、“多”、“少”属性、被四则运算后有意义的数字不能使用双引号；不具有“大”、“小”、“多”、“少”属性、表示代码、代号一类的数字需要使用双引号。这种数字分类方式，也正是单元格数字格式中数值类数字与文本型数字的分类方式。

A6与A7单元格内同是公式， A6单元格内公式“=2*5”没有使用双引号，被视为可运算，运算后得到公式“2*5”的结果；而A7单元格内使用了双引号，直接返回了字符串内容“2*5”。例如前文图3-8中的公式“IF(B2="美元",A2*6.5,IF(B2="欧元",A2*9.3,A2))”，如果B2单元格内容为美元，则希望得到“A2*6.5”的运算结果而不是公式本身字符串，所以没有使用双引号。

A8与A9单元格内分别输入了有双引号和没有双引号的“比例”，由于此前在该工作表上创建了名称“比例”为常量0.03，所以公式“=比例”将比例运算了一次，得到0.03；而公式“="比例"”中，由于使用了双引号，标记“比例”不可运算，从而返回了“比例”这两个文字。例如图3-10所示实例中的公式“=IF(B2>=1000000,"优",IF(B2>=800000,"良",IF(B2>=500000,"中","差")))”，当B2单元格中的数量大于或等于1000000时，需要返回文字“优”，而不需要对“优”进行运算，所以公式中使用了双引号。而实际上，在这种情况下，如果不使用双引号，公式将返回“#NAME？”错误，表示在运算“优”的时候，没有找到对应的函数或名称。





3.3.6 使用其他函数替代IF函数




由于IF函数在表示条件判断时描述方式比较单一、过于直白，导致其高能低效的特点，特别是在多条件的运算处理中力不从心，本节就使用其他函数替代IF函数的问题提出几个解决思路。

1．使用查找条件方式替代IF函数

IF函数是基于对多个条件的逐一判断处理数据的，由于需要逐一描述比较判断，所以不适合使用IF函数处理条件过多的问题。在图3-16所示的数据中，根据币种、金额，按照汇价计算折合人民币。由于判定条件达到10种，导致前面章节中使用IF函数处理比较困难，可以使用查找引用类函数找到币种对应的汇价，再使用查找引用函数返回汇价后，乘以数额得到折合人民币。




图3-16

例如本例中公式“=VLOOKUP(D2,$A$2:$B$11,2,0)*E2”，就使用了VLOOKUP这个非常常用的查找引用类函数，其中VLOOKUP(D2,$A$2:$B$11,2,0)表示在A2:B11区域中精确查找D2单元格内的币种，并返回其汇率。在这里， VLOOKUP(D2,$A$2:$B$11,2,0)的结果为7.37，即瑞士法郎的汇价，再乘以数额E2后得到其折合人民币金额。

在这个例子中，由于查找的对象为固定的几个币种，属于字符串的查找，所以使用了精确匹配。如果参照前文图3-10所示的数据，根据销售业绩评定等级，则需要根据数据范围进行条件判断，可使用VLOOKUP函数的模糊匹配。

2．使用寻找数据规律方式替换IF函数

在处理一些条件判断问题时，最重要的是找到多重条件间隐含的客观规律，只有找到这个规律，才能透过看似复杂的多重条件找到处理问题的便捷方法。

图3-17所示数据体现的是人事工作中一个较为常见的问题，在计算员工带薪假期时，由于计算因素较多（国家年假、公司福利假、已休假时间、工龄、累计假期等），经过计算会得到B列数据中含有小数的年假剩余天数，而公司中规定最小休假单位为半天，不足半天的部分将舍去，从而得到实际年假。




图3-17

如果使用IF函数进行处理，则需要判断计算假期数值的小数部分是否大于或等于0.5，如果大于或等于0.5，则返回0.5，并与整数部分相加得到实际年假；否则返回0，并与整数部分相加得到实际年假。例如李明的计算年假为2.83天，其小数部分0.83大于0.5，则计算年假为0.5与整数部分2的和，即2.5天。

经过观察，我们可以发现一些规律，实际年假的小数部分只可能为0和0.5，而0和0.5恰恰是整数除以2后其结果小数部分的值（奇数除以2，其结果的小数部分为0.5、偶数除以2，其结果的小数部分为0），那么公式的最后一步运算将会是一个整数除以2的运算。而根据公司要求，运算中包含着“舍去”的意义，所以使用INT函数取出数字的整数部分（该函数将对其参数直接取整，并非四舍五入）。因为最后一步运算中的“除以2”会使计算年假缩小一半，于是在处理前先将计算年假乘2抵消其缩小效果。

在C2单元格输入公式“=INT(B2*2)/2”后下拉填充。以汪梅年假为例，计算年假为1.6天，乘以2后得到3.2天，使用INT函数取整后得到3天，再除以2后得到1.5天年假；再以李明的年假为例，计算年假为2.83天，乘以2后得到5.66天，使用INT函数取整后得到5天，再除以2后得到2.5天。

3．利用其他函数特性替换IF函数

图3-18呈现的是提取身份证中性别信息的问题，由于15位身份证与18位身份证编码方式不同，二者性别位所在的位置也不同。对于15位身份证而言，最后1位是性别位，当性别位为奇数时，表示该身份证号的所有者为男性，偶数则为女性；而18位身份证的倒数第2位为性别位，当性别位为奇数时，表示该身份证号的所有者为男性，偶数则为女性。




图3-18

性别位上的数字同样都是奇数表示男性、偶数表示女性，所以判定方法相同，本例中重点探讨如何从两种类型的身份证中提取到性别位。

如果使用IF函数，则需要首先判断身份证类型是15位还是18位，再根据不同情况进行截取。而实际上，在B2单元格输入公式“=RIGHT(LEFT(A2,17),1)”就可以达到这个效果。其中LEFT(A2,17)表示将单元格文字自左向右截取17个字符，对于15位身份证而言，其文本长度不足17位，运算的结果为全部15个字符，而对于18位身份证而言，运算的结果为左侧17个字符，失去最后1个字符。再将这个加工过的字符串使用RIGHT函数自右向左截取1个字符，将得到15位身份证的最后一个字符和18位身份证的倒数第2个字符。

计算性别的完整公式为“=IF(MOD(RIGHT(LEFT(A2,17),1),2),"男","女")”，表示将性别位上的数字除以2后取余数，当余数为1时，表示该数字为奇数，返回“男”，否则返回“女”。

本节中列举了3个使用其他函数替代IF函数的例子，实际上，很多函数之间都可以相互取代，而每个函数特长不同，在某些情况下又是不可替代的。在前文提到的3个例子中，使用了几个没有在前面章节中介绍过的函数，这些函数都将在后续的章节中详细地介绍。





第4章 用函数统计和处理数值








4.1 单条件求和与计数








4.1.1 SUMIF单条件求和




现实工作中，经常会遇到这样的问题：图4-1所示是一张简易工资明细表，如果要计算各部门的工资小计，该用哪个函数呢？




图4-1

这是一个单条件求和的问题，不是要把所有的实发工资都加起来，而是要挑出同部门的进行求和，SUMIF函数可以解决这个问题。

假设把工资小计结果放在F2单元格里，则公式是：

=SUMIF(B$2:B$9,E2,C$2:C$9)

其余的公式只需要一个“下拉”操作就可以完成，如图4-2所示。




图4-2

无论员工总数是8人还是8000人，无论部门的排列是顺序的还是乱序，用这个公式都能在瞬间完成统计。

如果不涉及“下拉”，也可以将公式写成：

=SUMIF(B2:B9,E2,C2:C9)

关于绝对引用与相对引用，请参见第2章。

1．函数语法

SUMIF，用来对满足条件的单元格求和。一共有3个参数：

SUMIF(条件区域,条件,求和区域)

条件区域

是包含了求和条件的单元格区域，如图4-3所示。




图4-3

条件

是指需要满足的求和条件，如图4-4所示。




图4-4

求和区域

是需要进行求和的数据所在的单元格区域，如图4-5所示。




图4-5

2．扩展应用

条件变化

SUMIF的条件，可以是单元格引用，可以是字符串，也可以是字符串与单元格引用混合。引用字符串必须在外套一对半角双引用，字符串与单元格引用混合使用时，用连接符“&”连接。

SUMIF的条件除了使用固定的数字或文本外，还可以有所变化，这种变化通过两种符号体现出来：比较符号和通配符。比较符号就是大于小于等于之类的，各符号的写法已在错误！未找到引用源。小节中详细介绍，此处不再赘言，值得一提的是通配符。

通配符只有两个，如下所述。

代表0个或多个字符的半角星号“*”。

代表1个字符的半角问号“?”。

当这些符号搭配组合起来用时，其中精彩就尽显无疑。例如图4-6中，A1:D7是个各大区销量明细表，现在假设有两个问题摆在我们面前，如下所述。




图4-6

计算北区（A列地区中所有带“北”字）的总金额

这里的公式有几种写法，但都离不开通配符的使用。

公式一：

=SUMIF(A2:A7,"*北*",D2:D7)

一般情况下，带“北”字的条件写法是“*北*”，表示条件是包含“北”字的字符串，不管“北”字在这个字符串中的哪个位置，也不管这个字符串有多长（哪怕就一个“北”字）。

公式二：

=SUMIF(A2:A7,"*北",D2:D7)

本例中的“北”都是最右字符，所以后面那个*可以不要，直接写成“*北”，表示条件是以“北”字结尾的字符串，不管这个字符串有多长。

公式三：

=SUMIF(A2:A7,"?北",D2:D7)

本例的区域又有个特殊性，即所有的地区都只有两个字，那么条件也可以写成“?北”，表示条件只有两个字，并以“北”字结尾。

图4-7展示了以上3种公式的写法，至于具体用哪一个，可以根据实际情况和个人爱好任选。




图4-7

计算单价在平均单价35元以上的总金额

在这个公式中，条件就要写成“>35”，同理，条件以字符形式出现时，必须要用半角引号引用。

=SUMIF(C2:C7,">35",D2:D7)

但是作为数字本身是可以不加引号直接出现在公式中的，所以这里的条件也可以写成“">"&35”。

=SUMIF(C2:C7,">"&35,D2:D7)

大多数人会选择第一种写法，但如果数字“35”是放在单元格A15里被引用的，那就只能用第2种写法了：

=SUMIF(C2:C7,">"&A15,D2:D7)

如果B15单元格里的内容就是“>35”，那也可以直接把公式写成：

=SUMIF(C2:C7,B35,D2:D7)

各种写法的效果如图4-8所示。




图4-8

假如平均值是未知的，可以将一个计算单价平均值的函数AVERAGE作为条件，公式如下：

=SUMIF(C2:C7,">"&AVERAGE(C2:C7),D2:D7)

结果如图4-9所示。




图4-9

区域变化

整行或整列引用

SUMIF的区域参数可以整行或整列引用。例如图4-10的公式，如果B列没有其他不相干的内容打扰，公式也可以写成：

=SUMIF(B:B,E2,C:C)




图4-10

SUMIF不会因为引用整行或整列而影响运算速度。

条件区域与求和区域重叠

条件区域和求和区域的用法基本一样，当两者完全重叠时，求和区域可以省略。如图4-11所示，假如要计算各地区金额大于100的总金额，完整公式应写成：

=SUMIF(D2:D7,">100",D2:D7)




图4-11

但这里的条件区域和求和区域完全重叠了，公式就可以简化为：

=SUMIF(D2:D7,">100")

条件区域与求和区域错位

如图4-12所示条件区域和求和区域“错位”了，公式写成：

=SUMIF(B2:B9,E2,C3:C9)




图4-12

得出的结果，就不再是“生产部”对应的那5个单元格之和，而是都在“生产部”对应的C列单元格向下错了一位，实际计算了C3、C4、C5、C7和C8这5个单元格的和。

又假如我们把公式写成：

=SUMIF(B2:B9,E2,C3)

最后得出的结果和上面一个公式是一样的，可见SUMIF并不在乎求和区域有多大，只要最左上角位置定下来即可。

3．常见错误

如图4-13所示，A1:B7是日销量的流水账，现在要算每个月的销量总和。




图4-13

按照SUMIF思路，条件区域不能直接是A2:A7这个区域，因为条件是按“月”组合的，需要修改一下，用MONTH(A2:A7)得出A2:A7这个区域内每个单元格所在的月份，最终汇总1月份销量的公式写成：

=SUMIF(MONTH(A2:A7),1,B2:B7)

这个公式看上去好像没有任何问题，可是放进单元格却总是提示错误。

原来SUMIF的区域参数，包括条件区域和求和区域，必须是单元格区域的引用，其他一概出错，这是我们在学习中需要非常注意的一点。

这个例子用SUMIF解决就有难度了，因为它骨子里是多条件求和。统计1月的销量就是日期大于12月31日且小于2月1日的销量，关于多条件求和的解法，会在后面的小节中详细说明。

4.1.1.4 常用组合

有一种SUM和SUMIF的数组公式组合，如：

=SUM(SUMIF(A:A,">"&{2,8})*{1,-1})

其实就是：

=SUMIF(A:A,">2")-SUMIF(A:A,">8")

这种写法涉及到常量数组的应用，不属于本书范围，略有了解即可。

另外，SUMIF的区域参数可以是间接引用，所以也经常和INDIRECT、OFFSET等引用函数组合，有关这两个引用函数详见第5章。





4.1.2 COUNTIF单条件计数




现在，问题升级：如图4-14所示是一张简易花名册，如果要计算各部门各有多少人，该用哪个函数呢？




图4-14

这是一个单条件计数的问题，不是要统计一共多少人，而是要挑出同部门的人进行计数，COUNTIF函数可以解决这个问题。

假设把人数统计结果放在F2单元格里，则公式是：

=COUNTIF(B$2:B$9,E2)

其余的公式只需要一个“下拉”操作就可以完成，如图4-15所示。




图4-15

COUNTIF，用来对满足条件的单元格计数。它一共有两个参数：

COUNTIF(条件区域,条件)

看到这里，是不是有种“似曾相识”的感觉？

是的，COUNTIF和SUMIF非常相似，它们只有以下两个区别。

统计结果不同

SUMIF是求和，而COUNTIF是计数。不过这并不意味着SUMIF就不能计数，如果SUMIF的求和区域全部是数字“1”，那结果就是计数了，如图4-16所示。




图4-16

不过反之，要让COUNTIF求和，那是万万不能的。

参数数量不同

SUMIF有一个“求和区域”参数，而COUNTIF当然不存在“求和区域”一说，由此引出的“区域错位”，对COUNTIF来说也不存在。

当SUMIF在特殊情况下省略“求和区域”时，两者的参数结构完全一样，如图4-17所示。




图4-17

其他的参数特性、条件变化，以及常见错误等，二者完全相同。只要掌握了SUMIF函数，也就会很快掌握COUNTIF函数。





4.2 多条件求和与计数








4.2.1 SUMIFS和COUNTIFS条件求和计数




现在，问题进一步升级：如图4-18所示是一张简易工资明细表，如果要计算各部门、各职级的工资小计，以及各部门、各职级的人数，该用哪些函数呢？




图4-18

这是一个多条件求和与计数的问题，不是单一地按部门或者按职级设置条件进行统计，而是部门和职级分别满足两个指定条件时进行求和与计数，SUMIFS函数和COUNTIFS函数可以解决这些问题。

1．SUMIFS条件求和

假设把工资小计结果放在H2单元格里，则公式是：

=SUMIFS(D$2:D$9,B$2:B$9,F2,C$2:C$9,G2)

其余的公式只需要一个“下拉”操作就可以完成了，如图4-19所示。




图4-19

SUMIFS，用来对满足一个或多个条件的单元格求和。它的参数最少3个，最多255个，换言之，SUMIFS的条件，最少是1个，最多可达127个：

SUMIFS(求和区域,条件区域1,条件1,条件区域2,条件2……)

2．COUNTIFS条件计数

继续图4-18所示的问题，假设把人数统计结果放在I2单元格里，则公式是：

=COUNTIFS(B$2:B$9,F2,C$2:C$9,G2)

其余的公式也同样只需要一个“下拉”操作就可以完成了，如图4-20所示。




图4-20

COUNTIFS，用来对满足一个或多个条件的单元格计数。它参数最少是两个，最多254个，和SUMIFS一样，条件可以是1至127个：

COUNTIFS(条件区域1,条件1,条件区域2,条件2……)

3．SUMIFS和COUNTIFS的特点

这两个函数，就好像SUMIF和COUNTIF的“复数形式”，很多特性如出一辙，只要掌握了SUMIF函数和COUNTIF函数，再学习这两个函数就没有难度。不过，这两个函数还是存在一些区域别于SUMIF和COUNTIF的特点，如下所述。

SUMIFS和COUNTIFS是2007版和2010版专用函数

2003版没有这两个函数。

如果电脑中未安装2007版或2010版，在输入这两个函数时，单元格里会出现“#NAME?”。如果安装了可以打开高级版本的补丁，打开在2007版或2010版中已设置好这两个公式的工作簿，编辑栏里的公式前面就会多出“_xlfn.”部分，如图4-21所示。




图4-21

而且这个公式只能看不能改，双击或者按功能键【F2】进入单元格后再按【Enter】键，结果就会变回“#NAME?”，如图4-22所示。




图4-22

这也就意味着，即使电脑里装有可以打开高级版本的补丁，这两个函数仍然不能使用。所以，在2003版尚未完全退出市场之前，使用这两个函数仍有很大限制。

SUMIFS函数的参数顺序与SUMIF函数不同

SUMIF函数的求和区域是第3个参数，而SUMIFS的求和区域被移至第1个，如图4-23所示。这点小小的差异不并涉及到参数特性的改变，只要在使用过程中稍加注意即可。




图4-23

SUMIFS函数不能省略求和区域参数

SUMIFS函数参数最少是3个，即求和区域、条件区域和条件，在求和区域和条件区域完全重叠时，SUMIF可以写成：

=SUMIF(B2:B7,D2)

但是，SUMIFS却不能写成：

=SUMIFS(B2:B7,D2)

如图4-24所示。




图4-24

如果执意在图4-24所示的单元格D8中写入这样的公式，按下【Enter】键后，就会看到如图4-25所示的对话框。




图4-25

SUMIFS函数和COUNTIFS函数中所有区域参数大小都必须一致

所谓两个区域大小一致，是指这两个区域所包括的行数和列数都完全相同。如图4-26所示，SUMIF的求和区域参数，只要确定最左上角的位，即B1单元格，就会得出正确结果。而在SUMIFS中，如果将B1作为求和区域，或者将与条件区域A1:A7大小不一致的B2:B7作为求和区域，结果都会出现错误。因为求和区域是7行1列，条件区域也必须是7行1列。把求和区域改成B1:B7，或者把条件区域改成A2:A7，求和区域仍用B2:B7，计算结果才不会出现错误。




图4-26

SUMIFS函数的各区域参数，只要在大小一致的前提下，区域错位仍可以计算，只是计算的结果也是“错位”的，这一特性与SUMIF一样，如图4-27所示。




图4-27

COUNTIFS也一样，各条件区域可以错位，但大小必须完全一致。





4.2.2 SUM无条件求和与有条件求和计数




因为SUMIFS和COUNTIFS的使用存在着很大的限制，如图4-28所示是一张简易工资明细表，如果要计算各部门、各职级的工资小计，以及各部门、各职级的人数，在2003版中该用哪些函数呢？




图4-28

我们在1.3.3小节学习输入一个简单函数公式时，曾以SUM函数为例对一组数据进行求和。可见，SUM函数的基本用法是不论条件的求和，只要在其“势力范围”内的参数统统会被加到一起。如果要让SUM函数实现带条件求和计数的功能，需要对其参数进行一些小小的“改造”。

假设把工资小计结果放在H2单元格里，则公式是：

{=SUM((B$2:B$9=F2)*(C$2:C$9=G2)*D$2:D$9)}

把人数统计结果放在I2单元格里，则公式是：

{=SUM((B$2:B$9=F2)*(C$2:C$9=G2))}

如图4-29所示，公式前后多出了一对大括号，这对大括号并不是手工输入的，而是在输入公式以后，一改传统的直接按【Enter】键结束的操作，同时按下【Ctrl】、【Shift】、【Enter】合键结束后自动生成的。这就是与普通公式相区别的“数组公式”。数组公式要得到正确的结果，必须以上述组合键结束。




图4-29

注意，使用组合键必须是在单元格处于“编辑”状态才有效，也即光标可以在这个单元格内的各字符间移动，而非光标可以在不同单元格间移动。

如果不想用组合键，或者公式放在合并单元格里无法使用组合键，也可以把SUM函数换成SUMPRODUCT函数 

1


 ，即公式写成：

1 SUMPRODUCT函数是普通函数，用来计算数组结果，其实质是数组计算。

=SUMPRODUCT((B$2:B$9=F2)*(C$2:C$9=G2)*D$2:D$9)

=SUMPRODUCT((B$2:B$9=F2)*(C$2:C$9=G2))

其余的公式只需要一个“下拉”操作就可以完成。

1．公式结构

SUM数组或SUMPRODUCT的这一用法，可以对满足一至多个条件的单元格求和或计数。

求和公式结构如下：

SUM((条件区域1=条件1)*(条件区域2=条件2)*……*(条件区域n=条件n)*求和区域)

计数公式结构如下：

SUM((条件区域1=条件1)*(条件区域2=条件2)*……*(条件区域n=条件n))

求和公式仅比计数公式多了个“*求和区域”，其余部分完全一样。

这里的SUM函数里实质上只包含一个参数，因为中间用的是乘号“*”，而不是用于分隔各参数的半角逗号“,”，所以条件个数可以说是无限的，只要有这个耐心输入。

2．原理分析

我们仍以图4-28所示为例。

计数的公式

{=SUM((B2:B9=F2)*(C2:C9=G2))}

其中的“(B2:B9=F2)”是一个判断过程，让B2:B9这个区域内每一个值都和F2单元格的值比较是否相等。如果相等，则得出一个“逻辑真”值，即“TRUE”；反之不相等，则得出一个“逻辑假”值，即“FALSE”。

“(C2:C9=G2)”部分也是同样的原理，如图4-30所示。




图4-30

而逻辑值“TRUE”和“FALSE”在Excel里是可以参与计算的，规律如下。

“TRUE”参与计算时会被当作数字“1”。

“FALSE”参与计算时会被当作数字“0”。

根据小学算术的常识，我们知道，任何数乘以0的结果为0；任何数乘以1的结果，还是原来的数。据此，当“(B2:B9=F2)”乘以“(C2:C9=G2)”以后，就会得出如图4-31的结果。




图4-31

从图4-31中我们可以看出，满足所有条件的结果，即“TRUE”乘以“TRUE”的结果都是1，只要有一个条件不满足的，无论是“TRUE”乘以“FALSE”，还是“FALSE”乘以“TRUE”，还是“FALSE”乘以“FALSE”，结果都是0。

最后再把这些“1”和“0”相加，就是我们需要计算的计数。

求和的公式

{=SUM((B2:B9=F2)*(C2:C9=G2)*D2:D9)}

这个公式仅仅比计数的公式多了个“*求和区域”，这仍是利用任何数与1或0相乘规律，当计数公式进行到图4-31所示的结果时，不急于将这些“1”和“0”相加，而是再乘以求和区域，如图4-32所示。




图4-32

从图4-32中我们可以看出，满足所有条件的结果，即与“1”相乘的结果，就是对应的求和区域的结果，而不满足条件的结果，即与“0”相乘的结果，都是0。

最后再把这些数字和“0”相加，就是我们需要计算的求和。

3．扩展应用

条件变化

条件内容变化

SUM条件求和计数公式里的条件，可以是单元格引用，可以是字符串，可以是函数，也可以是各种混用。引用字符串必须再外套一对半角双引用，字符串与单元格引用混合使用时，用连接符“&”连接。

在如图4-33所示的表格中， A1:D7是个各大区销量明细表，条件有3种，分别列在A10、B10和C10单元格中。




图4-33

当给定的条件如A10单元格中所示，可以直接引用这个单元格，公式写成：

{=SUM((A2:A7=A10)*D2:D7)}

或者条件直接用字符串“华北”来表示，公式如下：

{=SUM((A2:A7="华北")*D2:D7)}

当给定条件有一部分在B10单元格中，另一部分需要用字符串来表示时，公式可以写成：

{=SUM((A2:A7="华"&B10)*D2:D7)}

当条件如C10单元格里的内容，比条件区域A2:A7里的内容“多”出来的时候，可以套用一个LEFT函数将C10单元格中的“华北”提取出来作为条件，公式如下：

{=SUM((A2:A7=LEFT(C10,2))*D2:D7)}

各种写法的效果如图4-34所示。




图4-34

条件符号变化

在“SUM((条件区域1=条件1)*(条件区域2=条件2)*……”这个公式里，“条件区域”和“条件”之间不一定非要 “相等”，也可以根据实际情况使用其他比较运算符号，例如不等于、大于等。

如图4-35所示的例子，如果要计算单价在35元以上的总金额，可以直接用大于号“>”连接条件区域和条件，公式如下：

{=SUM((C2:C7>A19)*D2:D7)}

但如果条件如B19单元格所示，已经带比较运算符，公式里条件区域直接等于条件，就会因为“找不到相同”而得出结果“0”。这时候必须套用RIGHT函数，提取出“>35”这个字符串中的“35”，并用“乘以1”的方式将RIGHT函数得出的文本型数字转化成数值型数字，才能计算出我们想要的结果，公式如下：




图4-35

{=SUM((C2:C7>(RIGHT(B19,2)*1))*D2:D7)}

各种写法效果如图4-36所示。




图4-36

可见在SUM条件求和计数的公式里，条件无论以什么样的形式出现，必须在条件区域内找到完全相同项，否则只会得出“0”。同理，在SUM条件求和计数公式里，不能使用通配符，详见本章4.2.2小节中的“4．常见错误”。

域变化

区域内容

SUM条件求和计数公式里的区域，包括条件区域和求和区域，可以是单元格引用，可以是数组运算的结果，也可以套用各种函数，不过这些都是函数进阶阶段才需要学习的内容，本书就不作介绍了。

区域错位

如图4-37所示，公式中的两个条件区域“错位”了，公式写成：

{=SUM((B2:B9=F2)*(C3:C10=G2)*D2:D9)}




图4-37

那么同时满足F2和G2两个单元格条件的结果，不再是D4单元格和D6单元格之和，而变成了D2单元格的值。原因是，第2个条件整体向下错了一位，但是，公式的计算过程仍然按照自身的规律执行，如图4-38所示。




图4-38

求和区域扩展

如图4-39所示的例子，A1:D7区域是各产品在不同产地不同型号的产量，其中产品型号分成两类，分别放在C2:C7和D2:D7两个区域里了。




图4-39

如果要计算上海产的电冰箱的总产量，公式难道要写成这样？

{=SUM((A2:A7=A10)*(B2:B7=B10)*C2:C7)+SUM((A2:A7=A10)*(B2:B7=B10)*D2:D7)}

试想一下，如果产品型号有上百种，那这个求和公式将是多么壮观？在这种情况下，我们可以将求和区域扩展，公式简化成：

{=SUM((A2:A7=A10)*(B2:B7=B10)*C2:D7)}

条件区域变化

如果两个条件分别是以行和列的形式出现的，表格结构如图4-40所示，是否还能用SUM条件求和计数的公式来统计呢？




图4-40

答案是肯定的，我们可以将公式写成：

{=SUM((A2:A7=A10)*(B1:C1=B10)*B2:C7)}

综上所述，条件区域和求和区域都可以有所变化，不像SUMIFS和COUNTIFS那样死板，但这并不意味着可以随意扩展，如果胡乱扩展，只会得到一个错误的结果，本章的4.2.2小节中的“4．常见错误”将详细介绍这方面的内容。

4．常见错误

使用通配符

在SUM条件求和计数公式中，条件使用通配符无效，如图4-41所示。




图4-41

SUM条件求和计数公式，是通过等号“=”连接条件区域与条件，既然是“等”，那么等式两边必须完全相同才能得到逻辑真“TRUE”的结果，而除了“*北”本身以外，其他无论是什么“北”，都无法与“*北”等同起来。所以，条件中使用通配符无效。

遇到这类情况，只能用RIGHT函数分别提取条件区域和条件最右字符，以这样的公式实现：

{=SUM((RIGHT(A2:A7)=RIGHT(A10))*D2:D7)}

整行或整列引用

在2003版中，SUM条件求和计数公式对区域不可以整行或整列引用。例如图4-42中H2单元格的公式如果写成如下的整列引用：

{=SUM((B:B=F2)*(C:C=G2)*D:D)}

就会出现如图4-42所示的错误。




图4-42

求和区域包含文本

那么在2007版或2010版中，是否就可以整行或整列引用了呢？是的。

但现实的具体使用过程中，我们仍然无法真正做到整行或整列引用，因为SUM条件求和计数公式对求和区域有特别的要求，这个区域内必须全部是数字，不可以有文本的内容，否则就会出现如图4-43所示的错误。




图4-43

一般情况下，表格总得有个“表头”，而“表头”大多是以文本形式出现，如此一来，即使在2007版或2010版中可以整列引用，现实中也很难用得起来，除非如图4-44所示不设置表头，或者将表头设置成数值，才能得到正确的结果。




图4-44

上述这种不设置表头，用Excel术语描述，就是将表头设置为“真空”。所谓“真空”，就是单元格和编辑栏里都没有任何字符，真真正正的“空”。

与“真空”相对的就是“假空”。所谓“假空”就是用公式得出的空值，说得通俗点就是编辑栏里有内容，单元格里是空。“假空”也会造成SUM条件求和计数公式结果错误。所以，用于求和的区域如果是通过公式得出的，需要避免生成“假空”值。最简单地生成“假空”值的方法，就是输入公式“=""”，如图4-45所示。




图4-45

另外，空格在Excel中也是一个“文本字符”，所以求和区域里如果出现空格，结果也会出错，而往往这类空格都是因误操作而输入，经常造成很大的困惑。

区域随意扩展

在本章4.2.2小节的“3.扩展应用”中，我们讲到SUM条件求和计数公式的条件区域和求和区域可以扩展变化，但是这种扩展必须遵循以下规律，否则就会出现错误。

条件区域是多行1列的，求和区域只能在列方向扩展，反之只能在行方向扩展。

如图4-46所示，条件区域是A2:A10这样一个9行1列的区域，求和区域无论如何变化，行数必须是同样的9行。




图4-46

条件区域都是1行或者1列的，所有条件区域的大小必须一致

如图4-47所示，条件区域是A2:A10这样的一个9行1列的区域，而其他条件区域没有一行多列的形式，那么所有的条件区域都必须同样是9行1列。




图4-47

条件区域分别是以行和列的形式出现时，成行的条件区域大小必须一致，成列的条件区域大小必须一致，求和区域的行列数与条件区域的最大行列数一致

如图4-48所示，条件区域1是A2:A10这样一个9行1列的区域，条件区域3是B11:D11这样一个1行3列的区域，那么再增加的条件区域，要么是9行1列，要么是1行3列，而求和区域必须是9行3列，不可扩展或缩小。




图4-48

另外还有一些扩展，需要再套用MMULT等复杂函数，这是进阶阶段才需要学习的函数，本书就不作介绍了。





4.3 其他带条件的统计








4.3.1 用求和除以计数计算平均




问题继续升级，如图4-49所示的是一张简易工资明细表，如果要计算各部门、各职级的平均工资，该用哪个函数呢？




图4-49

这是一个多条件平均的问题，在一时想不起求平均值用哪个函数时，完全可以用求和除以计数。假设把平均工资结果放在H2单元格里，则公式是：

{=SUM((B$2:B$9=F2)*(C$2:C$9=G2)*D$2:D$9)/SUM((B$2:B$9=F2)*(C$2:C$9=G2))}

其余的公式只需要一个“下拉”操作就可以完成了，如图4-50所示。




图4-50





4.3.2 AVERAGEIF和AVERAGEIFS条件平均




除了这个略显麻烦的求和除以计数以外，AVERAGEIF和AVERAGEIFS也可以进行条件平均，只不过，这两个函数都是2007版或2010版函数。

AVERAGEIF的用法和SUMIF完全相同；AVERAGEIFS的用法和SUMIFS完全相同，这些函数可谓是一个“系列”，学会了其中一个，其他基本上就可以触类旁通了。

唯一需要注意的是，因为被除数为0时会出现错误结果，所以在使用这两个函数时，需要避免没有满足条件的结果时的情况，如图4-51所示，A17单元格里的结果就是错误，因为条件区域A2:A9里没有满足条件的“生产部”。




图4-51





4.3.3 AVERAGE无条件平均和有条件平均




1．公式结构

在2003版尚未完全退出市场之前，使用AVERAGEIF和AVERAGEIFS这两个函数有很大限制，这时我们可以用AVERAGE进行有条件平均的计算。

AVERAGE本身是一个求平均的函数，它的基本用法和SUM完全一样。在进行条件平均计算时，AVERAGE的用法是不是也和SUM的条件求和用法一样呢？如图4-52所示。




图4-52

很不幸，H2单元格里所得出的并不是我们所期待的结果。

原因何在，这还要从源头分析：公式中的参数计算的结果是，符合条件的结果为D2和D7两个单元格的值，而其他不符合条件的结果就是0。0在求和中是不会产生任何影响，但是在计算平均时，就会影响结果了，所以我们需要将0变成不会被AVERAGE计算的值，这里就需要套用一个IF公式结构如下：

AVERAGE(IF((条件区域1=条件1)*(条件区域2=条件2)*……*(条件区域n=条件n)，平均区域))

这同样是个数组公式，需要按组合键结束。

将这个公式结构应用到上面的例子里，公式写成：

{=AVERAGE(IF((B2:B9=F2)*(C2:C9=G2),D2:D9))}

2．原理分析

我们知道，上述公式中“(B2:B9=F2)*(C2:C9=G2)”部分得出的结果，是一些1和0的数字，条件都满足的结果为1，只要有一个条件不满足的结果为0，如图4-53所示。




图4-53

这时将“(B2:B9=F2)*(C2:C9=G2)”作为IF的第一个参数，将“D2:D9”作为IF的第二个参数，IF会根据第一个参数的结果进行判断，第一个参数为1时，得到的结果是“D2:D9”所对应的位置；第一个参数为0时，得到的结果则是FALSE，如图4-54所示。




图4-54

因为AVERAGE计算过程中忽略逻辑值，所以那些“FALSE”不会从中“捣乱”，公式最后可以得出正确的结果，如图4-55所示。




图4-55

AVERAGE条件平均公式的扩展和常见错误都和SUM条件求和计数完全一样。





4.3.4 其他带条件的统计




统计函数套用IF函数的结构

因为逻辑值也不会影响统计，所以，除了平均之外，许多统计公式，如求和计数、求最大值的MAX、求最小值的MIN、求乘积的PRODUCT等，都可以用和AVERAGE条件平均的公式一样的结构。

条件求和的公式结构：

SUM(IF((条件区域1=条件1)*(条件区域2=条件2)*……*(条件区域n=条件n)，求和区域))

条件计数的公式结构：

SUM(IF((条件区域1=条件1)*(条件区域2=条件2)*……*(条件区域n=条件n)，1))

条件最大值的公式结构：

MAX(IF((条件区域1=条件1)*(条件区域2=条件2)*……*(条件区域n=条件n)，求最大值的数据区域))

条件最小值的公式结构：

MIN(IF((条件区域1=条件1)*(条件区域2=条件2)*……*(条件区域n=条件n)，求最小值的数据区域))

条件乘积的公式结构：

PRODUCT(IF((条件区域1=条件1)*(条件区域2=条件2)*……*(条件区域n=条件n)，求乘积的数据区域))

……

统计函数不套用IF函数的结构

针对有些统计，0在其中不会影响到结果，除了本章4.2.2小节所介绍的用于求和计数的公式结构外，求最大值、乘积等，也可以用这类公式结构。

条件最大值的公式结构：

MAX((条件区域1=条件1)*(条件区域2=条件2)*……*(条件区域n=条件n)*求最大值的数据区域)

条件乘积的公式结构：

PRODUCT((条件区域1=条件1)*(条件区域2=条件2)*……*(条件区域n=条件n)*求乘积的数据区域)

……

以上所有的公式均为数组公式，需要按组合键结束。





4.3.5 数组条件统计公式的缺点




这一系列的条件统计公式，因其灵活和在所有版本都通用，无疑给我们的使用带来方便。但是，这些公式也存在着一些不可避免的缺点，在使用过程中不可忽视。

运算速度慢

相比之下，同样是条件求和，SUM的数组要比SUMIFS慢很多，而且这种“慢”与数据量成正比，数据量越大，使用过程中就越能感受到这种“慢”，有时每输入一个数据，都要等待一两秒钟的时间后才能继续输入。无奈之下，只好将表格设置成“手动重算”。

2003版：“工具”→“选项”→“重新计算”→“手动重算”命令

2007版或2010版：“公式”→“计算”→“计算选项”→“手动”命令

当需要计算时按功能键【F9】重新计算整个工作簿，或者同时按下【Shift】键和功能键【F9】，重新计算当前工作表。

可是，表格在打印之前并不会自动重新计算，如果忘记重新计算就按下“打印”按钮，悲惨的事就一定会发生。

公式太长

因为公式的长度是与条件的多少成正比的，所以，当条件非常多时，整个公式就会变得很长。不要说编辑过程麻烦，就是编辑好了放在单元格里，每次一选取带这个公式的单元格时，都觉得非常“触目”。更重要的是，在2003版里，选取了带太长公式的单元格，还会影响操作，如图4-56所示，选取H2单元格后，前3行的内容都被挡住了。

虽然这个问题在2007版或2010版里已得到解决，但每次都要拖放调整编辑栏的高度，终究让人觉得不便，如图4-57所示。




图4-56




图4-57





4.4 数据库函数用于条件统计




问题进一步升级，同样是进行条件统计，如果既不想被又长又慢的数组公式拖住，又不想使用2007版或2010版；或者即使使用了高级版本，但需要计算的是最大值、最小值之类的，任何一个版本都没有MAXIF(s)、MINIF(s)之类的公式。这种情况下，该用哪些公式呢？

数据库函数可以解决这个问题。





4.4.1 数据库函数的成员




在函数类别中，有一个类别，一共包括12个函数。虽然这12个函数功能各有不同，但是其参数和各参数的使用方法完全一样，这就是“数据库函数”。在插入函数的对话框里，我们可以从“或选择类别”下拉列表框中找到“数据库”，如图4-58所示。




图4-58

除了参数及用法完全相同外，这组函数还有个特点，就是函数名全部以字母“D”开头（D为“数据库”Database的首字母），后面跟的单词正好是与其功能相对应的函数名，例如“DAVERAGE”就是用于求平均（AVERAGE）的数据库函数；“DSUM”则是用于求和（SUM）的数据库函数，等等。





4.4.2 数据库函数的用法




因为12个数据库函数的用法都完全一样，所以这里，我们以DSUM求和为例来说明全部数据库函数的用法。

1．函数语法

DSUM，用来对满足某区域条件的单元格求和。它一共有3个参数：

DSUM(数据库,求和列表头或位置,条件)

数据库

就是指整个数据源表，如图4-59所示。




图4-59

求和列表头或位置

指用于求和的那一列的表头，如图4-60所示的求和一列是D列，其表头就是D1单元格。




图4-60

另外，这个参数也可以用数字来表示，这里的D列位于整个数据库（即第1个参数）的第4列，所以这个公式也可以写成：

=DSUM(A1:D9,4,F1:G2)

条件

即需要满足的条件，如图4-61所示。




图4-61

这里的“条件”严格来说是一个“区域”，除包括若干个“条件”外，还包括了这些条件所在的“表头”。例如图中的条件应是“生产部”和“X”，它们分别属于“部门”和“职级”，所以，整个条件参数包括了4个单元格的内容，即条件1“生产部”和其表头“部门”，条件2“X”和其表头“职级”。

2．扩展应用

整行或整列引用

DSUM的第1个参数可以整行或整列引用。例如图4-62所示的公式，如果A:D列没有其他不相干的内容打扰，也可以写成：

=DSUM(A:D,D1,F1:G2)




图4-62

DSUM不会因为引用整列或整行而影响运算速度。

条件变化

DSUM的条件，必须是单元格的引用，不可以是常量数组。这也就注定了数据库函数的一个缺点：必须使用辅助单元格。

DSUM条件的变化是非常丰富多彩的，“与”条件和“或”条件都可以用，还可以使用比较运算符和通配符，甚至还可以在条件所在的辅助单元格区域内设置公式。

“与”条件

上述所有例子，都是“与”条件，即必须同时满足所列条件。使用“与”条件时，作为条件的辅助单元格区域内的所有条件是横向排列，如图4-61所示。

“或”条件

只要满足若干条件中的至少一个，也就是“或”条件，公式的写法依然不变，只是将条件所在的辅助单元格区域的结构稍做改变。如图4-63所示的例子，需要计算生产部或者市场部实发工资的小计，公式没有变化，仍是3个参数：

=DSUM(A1:D9,D1,F1:F3)

但条件的结构发生了变化，原来横向排列的变成了纵向排列，如图4-63所示。




图4-63

当然，“与”条件和“或”条件还可以混合起来用。如图4-64所示的例子，是计算生产部X级和市场部Z级人员的工资小计。




图4-64

比较运算符和通配符

DSUM函数的条件也可以用比较运算符和通配符。如图4-65所示的例子，需要计算的是所有北区且单价在35元以上的金额小计，公式仍没有任何变化，只是在条件辅助单元格区域里使用了比较运算符和通配符。




图4-65

设置公式

DSUM函数的条件还可以直接用公式，如图4-66所示。




图4-66

图中工资小计的公式仍是毫无悬念的：

=DSUM(A1:D9,D1,F1:F2)

但是条件发生了很大的变化，条件的“表头”（F1单元格）不见了，只剩下个光秃秃的空单元格；条件的“内容”变成了一个结果为“TRUE”的公式：

=AND(B2="生产部",C2="X")

这个公式本身并不难理解，就是同时满足部门为“生产部”，职级为“X”时，结果为逻辑真“TRUE”，只要有至少一个条件不满足，结果就为逻辑假“FALSE”。

由此可见，用公式来设置条件时，条件必须是一个至少两行的区域，表头为空，内容为一个结果为逻辑真，即“TRUE”或大于0的整数数字；或者逻辑假，即“FALSE”或0的公式。

关于逻辑判断公式，如果记不太清楚了，那就去复习一下第3章吧。

假如条件所在的辅助单元格区域全部是空的，结果又将如何呢？如图4-67所示。




图4-67

结果变成了无条件求和。

所以，DSUM的条件所在的辅助单元格区域也可以整列引用，不过这样一来，计算的结果会出现点意外，原因就是条件所在的辅助单元格区域纵向排列就是“或”条件求和，其中一个条件是“无条件”了，结果自然就出现偏差了，试试将图4-62所示的公式写成：

=DSUM(A:D,D1,F:G)

看看结果是不是和“无条件”求和一样了。

不过，因为横向排列是“与”条件求和，所以条件所在的辅助单元格区域被整行引用，不会出现任何偏差。

仔细回味一下，数据库函数的用法是不是和高级筛选非常像呢？

3．常见错误

数据库函数的基本用法实在太简单了，所以想出错也比较困难。唯一有可能出现问题的就是条件辅助单元格里的“表头”。这个“表头”必须和数据库中的“表头”完全一致。数据库中的表头名是“职级”，那么在条件参数里，这个表头也必须是“职级”，如果写成“级别”或者“职别”，都会出错。

4．常用组合

由于DSUM第3个参数必须是引用，这就注定了使用这一函数必须要带辅助单元格。这样一来，好像如图4-68所示的H列的结果无法像SUMIFS或SUM数组一样用一个公式下拉完成。




图4-68

事实上，DSUM也是可以做到的，不过要与SUM函数配合使用，这个公式在H2单元格的写法如下：

=DSUM(A$1:D$9,D$1,F$1:G2)-SUM(H$1:H1)

先看公式的前半部分“DSUM(A$1:D$9,D$1,F$1:G2)”。

这个公式在H2单元格里计算后的结果，是生产部X职级的工资小计，当这个公式下拉到H3单元格，因为绝对引用符号的作用，公式就变成了：

=DSUM(A$1:D$9,D$1,F$1:G3)

而结果除了生产部X职级的工资小计外，还增加了生产部Y职级的工资小计，如图4-69所示。




图4-69

如此一来，H3单元格的结果比我们预想的多出了生产部X职级的部分，需要减去才能得出正确的结果，这时当然可以将原来H2单元格里的DSUM公式拿来用，将公式写成：

=DSUM(B$1:D$9,D$1,F$1:G3)-DSUM(B$1:D$9,D$1,F$1:G2)

不过这样一来，公式下拉是没有问题了，H3:H6这4个单元格都能得出正确的结果，但往回上拉后，H2单元格会得出一个错误值的结果，如图4-70所示，所以需要找另一种方式来代替后面那一串DSUM。




图4-70

再来看公式中的后半部分，在H2单元格里，这部分“SUM(H$1:H1)”看似很多余，因为H1单元格里是文本，而SUM在对带文本的单元格进行求和时会忽略，所以结果是0。但当这个公式下拉到H3单元格时，这一部分会变成

SUM(H$1:H2)

计算的结果就变成了H2的值，而这个值，正是“DSUM(A$1:D$9,D$1,F$1:G3)”计算后多出来的生产部X职级的工资小计部分。

如此一来，我们就可以用这个SUM代替之前会出错的那串DSUM，同时避免了公式在H2单元格出错的问题，如图4-71所示。




图4-71

另外，DSUM第1和第3个参数可以是间接引用，所以也经常和INDIRECT、OFFSET等函数组合，尤其是OFFSET，可以对动态区域进行引用。关于这两个函数的使用，详见第5章。





4.5 设计合理的表格结构避免统计麻烦








4.5.1 避免合并单元格的条件求和




所谓合并单元格，实际上只保留了被合并区域最左上角的一个单元格的内容。所以，带合并单元格的表格结构，上述所有的公式都无法求得正确的结果，如图4-72所示。




图4-72

这里3个公式都没有问题，但是得出的结果都不正确。问题就出在了B列和C列用了合并单元格。

虽然针对合并单元格的条件统计用函数最终也能完成。但是，这需要用到非常复杂的以二分法为查找原则的LOOKUP函数。既然如此，我们为什么不放弃合并单元格，而非要走疑难路线呢？





4.5.2 避免跨工作表/簿的条件求和




跨工作表求和需要用到多维引用，而跨工作簿求和，因为INDIRECT和OFFSET函数不支持跨工作簿计算，所以就要动用VBA代码或者SQL了。所以，在现实工作中，尽可能将数据放在一个工作表里。这样，用简单的函数就可以进行统计。





4.6 附：各类统计公式的特点




各类统计公式的特点如表4-1所示。

表4-1








第5章 用函数查找数据








5.1 单条件查找




日常在数据汇总时，我们经常会遇到根据产品编号查找产品名称、规格，或者通过学号查找姓名等类似的情况。图5-1所示是根据学号查找姓名，使用VLOOKUP函数就可以非常方便地解决这个问题。




图5-1

在单元格输入查找公式：

=VLOOKUP(D2,A1:B5,2,0)

即可查找出D2单元格对应的学号。

VLOOKUP这个函数可谓是最基础的Excel函数之一，也是最常用的一个函数。下面我们一起来看看这个函数是怎样完成这个查找过程的。





5.1.1 VLOOKUP函数单条件查找应用




1．VLOOKUP函数语法

VLOOKUP函数的语法结构如下：

VLOOKUP(查找值,数据区域,返回列数,逻辑值)

查找值：是需要在数据区域的第一列中查找的数值，可以为数值、文本、引用或者是公式计算出的结果。只有在查找值在数据区域的第一列存在时，公式才能返回正确的值。如果查找的数值在第一列中不存在，则会返回错误值“#N/A”，如图5-2所示。




图5-2

同时，在这个错误值的单元格的左上角，会出现一个绿色的小三角。选中这个单元格的时候，左侧就会出现一个小小的提示框，当鼠标移到提示框上面的时候，会出现这个错误值的说明，如图5-3所示。




图5-3

数据区域：是包括查找值和返回值在内的区域，一般表示为“$A$1:$B$10”的形式，以便于公式填充，也可以将区域定义为一个名称。

返回列数：在数据区域的第一列找到查找值后，返回查找值对应的那一行的第几列的单元格的内容。

逻辑值：指明函数返回时是精确匹配还是近似匹配。

当逻辑值为TRUE或者1的时候，Excel的帮助文件中要求的是：“数据区域的第一列必须按升序排列：…，-2，-1，0，1，2，…，-Z，FALSE，TRUE”，实际上是因为这时候公式的计算过程将按照二分法的查找机制进行查找。只有当第一列中数据按升序排列时，才能查找出小于查找值的最大值。

提示：有关Excel中二分法的应用，请查阅函数进阶教材。

看起来逻辑值TRUE的用处似乎不大，实际上在分段数据查找的时候还是很方便的，如所得税、提成、成绩等。请看如图5-4所示的例子。




图5-4

在图5-4中，可以看到评分表的第一列是已经升序排列的。注意看看，张三的50米成绩是8.12秒，在评分表中是没有这个数据的，那么他对应的分数是表中8.11秒的分数，这时候直接用公式“=VLOOKUP(E2,$A$3:$B$18,2,TRUE)”就可以得出了。

当逻辑值为TRUE或者1的时候，这个参数可以忽略不写。以上的公式可以写成：

=VLOOKUP(E2,$A$3:$B$18,2)

当逻辑值为 FALSE或者0的时候，那么数据区域不必进行排序，这时候意味着公式进行精确查找。如果找不到，则返回错误值“#N/A”。

当逻辑值为FALSE的时候，公式经常忽略这几个字母不写，图5-1中的公式可以写成：

=VLOOKUP(D2,A1:B5,2,)

逻辑值使用fALSE进行精确查找，是VLOOKUP的基本功能，也是最常用的功能。对于数据量大的查找，其速度甚至比菜单中的查找功能还快。精确查找适用于文本，也适用于数值；但对数值查找时须注意格式一致，否则会出错。

2．应用实例

图5-5所示的是个税计算（2011年9月1日起）的税率表。




图5-5

税率的计算方法是：

应纳税所得额-起征点=含税级距额度

根据含税级距额度查找对应的适用税率和速算扣除数，应交税额=含税级距额度*税率-速算扣除数。如图5-6所示。

应交税额C11单元格公式为：

=C10*VLOOKUP(C10,B2:C8,2)-VLOOKUP(C10,B2:D8,3)




图5-6

3．常见的错误及解决办法

语法上的错误，如下所述。

以下写法是刚开始学习VLOOKUP这个函数的时候，容易出现的错误写法：

● =VLOOKUP（A1:C10,A1:C10,3,0）

这是将第一个参数写成一个区域了，实际上应该是一个单元格。

严格地讲，并不是说第一个参数不能写成区域，写成区域属于数组运算，要用【Ctrl+Shift+Enter】快捷组合键来确认。

● =VLOOKUP（A12,C1:C10,3,0）

这是将第二个参数误写成了单一列。

在实际应用中，我们会发现，第二参数数据区域的列数，是必须大于或等于第三个参数的。也就是或，上面的公式第三个参数要求返回的是第3列，那么数据区域就必须有3列或以上。

有时候我们会很迷惑，在数据区域明明看到有这个值，为什么返回是错误的？这时候在数据区域用菜单的查找功能也确实是能找到的。

这时首先注意一个情况，用VLOOKUP查找时，查找的数据和区域第一列的数据必须格式是一致的，否则返回的也是错误值。

图5-7显示的是，D2单元格的数值在左边的查找区域中是存在的，但是格式是文本，而数据区域第一列的格式是数值，导致查找返回的是错误值。




图5-7

对于Excel2003及以上的版本，我们通过单元格上的绿三角还比较容易区分格式是否不同，但在Excel2000及以下版本中，只能通过看其属性了。

知道问题就容易解决了。最简单的就是将D2转换成数值，技巧的方法很多，有个最简单的办法就是把公式写成：

=VLOOKUP(--D2,A1:B5,2,0)

这里的两个减号的作用是将D2的文本转换成数值。

在从各种数据库软件导出数据后，再对其使用VLOOKUP进行查找，也会发现出现“#N/A”的情况。这是由于数据库导出的数据，往往在数值后面带有不可见的字符，这时候使用精确查找的话，就会返回“#N/A”。

这时候可以通过以下的方法来检查。

选中第一列的数据，用鼠标点进编辑栏，在数值的前面或后面（通常是在后面），会发现有“空格”，实际上这极有可能是不可见的控制字符。在VLOOKUP计算的时候，会把这些字符也当作需要查找的字符，而公式往往是手动输入的，这就出现了数据区域中没有查找值的情况。

用其他lEN()公式进行检查，空白单元格输入“=LEN(A1)”来检查A1单元格的字符串长度，发现比可见的字符多。

这时候需要对数据的第一列进行处理。方法可以是用lEFT、MID等函数进行提取，或者用分列的方法处理，如果是数值型的数据，也可以用选择性粘贴乘一个1来解决。

在数据处理中，当然有真的查找不到的时候，而这时我们并不想“#N/A”显示出来，

就可以用一个IF判断来处理：

=IF(ISNA(VLOOKUP(D2,A1:B5,2,0)),””,VLOOKUP(D2,A1:B5,2,0))

或者是：

=IF(COUNTIF(A1:A5,D2),VLOOKUP(D2,A1:B5,2,0),"")

当VLOOKUP公式返回的是日期或时间的话，因为Excel单元格格式默认是“常规”，所以实际返回的是时间序列值，如图5-8所示。




图5-8

G2的公式是“=VLOOKUP(E2,A2:C5,3,)”，对应的日期“2011-7-2”被返回成为日期序列号40726。除了调整单元格格式外，还可以将VLOOKUP和TEXT函数相结合：

=TEXT(VLOOKUP(E2,A2:C5,3,),”yyyy-m-d”)





5.1.2 INDEX+MATCH进行查找




除了VLOOKUP外，INDEX和MATCH两个函数搭配也是查找数据的好手，而且在查找功能上比VLOOKUP更加灵活。

1． MATCH语法及使用

MATCH，用来查找数据在区域中的位置，而不是查找值本身。这个函数一共有3个参数：

MATCH(查找值,查找区域,逻辑值)

查找值可以是数值、文本或者逻辑值。

查找区域是一个连续的单元格区域，可以是一行，或者一列。查找区域也可以是一个数组或数组的引用。

逻辑值和VLOOKUP类似，当逻辑值是FALSE或者0的时候，函数进行精确查找；当逻辑值是1或者−1的时候1进行模糊查找。

2．应用实例

如图5-9所示，单元格E2的输入公式：

=MATCH(D2,B2:B8,0)

或者省略0不写：

=MATCH(D2,B2:B8,)

是返回含税级距55000在B2:B8中的位置，位置是第5个数据，即相当于级数。




图5-9

当要查找的数据是B2:B8的含税级距范围内数据时，就可以用MATCH的模糊查找功能，如图5-10所示。




图5-10

单元格E2的输入公式是：

=MATCH(D2,B2:B8,TRUE)

或者省略第三个参数：

=MATCH(D2,B2:B8)





5.1.3 INDEX语法及应用




INDEX 函数返回数据区域中的数值或对数值的引用。

这里所指的数据区域，可以是一行或一列的区域，也可以是多行多列的区域。

INDEX函数有两种形式：数组和引用。数组形式通常返回数值或数值数组；引用形式通常返回引用。

1．语法

数组形式

INDEX(数据区域，行序号或列序号)

返回数组中指定单元格或单元格数组的数值。

这里的数据区域可以是单元格区域或数组常数。而行序号或列序号则指向数据区域中的某一单元格，否则，INDEX返回错误值“#REF! ”。

如图5-11所示，可以通过这两种方式来取值。

根据D2的级数，在E2返回对应的含税级距，那么公式有两种写法：




图5-11

单元格区域写法：

=INDEX(B2:B8,D2)

意即在单元格B2:B8中，返回第4个数据。

数组常数写法：

=INDEX({1500;4500;9000;35000;55000;80000;80001},D2)

意即在{1500;4500;9000;35000;55000;80000;80001}数组常数中，返回第4个数据。

这两种写法并没有优劣之分，只是根据实际情况的不同来采用。

引用形式

INDEX(数据区域的引用，行序号，列序号，区域序号)

返回引用中指定单元格或单元格区域的引用。

这里的数据区域的引用，是指对一个或多个单元格区域的引用。

行序号是引用中某行的行序号，函数从该行返回一个引用。

列序号为引用中某列的列序号，函数从该列返回一个引用。

在数据区域的引用为多个的时候，需要通过区域序号来指定返回哪个数据区域的引用。

简单地说，INDEX引用形式，是指选定多个区域引用中的其中一个，返回行序号和列序号交叉的单元格区域的引用。在区域序号省略的时候，将自动使用第一个区域。

同样，行序号和列序号必须指向其中的单元格，否则，INDEX 返回错误值“#REF! ”。

2．应用实例

应用实例如图5-12所示。




图5-12

当使用公式：

=INDEX(A2:C5,2,3)

返回的是区域A2:C5（图中方框）中的第2行和第3列的交叉的单元格（图5-12中椭圆框所示）的引用6.3。

当使用公式

=INDEX((A2:A5,B2:C5),3,2,2)

返回的是第2个区域B2:C5中第3行和第2列交叉的单元格的引用5.9。

看图5-13中的公式，可以加深对INDEX引用的理解。




图5-13

在图5-13所示的E2单元格中，输入了公式

=INDEX((A2:A5,B2:C5),0,2,2)

这里的意思是对第2区域B2:C5的第2列的引用。在单元格中显示的是数值4.9，这时候在编辑栏中选中整个公式，然后按【F9】键查看，就可以看到图5-14所示的结果。




图5-14

这里可以很明显地看出，结果是4个B2:C5的第2列的引用所形成的数组常数。

作为进一步的理解，在这个公式外面再套用公式，是可以对其做进一步计算的，比如：

=COUNT(INDEX((A2:A5,B2:C5),0,2,2))，计算区域引用的个数，结果是4。

=SUM(INDEX((A2:A5,B2:C5),0,2,2))，计算区域引用的合计数，结果是23.4。

3． INDEX和MATCH配合进行查找

INDEX和MATCH可以说是Excel中最常用的函数组合了，差不多VLOOKUP能查找的，通过INDEX+MATCH都可以做到，称之为最佳拍档并不为过。

图5-1 的VLOOKUP公式，就可以用INDEX+MATCH来代替：

=INDEX(B2:B5,MATCH(E2,A2:A5,))

图5-3中当查找值为文本时，用INDEX+MATCH也可以变通：

=INDEX(B2:B5,MATCH(--E2,A2:A5,))

有兴趣的朋友可以尝试把5.1.1小节中的VLOOKUP公式，都用INDEX+MATCH来替换。

作为比较，两种查找方式在结果上没什么区别。相对来说，VLOOKUP只涉及一个函数。对于入门的学习者来说相对容易理解，适合入门学习函数时第一时间学习，建立对函数运算的概念和思路。INDEX+MATCH在多条件查找和反向查找中更加灵活和多变，并且更适合数据的进一步处理。





5.1.4 LOOKUP函数及应用




LOOKUP 函数可返回一行区域或一列区域、数组中的某个值。这是一个很特别的函数，在分级查找或进行特殊查找时，使用此函数往往有意想不到的结果。

1． LOOKUP语法

LOOKUP函数有两种语法形式：向量形式和数组形式。

当要查询的数值列表较大，或者数据列表会改变时，通常使用向量形式，这种形式的语法有3个参数：

LOOKUP(查找值，查找区域，返回结果的区域)

查找区域是一行或者一列的区域。

返回区域是和查找区域相当大小的一行或者一列的区域。

在LOOKUP函数语法中，有两个特性在公式中得到充分的应用。

如果LOOKUP找不到查找值，就会使用小于查找值的最大值。

查找区域默认按照升序排序。在Excel的帮助中，写的是这个区域必须按照升序排列。这两个说法有些不同，但实际意思是一样的。因为LOOKUP的查找过程是按照二分法的查找机制来进行的，不管区域排序如何，LOOKUP都是作二分法查找，所以很多EXCEL的使用者在没按照升序排列的时候，得出的是错误值。正因为如此，EXCEL的帮助中注明必须按照升序排列，以保证结果的正确性。

图5-11所示的查找，由于查找数据是升序排列，所以可以直接用LOOKUP来完成：

=LOOKUP(D2,A2:B8)

图5-15所示的表格中，若查找区域不是升序的时候，则LOOKUP会返回错误值。




图5-15

公式是：

=LOOKUP(E2,A2:A5,B2:B5)

所以在这个例子中，就必须使用VLOOKUP函数或者INDEX+MATCH函数。

=VLOOKUP(E2,A2:B5,2,)

=INDEX(B2:B5,MATCH(E4,A2:A5,))

需要注意的是，除了数值型数据，文本数据也是有排列顺序的。

当要查询的数值列表较小，或者保持不变的时候，通常使用数组形式，这种形式的语法有两个参数：

LOOKUP(查找值，数组常数或者数据区域)

同样，数组形式的时候，也会有二分法的两个查找特性。

2．应用实例

在图5-所示的表6中，也可以用LOOKUP函数来实现应交税额的计算：

=C10*LOOKUP(C10,B2:C8)-LOOKUP(C10,B2:D8)

除此以外，LOOKUP函数有以下一些常有的方式。

查找最后一次的数据

如图5-16所示，单元格C8计算的是最新的成本价格，可以使用公式：

=LOOKUP(10000,C2:C5)

在这个公式里面，查找值只要是和预计的成本价相比足够大的一个数字即可。




图5-16

用写法上的变通来弥补LOOKUP要求升序的缺陷。图5-15所示的LOOKUP公式导致了查找错误，变通成以下形式，就可以正确查找了：

=LOOKUP(1,0/(A2:A5=E2),B2:B5)

这种LOOKUP(1,0/( ))的形式，完全避免了升序的要求。如果在计算上，有多个符合查找值的记录，则会返回最后一个记录。





5.2 多重条件查找




以上几个函数都可以用于多重条件的查找，所不同的是，公式的繁简程度会有所不同。下面我们通过一个例子来说明几个函数在双重查找上的不同用法。

如图5-17所示是需要通过产品和部门两个条件，来查找对应的销售金额，以组成一个新的汇总表格。




图5-17

VLOOKUP的写法如下：

F2=VLOOKUP($E2&F$1,IF({1,0},$A$2:$A$13&$B$2:$B$13,$C$2:$C$13),2,)

公式说明：

在第1参数中，两个查找条件合并，并且分别在列号和行号上添加了绝对符号，是为了方便下拉填充。

第2参数比较长，“IF({1,0},$A$2:$A$13&$B$2:$B$13,$C$2:$C$13)”，这里是利用了IF函数的特性，把“$A$2:$A$13”和“$B$2:$B$13”组成VLOOKUP查找的第一列，“$C$2:$C$13”则是查找的第二列。正因为如此，这个公式需要用数组方式确定，按（Ctrl+Shift+Enter）组合键。

INDEX+MATCH的写法如下：

F2=INDEX($C$2:$C$13,MATCH($E2&F$1,$A$2:$A$13&$B$2:$B$13,))

公式说明：

此公式也是数组公式。

MATCH中的第1参数和第2参数都分别进行了合并，即合并的查找值，在合并的两列中查找位置。INDEX根据这个位置值来返回对应的数据。

在这个例子中，INDEX+MATCH还有一种写法：

F2=INDEX($C$2:$C$13,MATCH(1,($E2=$A$2:$A$13)*(F$1=$B$2:$B$13),))

同样是数组公式。

在这个公式里面，“($E2=$A$2:$A$13)*(F$1=$B$2:$B$13)”分别把两个查找条件和相对应的列做是否相等的判断，在相等的情况下，各自产生逻辑真TRUE。当两个条件都是符合的时候，两个逻辑真TRUE相乘会得出数值1，其他的都是数值0。例如在F2单元格上用F9计算“($E2=$A$2:$A$13)*(F$1=$B$2:$B$13)”这部分公式，则会看到图5-18所示结果。




图5-18

这里的“MATCH(1,{1;0;0;0;0;0;0;0;0;0;0;0},)”，意味着查找出来的结果是标题以下第1行是符合这个单元格查找条件的。

LOOKUP写法如下：

F2=LOOKUP(1,0/(($A$2:$A$13&$B$2:$B$13)=$E2&F$1),$C$2:$C$13)

公式说明：

这是几个函数写法中唯一的普通公式，不需要数组方式确定。

“0/(($A$2:$A$13&$B$2:$B$13)=$E2&F$1”这部分可以任意添加多个条件判断，格式上非常固定，是“LOOKUP(1,0/( ))”这种查找形式的拓展应用。

经常用函数进行数据查找的读者，建议记下这个格式写法。

以上例子是双重条件查找的公式写法，当涉及多重条件查找的时候，条件单元格和对应的查找区域可以添加到对应的公式里面去，运算的原理完全没有变化。

需要提及的是，当数据量很大，也就是说查找条件的个数比较多，或者查找区域行数很多的时候，几个公式所占用的系统资源和运算速度会有所不同。数组公式历来在这些情况下都不是首选，使用数组公式来查找的用户，几乎每个人都有眼睁睁看着计算进度百分比一动不动，甚至Excel停止响应的经历，即使重启电脑做同样的操作，通常得到的是同样的结果。建议数据量特别大的时候，改用VBA的数组或者字典方式，具体方法不在此赘言。





5.3 反向查找




所谓反向查找，是和日常的从左到右查找相对而言的，VLOOKUP就是最典型的从左到右查找。反向查找就是返回数值是在查找数值的左边，即从右边向左边查找。

Excel里面的所有查找函数，都有各自反向查找的方法，有的本身就具有这个功能。





5.3.1 用VLOOKUP实现反向查找




图5-19所示的表格需要根据E2的姓名，反向查找出对应的学号。




图5-19

用VLOOKUP公式有以下两种写法：

F2=VLOOKUP(E2,IF({1,0},C2:C5,A2:A5),2,0)

F2=VLOOKUP(E2,CHOOSE({1,2},C2:C5,A2:A5),2,0)

公式说明：

两个公式的“IF({1,0},C2:C5,A2:A5)”和“CHOOSE({1,2},C2:C5,A2:A5)”，目的是一样的，利用公式内的常量数组{1,0}和{1,2}，重新构建查找的数据区域。在此可以把这理解为按照公式的编写顺序构建新区域，先提取C2:C5数据建立为新区域的第一列，再提取A2:A5建立新区域的第二列。因此这里VLOOKUP的第三参数都是2而不是其他列序号。

当用于多重条件查找的时候，两种公式的写法都可以达到同样效果。





5.3.2 用INDEX+MATCH进行反向查找




由于INDEX的第1个参数可以灵活设置，因此INDEX+MATCH本身就具有反查的功能，公式也非常简洁：

F2=INDEX(A2:A5,MATCH(E2,C2:C5,0))





5.3.3 用LOOKUP进行反向查找




用LOOKUP查找文本的时候，也存在排序的问题，而且比数值型数据的排序更复杂，因此更宜采用“LOOKUP(1,0/())”的形式：

F2=LOOKUP(1,0/(C2:C5=E2),A2:A5)

需要说明的是，这种公式写法中，第1参数的1，和第2参数的0并非一成不变，只需要保持第1参数不小于第2参数中的分子即可。比如上面的公式也可以写成：

=LOOKUP(0,0/(C2:C5=E2),A2:A5)

或者：

=LOOKUP(9,1/(C2:C5=E2),A2:A5)





5.4 INDIRECT函数




在工作表中引用单元格的时候，通常直接使用等号。而当需要引用的单元格地址或者地址的部分内容在其他单元格中存储的时候，就不能直接用等号了。

如图5-20所示。




图5-20

在单元格A列输入了数值1～4，C1单元格的内容是A3。当需要通过单元格C1来引用A3的值的时候，以下D1输入的公式是错误的：

="="&C1

当然这时候可以通过把D1单元格选择性粘贴成数值后分列得到A3的值，但这个过程显然不是我们需要的。

这时候就需要用INDIRECT函数了。

INDIRECT也叫做引用函数，英文的意思是间接、迂回。这意味着这个引用函数是通过“迂回”的方式来引用单元格内容的。





5.4.1 INDIRECT函数语法




INDIRECT函数语法如下：

INDIRECT(单元格的引用，逻辑值)

单元格的引用包括 A1样式的引用、R1C1样式的引用、定义名称或对文本字符串单元格的引用。如果单元格的引用是错误的或者不存在的，INDIRECT返回错误值“#REF!”。

A1样式和R1C1通过设置来转换，R1C1中的R和C分别代表行和列的意思，R1C1样式表示行标和列标都是用数字方式来标注。设置方式是，执行“工具”→“选项”菜单命令，进入“选项”对话框，选“常规”选项卡，如图5-21所示。




图5-21

把“R1C1”引用样式复选项选中以后，工作表的列标将显示成图5-22所示。




图5-22

Excel默认的是A1样式，日常的数据引用也大多是A1样式。

逻辑值指明单元格引用的类型。

如果逻辑值为 TRUE 或省略，则为A1样式的引用。

如果逻辑值为 FALSE，则为R1C1样式的引用。

另外，如果引用是针对另一个工作簿的，也称之为外部引用，则那个工作簿必须被打开。如果源工作簿没有打开，函数 INDIRECT 返回错误值“#REF!”。





5.4.2 基本应用




1．根据以上语法，我们开始尝试“迂回”引用了。如图5-23所示




图5-23

D2的公式为：

=INDIRECT(C1)

当我们尝试用F9来检查INDIRECT里面C1的时候，会发现是“=INDIRECT("A3")”，这时候就可以理解这里的“间接、迂回”的意思了：C1里面存储了另一个单元格的地址A3，当我们想引用A3的时候，就需要通过C1来“迂回”了。

换一个角度来理解，如果公式直接写成“=INDIRECT("A3")”，则表示始终引用单元格A3。

容易混淆的是，当输入公式“=INDIRECT("C1")”时，结果将是A3。关于这点，可以根据上一句的说法来理解。“=INDIRECT("C1")”表示始终引用单元格C1，而C1的内容是“A3”这两个字符，所以结果是A3。

2．跨工作表的引用

首先看两张工作表如图5-24所示。




图5-24

在工作表Sheet1中是学号等基本信息，Sheet2中需要根据单元格A5指定的工作表名来引用3项内容：

A2=INDIRECT($A$5&"!r3c",)

公式说明：

因为需要始终引用工作表名，并且需要右拉填充，所以在公式中，“INDIRECT($A$5”是固定的。

“&”后面的叹号“! ”表示前面的内容是工作表名。

“r3c”表示引用的是第3行和当前列。

特别需要注意的是，当工作表的表名中有空格时，以上写法依然会导致错误。请见图5-25。




图5-25

如果使用公式“=INDIRECT("sheet 1!A3")”，会显示出错结果，如图5-26所示。




图5-26

解决办法是在公式中的表名外加上单引号，如图5-27所示。




图5-27

公式：=INDIRECT("'sheet 1'!A3")

因此，在工作表名中，要么不使用空格，要么就在引用的时候加上单引号。





5.5 OFFSET函数与动态引用




在英文中，OFFSET是偏移的意思，这个函数是以指定的单元格为基准点，通过指定的偏移量得到新的单元格引用。因此，我们也把这个有趣又实用的函数称为偏移函数。





5.5.1 OFFSET函数语法




OFFSET函数一共有5个参数：

OFFSET(引用基点，偏移行数，偏移列数，新区域行数，新区域列数)

引用基点，可以是一个单元格引用，也可以是单元格区域引用。

偏移行数和偏移列数，正数表示行数向下偏移、列数向右偏移，负数表示向上和向左偏移。

新区域的行数和列数，表示可以返回这个行数和列数单元格区域引用。我们用一个例子来理解这个语法，如图5-28所示。




图5-28

公式为：

=OFFSET(A2,3,2,2,2)

最后的结果是红色细框的4个单元格C5:D6。

公式的运算过程是：以第1参数作为引用基点，根据第2和第3参数，向下偏移3行，并且向右偏移两行，即C5作为新的基准点。以新的基准点为基础，取第4和第5参数分别引用两行和两列单元格区域C5:D6，作为公式最后的结果。

当第4、第5参数省略的时候，表示不再以新的基准点来获取多行多列。这时候当第1参数是一个单元格引用时，则函数返回对新的基准点的引用。比如“=OFFSET(A2,3,2)”，则结果为对C5的引用。

当第1参数是一个单元格区域的引用时，则函数返回以新的基准点为基础的，和第1参数同样矮小的单元格区域的引用。比如“=OFFSET(A2:B3,3,2)”，则结果为以C5为新的基准点的区域C5:D6，新的区域和第1参数同样大小。

函数中的第4和第5参数同样也可以是负数，如图5-29所示中的公式：

=OFFSET(A2,3,2,-2,-2)：




图5-29

此公式的意思是，从新的基准点C5开始，向上向左分别选定两行和两列来引用，结果是B4:C5。





5.5.2 OFFSET函数的扩展应用




OFFSET函数返回的是对单元格的引用，所以可以对其结果做进一步运算，比如求个数、求和等。

公式1：

=COUNT(OFFSET(A2,3,2,2,2))

公式2：

=SUM(OFFSET(A2,3,2,2,2)

以上公式1，是返回的区域引用中的数据个数，公式2是返回区域引用中的数据的合计数。

另外由于OFFSET函数的第4、第5参数可以是对其他单元格的引用，或者是其他公式计算的结果，因此可以对数据统计、查找或者图表制作进行多动态引用。

图5-30就是一个动态引用的例子。




图5-30

公式为：

E3=SUM(OFFSET(B1,COUNTA(B:B)-1,,-3))

可向右拖曳填充。

公式说明：

“COUNTA(B:B)-1”计算B列的数据个数，但包括了标题行，所以需要减1，目的是把新的基准点定位于最下面的单元格。

第4个参数表示向上偏移3行，引用3个单元格数据。

由于这里是动态的引用，所以当每天添加新的出入库数量的时候，以上公式会自动调整引用的单元，合计数会相应变化。





5.6 查找与引用实例








5.6.1 制作工资条




如图5-31所示是工资的明细表。




图5-31

需要将工资明细表制作成图5-32所示的工资条。




图5-32

以下是几个常用的转换公式：

A1=IF(MOD(ROW(),3)=0,"",IF(MOD(ROW(),3)=1,明细表!A$1,INDEX(明细表!$A:$F,(ROW()+1)/3+1,COLUMN())))

A1=IF(MOD(ROW(),3)=0,"",IF(MOD(ROW(),3)=1, 明细表!A$1,INDIRECT(" 明细表!r"&(ROW ()+1)/3+1&"c",)))

A1=IF(MOD(ROW(),3)=0,"",IF(MOD(ROW(),3)=1,明细表!A$1,OFFSET(明细表!$A$1,(ROW() +1)/3,COLUMN()-1)))

公式说明：

这几个公式都是普通公式，在结构上也都是一样的，都是用了两层IF做判断。

“MOD(ROW(),3”取当前行号对3的余数，来判断是返回标题行、各人的工资数据、还是空白行。因此公式可以理解成为：当前行是3、6、9的时候，不做任何引用，每个单元格都返回空白；当前行号是1、4、7的时候，引用明细表的标题行；当前行是2、5、8的时候，通过“(ROW()+1)/3+1”运算，依次引用明细表的2、3、4行。

3个公式的不同在于引用各人的工资数据时，采用了不同的函数。

需要提醒的是，公式2采用了INDIRECT的R1C1样式。INDIRECT采用什么样式，取决于需要引用的行号和列号是否为一个变量，当在公式中行列号是变化的时候，通常采用R1C1样式，如果是固定的时候，采用A1样式。





5.6.2 制作二级下拉菜单




所谓二级下拉菜单，比如当我们单击电脑桌面左下角的“开始”按钮后，执行“所有程序”→“附件”或者“Microsofe Office”命令，再弹出的菜单就叫作二级菜单。

本节通过INDEX、INIRECT和OFFSET3种方法分别来制作二级下拉菜单。

如图5-33所示是需要制作二级下拉菜单的基础数据和二级下拉菜单的效果。当一级菜单选择“福建省”的时候，二级菜单有这个省份所属的6个城市可供选择。




图5-33

操作步骤如下所述。

第一步：建立一级菜单。

选中单元格F2，进入Excel的菜单：执行“数据”→“有效性”菜单命令。进入“数据有效性”的设置界面，如图5-34所示。




图5-34

在“允许”下拉列表框中选择“序列”，在“来源”框中输入“=$A$1:$D$1”，一级菜单就建立好了，如图5-35所示。




图5-35

一级菜单也可以通过定义名称的方法来设置，方法是：

选中“$A$1:$D$1”，在名称框中输入“省份”二字（也可以通过菜单“插入”→“名称”→“定义”来设置命令），单击“确定”按钮，如图5-36所示。然后在以上数据有效性的设置中，在“来源”框中输入“=省份”。




图5-36

第二步：建立二级菜单。

在设置数据有效性前，先设置几个定义名称，以便于后面的操作。

选中A2:A6，在名称框输入“广东省”；同样地，把各列的城市的数据区域均以省份来定义。另外对A2:D7设置名称为“城市”。最终在名称定义列表中可以看到以下定义内容，如图5-36所示。

现在可以开始设置二级下拉菜单了。

使用INDEX函数设置。选中单元格G2，执行“数据”→“有效性”菜单命令，如图5-37所示。

在“来源”框中输入“=INDEX(城市,0,MATCH($F$2,省份,))”，这时候G2单元格的二级下拉菜单就设置好了。




图5-37

这里的设置相对简单，但缺陷就是默认每个省份的城市都有6个，在选择湖南省、四川省等省份的二级下拉菜单时，下拉菜单最后面都有空白行。同时，假如新增加城市超过6个的时候，需要修改“城市”定义名称的引用位置的范围。要避免上述两个缺陷，采取的方法如下。

使用INDIRECT函数设置。在数据有效性设置中，在“来源”处输入“=INDIRECT($F$2)”。

这个设置不会在二级下拉菜单中有空白行，但不利于新数据的添加。比如“广东省”下面一共添加有10个城市，那么需要把“广东省”的定义名称的引用位置修改为“= $A$2:$A$11”。

使用OFFSET函数设置。在数据有效性设置中，在“来源”处输入“=OFFSET($A$2,,MATCH($F$2,省份,)-1,COUNTA(OFFSET(A:A,,MATCH($F$2,省份,)-1))-1)”。

这里的OFFSET函数采用了动态引用，避免了上述的两个缺陷，代价是公式比刚才两个都长，而且相对不太容易理解。

这个公式甚至还省略了第5个参数。前半部分“OFFSET($A$2,,MATCH($F$2,省份,)-1”，是把新的引用区域的基准点定位到相应省份的第一个城市。后半部分目的是为了计算出对应有多少个城市。

建议把这个公式单独放入一个单元格中计算，用F9计算各部分结果，并且对半部分独立地运算以便更容易地查看过程和结果。

假如能完成以上二级下拉菜单，那么恭喜您！查找和引用的学习您已经过关了。





5.6.3 图片的查找与引用




在查找和引用中，我们遇到过需要查找图片、照片的时候，比如需要调用列表中的人员资料，包括照片，如图5-38所示。




图5-38

在档案表中，通过对单元格的设置，把对应的图片都放到了单元格里面。

在调用表的过程中，用VLOOKUP或其他函数引用姓名，却不能引用出图片来。

这是因为VLOOKUP返回的是数值，而不是引用。INDIRECT和OFFSET返回的是引用，但是也必须通过名称定义才能实现。

引用对应的照片，需要以下步骤。

第一步，在G3单元格中插入一幅剪贴图。

第二步，进入名称定义，定义一个名称“照片”，引用位置输入公式：

=INDIRECT("C"&MATCH($E$3,$A$3:$A$5,)+2)

执行结果如图5-39所示。




图5-39

第三步，选定G3单元格的剪贴图，在编辑栏输入“=照片”。这时候照片就引用出来了。

试着修改E3的序号看看，对应的照片是不是跟着变化了？

2007版Excel的图片设置和2003版的稍有不同，定义名称的引用需要做如下修改：

=INDIRECT(ADDRESS(MATCH($E$3, $A$3:$A$5,)+2,3,1,1,”Sheet1”))





第6章 用函数处理文本




众所周知，Excel是进行数据统计和分析的工具，但这并不意味着，Excel不能处理文本。相反，在某些方面，Excel处理文本的功能还十分强大。这主要体现在Excel对文本数据的认读、转换、信息统计、查找替换、拆分与合并上。下面将逐一为您介绍。

通过本章的学习，您将掌握Excel内置函数库中绝大部分文本函数，以及个别与文本处理有关的信息函数。





6.1 什么是文本




在单元格中输入数据，Excel会自动识别录入的数据类型。如果录入的是数字， Excel会自动将其识别为数值，并自动右对齐。如果在单元格中录入如FALSE、TRUE这两个逻辑值，以及#N/A这样的错误值时，Excel也会自动识别，并将其居中对齐。当录入公式时，Excel会自动返回公式结果，并根据返回的结果类型，自动设置对齐方式。除此之外，录入的其他数据都是文本了，Excel会自动将其左对齐。如图6-1所示。




图6-1

有一种特殊情况，作为数字，在Excel中有两种表现类型：一种是默认的数值类型，另一种情况它可以被当作文本使用。要把数字当作文本使用，我们必须事先告诉Excel。如，提前将单元格数字格式设置为文本，或在输入的数字前先加一个英文单引号。

这就是为什么当我们输入18位的身份证号码时，15位以后的数字会变成0。因为如果你没有提前通知Excel，它会自动将18位的身份证号码识别为数值，而Excel最高只能识别15位的数值，超过15位的数值信息将会丢失，以0显示。

如果我们在输入18位身份证号码前，将单元格数字格式设置为文本，或在输入前加入一个英文单引号，这就相当于通知Excel，以下录入的是文本类型。作为文本， Excel不会有15位的限制。

注意，提前录入的那个英文单引号，不会在单元格中显示，但在编辑栏可以看到它。默认情况下，当数字以文本类型存储时，单元格左上角会出现一个绿色的小三角，此单元格被选中时，它的旁边还会出现一个中间有感叹号的智能标记。鼠标移动到智能标记上稍作停留，Excel会出现一行提示“此单元格中的数字为文本格式，或者其前面有撇号”，如图6-2所示。单击此智能标记，会展示一个信息栏，如图6-3所示。点击信息栏的第二项“转换为数字”，可以将文本型的数字转换为数值型。




图6-2




图6-3

综上所述，Excel中的文本包括：中文字符、英文字母，以及被提前设置了文本类型的数字和逻辑值等。注意，这里说的是“提前”设置成文本类型。

我们可以做个试验，在单元格中输入123这样的数字，再将其单元格数字格式设置为文本，虽然单元格中的数字会更改为左对齐，但用ISTEXT函数检查，发现此单元格数据并不会改变为文本类型。同样的道理，提前设置了文本类型后，录入的数字被确认为文本型数字。之后再将其单元格数字格式改为“数值”，也不能真正改变其数字类型。用SUM函数对这样的数字进行求和，得到的是0。





6.1.1 文本、字符、字符串与字节




对于文本类型的数据，我们常会见到这样几个称谓，字符、字节、字符串和文本。

字符，是文本最小的独立单位，如一个汉字是一个字符，一个单独的符号是一个字符，而一个英文字母，无论大小写，也是一个字符。

字节，是用来衡量字符在电脑中所占用的空间大小的。很多人都知道，汉字比英文字母“大”。这个“大”，其实指的就是一个字符的字节数。一个汉字字符占用两个字节，也就是双字节，一个字母占用一个字节，即指单字节。一般情况下，插入的符号或特殊符号，占用的也是两个字节。

字符串，顾名思义，是指两个以上的字符组成的文本。而文本概念常与字符串通用。本章所述的文本是涵盖了字符和字符串的广义概念。

前面我们曾说过，单元格数字格式设置为文本类型时，输入的身份证号码不会有15位的限制。那么单元格中可以存储多长的文本呢？

在Excel中，一个单元格最多可容纳32767个字符，可容纳的公式总长度为1024个字符，包括等号。公式中可容纳的字符串长度为255。





6.1.2 数值型转文本型




我们已经知道，数值型数字通过改变单元格数字格式，无法转换为文本型。操作上有很多解决办法，而使用函数的解决方法，通常是将数值连接一个空文本，如图6-4所示。




图6-4





6.1.3 VALUE文本型转数值型




通过前面的介绍，我们还知道，文本型数字通过改变单元格数字格式，无法转换为数值型。同样，技巧操作上也有很多解决办法。而常用的函数解决方法有如下两个。

1．将文本型数字加减0，或乘除1

加减0或乘除1，完全是为了保证原始数据不会增大或变小。实际上，文本型数字只要一遇到加减乘除符号，就会立即改变为数值型。如果把文本型数字比喻成“秀才”的话，那么加减乘除符号就是“兵”了。“秀才遇见兵，有理说不清”，管你是不是文本，都要改变为数值，参与数值运算。如图6-5所示。




图6-5

2．用VALUE函数将文本型数字转换为数值

VALUE函数的使用十分简单，只需把待转换的文本型数字放入它的“怀抱”（那对括号）即可，如图6-6所示。




图6-6

3． VALUE函数语法、参数

VALUE函数只有一个参数，即VALUE（文本）。文本可以是带双引号的文本，也可以是单元格引用的文本。

如图6-6所示，参数就是一个单元格引用文本，而图6-7则是以带双引号的文本为参数。




图6-7

参数“文本”，可以是Excel能够识别的任意数值，包括“带有格式的数值”，如日期、时间、带货币符号的数字、带百分号的数字、科学计数数字、分数数字等，如图6-8所示。




图6-8

如果VALUE函数的参数，不是Excel所能识别的数值格式，那么函数结果将返回错误值 “#VALUE!”。

如图6-9所示，参数为“一二三四”，在Excel中这不是标准的数值格式，而是一个文本，因此用它作为VALUE函数的参数，将返回错误值。




图6-9

4． VALUE函数应用实例

我们可能会遇到这样一种情况，要对一组数据进行统计计算，但这组数据都有单位。为了能够进行计算，我们运用函数将数据中的单位删除。可是得到的数字仍然不能直接用函数统计。原因是，得到的数字是文本型数字。

如图6-10所示，C2单元格输入公式“=SUBSTITUTE（B2，"元/斤"，""）”，将公式向下复制到C10单元格。意思是将B2:B10单元格中的“元/斤”替换为空文本，目的是删除“元/斤”，只保留数字。




图6-10

在B11单元格输入公式“=AVERAGE（B2:B10）”，向右复制到C11单元格，目的是分别对B列和C列的数据求平均值。尽管C列数据已将“单位”删除，可C11单元格得到的结果仍是错误的，因为删除“单位”后的数字仍是文本类型。

我们再加一列辅助列，在D2单元格输入公式“=VALUE（C2）”，向下复制公式到C10单元格，在C11单元格输入公式“=AVERAGE（D2:D10）”，才能得到正确结果。

5．扩展应用

VALUE函数似乎很有用，平常我们却很少使用它。因为VALUE函数不如前面所述的“秀才遇见兵”，来得容易便捷。如图6-10所示D2单元格输入公式“=C2*1”，一样可以解决问题。

6．常见错误

无论是用“秀才遇见兵”，还是用VALUE函数，将文本型数字转换为数值型，常见错误都是：试图转换的内容不是正确的数值格式。

比如，我们可能试图将“2011.2.14”这样一个文本转换为数值。如图6-11所示，A1单元格公式为“="2011.2.14"+0”，得到了错误结果。如果我们将“2011.2.14”改正为“2011/2/14”，或是“2011-2-14”，将会得到正确结果。




图6-11





6.1.4 ISTEXT判定是否文本




可能有人会问，有时候文本和数值长的那么像，简直难以区分，我们该如何运用公式，判定其是不是文本呢？

方法也不止一个，其中ISTEXT就是一个可以判定内容是否为文本的函数。如果内容是文本，函数结果返加TRUE，如果不是文本，函数结果就返回FALSE，十分直观，如图6-12所示。




图6-12

ISTEXT是个信息函数，在函数库的“信息函数”中才能找到它。





6.1.5 TYPE判定数据类型




TYPE函数也可以判断参数是否是文本，如果是文本，函数结果将返回2。如果是数值，将返回1。如果是逻辑值，则返回4。如果参数是错误值，会返回16。如果参数是数组，就会返回64。

因此TYPE函数可以同时区分数值、文本、逻辑值、错误值和数组。





6.1.6 T测定文本并按原文返回




T虽然只有一个字符，但它也是一个很重要的函数。它的用法是T（参数），如果参数是文本，函数结果将文本原样返回。如果参数不是文本，如数值、逻辑值，函数结果将返回一个空文本，如图6-13所示。




图6-13

说它重要，但在平常的公式中，却很少用到它。在函数高级教程中，讲述三维数组公式时，我们将会重新认识它。





6.1.7 不区分大小写文本比较




所谓文本比较，也就是看看两个文本是否一致。一般情况下，文本比较是不需要区分大小写的。如图6-14所示，在C1单元格输入公式“=A1=B1”，得到的是TRUE，说明大写A与小写a是一致的。

可是有时候，我们明明看着两个文本是“相同”的，直接比较的结果却出人意料。如所图6-15所示，C1单元格中的公式依然是“=A1=B1”，得到的结果却是FALSE。




图6-14




图6-15

这是因为，A1单元中，我们输入的是“小 妖”，中间有一个空格。而在B1单元格中，输入的却是“小妖”，中间没空格，只是将单元格的“水平对齐方式”设置为“分散对齐”，使A1与B1单元格的内容看上去是一致的。

看到这里，可能有人会笑。谁会那么傻，硬要将两个本来不一样的文本，搞得看似一样？谁又没事儿干，将两个看似一致，其实根本不一致的文本进行比较呢？

其实这个问题我们在工作中常会遇到。如图6-16所示的A列，我们可能会将两个字符的名字中间加一个空格，以保证其能与3个字符的名字对齐。




图6-16

这样虽然好看，但当需要按姓名查找其他内容时，就怎么也找不到。如图6-16所示中F3单元格公式返回的结果就是错误值。那是因为，E列内容是将单元格的“水平对齐方式”设置成“分散对齐”。所以平时要养成良好的习惯，不做画蛇添足的事情。





6.1.8 Excel区分大小写比较




在大多数情况下，比较文本时不需要区分大小写。可是也会有个别情况，需要区分大小写进行比较。比如，要求EXCEL与Excel不做同一个文本处理，这时就需要用到EXACT函数。如图6-17所示，C1单元格的公式为“=A1=B1”，而D1单元格的公式为“=EXACT（A1，B1）”，两个公式得到不同结果。




图6-17





6.2 文本转换




文本转换是文本函数的重要功能，这包括字符与字符代码之间的转换、将数字转换为某一特定格式、字符大小写的转换等。尤其是字符大小写的转换，如人民币小写转大写，工作中经常会遇到。





6.2.1 字符代码




我们在屏幕上能看到的所有字符都有一个代码编号。每一个字符就好象一件物品，在字符库房里有唯一的摆放位置。这个代码编号就好比字符的位置编号。

比如，我们首先将输入法调成“中文（中国）”，然后按住【Alt】键的同时，在小键盘输入“41385”，你会发现单元格中输入了一个字符“々”。“41385”就是字符“々”的代码。字符“々”在查找引用函数中经常被作为最大字符使用。迄今为止，我仍不知道此字符的读音，也不知道用什么输入法如何输入这一字符。每当我要用到它时，除了使用以上方法外，就是依次点击“插入”→“特殊符号”→“标点符号”命令选项，查找并插入这一符号，如图6-18所示。




图6-18

在Excel中，可以使用CODE函数取得一个字符的代码，也可以使用CHAR函数将一个字符代码转换为字符。





6.2.2 CODE字符转换为数字




在工作中，我们可能会需要知道某一字符的代码，以便输入或计算方便。如，我们想知道字符“々”的代码，以便日后直接使用“ALT+代码”方式录入此字符。再如，我们想知道列标“K”是第几列，以便下一步的计算。这时CODE函数就派上了用场，如图6-19所示。




图6-19

1．CODE函数语法、参数

CODE函数的参数只有一个，即CODE（文本）。此参数既可以是引用的文本，也可以是直接的文本字符。如图6-19所示就是以引用的文本作为参数，而图6-20则是以直接的文本字符作为参数。




图6-20

如果CODE函数的参数是一个字符串，则函数只返回字符串首字符的代码。如图6-21所示，B1与B2单元格得到的字符代码为65，因为它们都返回了字符“A”的代码，而B3与B4单元格得到的字符代码为54992，这是字符“中”的代码。




图6-21

2．应用实例

例1：统计有多少不重复的姓，如图6-22所示的公式。




图6-22





6.2.3 CHAR数字转换为字符




CHAR函数的作用与CODE函数正好相反，它的作用是将字符的数字代码转换为字符。

如图6-23所示，A列为姓名，B列为职务，工作中常会遇到产生如C列格式的内容。我们的做法是在C列输入公式“=A2&CHAR（10）&B2”，然后选择C列，设置单元格格式，在“对齐”选项卡中，将“文本控制”方式设置为“自动换行”，如图6-24所示。公式中“CHAR（10）”的作用是生成一个“换行符”，数字“10”是“换行符”的字符代码。




图6-23




图6-24





6.2.4 DOLLAR数字转换为美元货币格式




在EXCEL中，通过“设置单元格格式”，可以将数字设置成多种显示格式。如，可以将数字设置为美元货币格式或人民币货币格式。但是，如果数字是由公式产生，而数字显示格式还要通过“设置单元格格式”操作。就显得过于繁琐。文本函数中提供了这样的函数，DOLLAR函数就是将数字转换为美元货币格式的函数，如图6-25所示。




图6-25





6.2.5 RMB数字转换为人民币货币格式




与DOLLAR函数用法相似，RMB函数是将数字转换为人民币货币格式的函数，如图6-26所示。




图6-26





6.2.6 FIXED数字转换为千位分隔符文本




FIXED函数用来将数字直接转换为千位分隔符文本，如图6-27所示。




图6-27





6.2.7 TEXT数字转换为格式文本




TEXT函数是文本函数中使用率最高的函数之一。它的作用类似于单元格自定义格式。比如说，我们在Excel中输入“2011-2-14”这样的日期，默认状态下，Excel就以这样的格式保存。即使我们输入的日期是“2011年2月14日”，Excel仍默以“2011-2-14”格式保存。如果我们想让它以“年月日”格式保存，就必须依次用鼠标右键单击单元格，选择“设置单元格格式”选项，在弹出的对话框中选择“数字”标签，在“分类”栏选择“日期”选项，在右侧的“类型”栏中选择一个“年月日”格式，如图6-28所示。




图6-28

“类型”栏都是一些内置的，或者曾经使用过的格式。如果想查看格式代码，可以在“类型”栏中选择一种类型后，再选择“分类”栏中的“自定义”选项，在右侧的“类型”栏中会出现刚才所选类型的格式代码，如图6-29所示，“yyyy"年"m"月"d"日";@”就是“年月日”格式的格式代码。




图6-29

利用TEXT函数可以达到同样的效果，但必须了解这些格式代码，对于初学者，知道如何查找符合要求的格式代码十分重要。

1．用函数返回特定日期格式文本

如图6-30所示，刚才的问题用TEXT函数解决，B1单元格的公式为“=TEXT（A1，"yyyy年m月d日"）”。




图6-30

2．函数语法、参数

TEXT函数有两个参数，语法结构是TEXT（数值,格式代码）。

数值，可以是一切值。只要在单元格中能被设置自定义格式，就可以作为TEXT的第一个参数。

格式代码，如果是数字，可以不带双引号。如果是文本，需要用双引号括起来。代码内容与自定义格式基本一致。但有几点不同，如下所述。

（1）格式代码内部的引号需另加一对引号括起来。如图6-30所示中的公式，原自定义格式代码为“yyyy"年"m"月"d"日";@”，那么公式应该写成“=TEXT（A1，"yyyy"年"m"月"d"日";@"）”，但是当输入这样一个公式时，会弹跳出一个如图6-31所示的警告提示。




图6-31

这是因为在公式中，格式代码最外一层双引号是代表文本，而代码内部的双引号必须在其外侧再套一层双引号，如图6-32所示，公式才不会提示错误。




图6-32

此段格式代码可以简化，简化后的公式就是“=TEXT（A1，"yyyy年m月d日"）”。

（2）星号*在自定义格式中用来重复某个字符，但在TEXT函数中不支持。

（3）自定义格式中可以设置数据显示颜色，但TEXT函数做不到这一点。

（4） TEXT函数格式代码可以使用公式，而自定义格式做不到。

3．应用实例

例1：比如，工作中我常要录入一些日期，可我又不愿意按Excel的标准格式录入，因为每输入一个标准日期都要输入两个“-”，很是麻烦。常把日期录入为八位数值，如20110114之类。这样的“日期”，在日后需要按日期显示或计算时，可以采用TEXT函数将其转换。如图6-33所示，在C2单元格输入公式“=TEXT（B2，"0-00-00"）”，将B2单元格的八位数值转换为日期格式的文本。而在D2单元格输入公式“=DATEDIF（C2，TODAY()，"y"）”，计算出具体年龄。




图6-33

例2：再举一个例子，工作中，我们免不了要进行金额统计。可是合计后的金额，我们总希望能显示大写人民币的样子。如图6-34所示，D6单元格公式为“=TEXT （C6，"[DBNum2]G/通用格式元"）”，此公式也可以写成“=TEXT（C2，"[DBNum2]"）&"元"”。

当金额是小数时，这个公式会更为复杂，但基本原理是不变的。




图6-34

4．扩展应用

正如自定义格式用途十分广泛一样，TEXT函数也有非常奇妙的用法。

例1：看一个四舍六入五保留的例子。如图6-35所示，TEXT函数的第二个参数是由公式生成的变动的“格式代码”，其中公式“IF（RIGHT（B2）*1=5，0，""）”表示，如果工资的最后一位等于“5”，则返回“0”，否则返回“空文本”。再与前面的字符“0.0”相连接，就生成了格式代码“0.00”或“0.0”。也就是说，如果工资的最后一位等于“5”，格式代码是“0.00”，如果不等于“5”，格式代码就是“0.0”。格式代码“0.0”保证了工资最后一位“四舍六入”，而代码“0.00”，则可以保证工资最后一位是“5”时，保留原值。




图6-35

例2：比如我们想知道自己的农历生日，以前有过很复杂的公式，而TEXT函数却有一条秘密通道。如图6-36所示，在2010版中，在B2单元格中输入公式“=TEXT（A2，"[$-110000]e-m-d"）”，可以返回农历日期。公式中代码参数区间为110000～139999、910000～939999或1910000～1939999，均可返回农历日期，但必须是在2007以上版本中。




图6-36

5．常见错误

（1）格式代码容易出错，并不是所有可用的自定义格式代码，都可以直接照搬到函数中使用。如图6-37所示，当格式代码中出现双引号，每一个引号都必须成对出现。公式中“格式代码”部分的意思是，如果名次小于等于“3”，就显示为“"竞赛活动"奖励”，否则显示为“空”。




图6-37

（2） TEXT函数的结果是文本，如果要进行计算，必须转换为数值，这与自定义格式不同。

6．常组合使用的函数

常组合使用的函数包括统计函数、日期函数，以及其他文本函数。





6.2.8 NUMBERSTRING数字转换为大写文本




如图6-34所示，可以使用TEXT函数将数字转换为各种格式的大写文本。但是同时，Excel也给出了另一个将数字转换为大写文本的函数，这就是NUMBERSTRING函数，只是这个函数是一个“隐藏”函数，在“插入函数”对话框中并不能找到它。

1．函数语法、参数

NUMBERSTRING函数有两个参数，即NUMBERSTRING（待转换的数字,代码）。它的代码只有3种，分别为1、2、3，产生3种不同的大写格式，如图6-38所示。




图6-38

2．应用实例

我们常会遇到需要对大写数字如一、二、三、四……十一、十二等进行排序，但Excel默认会按拼音排序，不能达到我们的目的。这时，可以输入1～12这样的数字进行排序，然后利用自定义格式将其显示为大写，或利用TEXT函数、NUMBERSTRING函数将其转换为大写。如图6-39所示，B2单元格公式为“=TEXT （A2，"[DBNUM1][$-411]G/通用格式"）”，代码中加入了“[$-411]”，可以使“一十”以上的文本去掉前面的“一”，这符合我们的习惯。但当数值达到百位时，需要对公式做些变动，因为我们不习惯将“一百一十一”，称之为“百十一”。




图6-39





6.2.9 DATESTRING数字转换为日期文本




与NUMBERSTRING函数相似，DATESTRING函数也是一个“隐藏”函数，它只有一个参数。如公式“=DATESTRING（"2011-5-25"）”返回字符“11年05月25日”。





6.2.10 LOWER文本转换为小写




Excel不仅可以对数字进行大小写和格式上的转换，也可以对文本进行非常灵活的转换。如LOWER函数就是将文本转换为小写的函数。这个函数只有一个参数，如图6-40中B列所示，LOWER函数可以将英文字母全部转换为小写。B2单元格公式为“=LOWER（A2）”。




图6-40





6.2.11 UPPER文本转换为大写




用法同上，UPPER函数可以将英文字母全部转换为大写，如图6-40中C列所示， C2单元格公式为“=UPPER（A2）”。





6.2.12 PROPER文本首字大写




用法同上，PROPER函数可以将英文文本的首字母转换为大写，而其他字母转换为小写。如图6-40中D列所示，D2单元格公式为“=PROPER（A2）”。





6.2.13 WIDECHAR单字节转换双字节




在6.1.1小节，我们介绍了字节的概念，这里不再赘述。WIDECHAR函数只有一个参数，它可以将英文“单字节字符”转换为“双字节字符”。如图6-41中B列所示， B2单元格公式为“=WIDECHAR（A2）”。




图6-41





6.2.14 ASC双字节转换单字节




ASC函数与WIDECHAR函数的作用正好相反，它是将英文“双字节字符”转换为“单字节字符”。如图6-41中C列所示，C2单元格公式为“=ASC（B2）”。

值得注意的是，无论是大小写转换还是单双字节转换，函数只对英文字符起作用，而中文字符却无动于衷，这一特性可以用来区别中文和英文文本。





6.3 文本字符信息统计








6.3.1 LEN统计字符数




LEN函数是文本函数最为常用的函数之一，用它来统计字符串的字符数十分方便。

1．函数语法、参数

LEN函数参数只有一个，即LEN（字符串）。它可以返回字符串长度，即字符数。字符串中的每一个中文字符、英文字符、标点符号等都算一个字符，如图6-42所示。




图6-42

2．应用实例

例：由身份证号码提取出生日期，首先要判断身份证号码的位数，15位的身份证号码，第7、8位为出生年份，第9、10位为出生月份，第11、12位为出生日期。18位身份证号码，第7～10位为出生年份，第11、12位为出生月份，第13、14位为出生日期。在图6-43中，B2单元格的公式为“=TEXT（IF（LEN（A2）=18，MID（A2，9，6），MID （A2，7，6）），"00-00-00"）*1”，公式中先由IF函数判断A2单元格身份证号码是否为18位，如果是18位，则从字符串的第9位开始截取6位数字，如果不是18位，则从字符串的第7位开始截取6位数字，然后将截取出来的6位数字用TEXT函数格式化成“日期格式”的文本，最后乘以1是将文本转换成数值。如果想得到如图6-43所示的标准日期格式，还需要将B列单元格格式设置成“日期格式”。




图6-43





6.3.2 LENB统计字节数




与LEN函数用法相似，但LENB函数统计的是字符串字节数。如图6-44所示A列为单双字节混和文本，B列公式为“=LEFT（A1，LENB（A1）-LEN（A1））”，式中先用“LENB（A1）-LEN（A1）”取得字符串前面双字节部分的字符数，然后用LEFT函数从文本左边开始，按双字节字符个数提取字符。C列公式为“=RIGHT（A1，2*LEN（A1）-LENB（A1））”，RIGHT函数是从文本右边开始提取字符。




图6-44





6.3.3 返回中文拼音




Word中有“拼音指南”功能，可为文本添加拼音。在Excel中，也有“拼音指南”，但却不能自动添加拼音。如果需要在Excel的文本上添加拼音，可以在Word中为文本添加好拼音后复制到Excel，也可以在Excel中利用“拼音指南”手动编辑拼音。添加了拼音的文本，可用PHONETIC函数单独返回拼音文本，如图6-45所示。这个函数并不在“文本函数”中，它躲藏在“信息函数”中。函数的帮助信息中介绍此函数为“提取文本字符串中的拼音字符”，并且“该函数只适用于日文版”，看来并非如此。实际上我们发现，这个函数还有一个不为人知的功能，后面会再一次介绍它。




图6-45





6.4 文本查找




在文本中定位和查找某一指定字符串，是工作中经常要遇到的问题。Excel的查找功能与Word略有不同，它是按单元格查找。但是可以利用函数精确定位所要查找的字符串在文本中的位置。





6.4.1 FIND区分大小写查找




FIND函数可以区分大小写查找指定字符的位置。如图6-46所示，B1单元格公式为“=FIND（"a"，A1）”，它返回小写字母“a”在字符串中的位置“5”；而C1单元格的公式为“=FIND（"A"，A1）”，它返回大写字母“A”在字符串中的位置“1”。




图6-46

1．函数语法、参数

实际上FIND函数有3个参数，语法是FIND（被查字符串,备查文本,在文本的第几位开始）。当第三个参数省略时，默认从文本首位开始查找。当函数查不到所要查找的字符串时，返回错误值“#VALUE!”。

如图6-47所示，B2单元格公式为“=FIND（B1，A2，2）”，第一个参数“被查字符串”是B1单元格中的字符“A”，第二个参数“备查文本”是A2单元格的文本“A级苹果APPLE”，第三个参数是在文本的第“2”位开始查找。因此掠过了文本中的第一个“A”，而返回了第二个“A”的位置“5”。




图6-47

当第三个参数省略时，默认从文本首位开始查找。如图6-47所示，C2单元格公式为“=FIND（C1，A2）”，省略了第三个参数，因此返回了文本中第一个“A”的位置“1”。

函数的第一个参数既可以是单字符，也可以是指定的字符串。如图6-47所示D2单元格公式为“=FIND（D1，A2）”，实际上是查找字符串“苹果”的位置。

2．应用实例

工作中常会根据文本中某一特定字符的位置提取部分字符串。如图6-48所示， B1单元格的公式为“=MID（A1，FIND（"X"，A1）+1，FIND（" "，A1）-FIND（"X"，A1）-1）”，此公式是以字符“X”和“空格”为标志位置截取字符串的。




图6-48

与LEN函数一样， FIND函数也有一个兄弟FINDB函数，但较少使用。





6.4.2 SEARCH忽略大小写查找




SEARCH函数语法与FIND函数相似，但它在查找时忽略大小写。

如图6-46所示，如果将公式中的FIND函数改为SEARCH函数，那么无论公式“=SEARCH（"a"，A1）”，还是“=SEARCH（"A"，A1）”都返回1。





6.4.3 SEARCH使用通配符查找




在Excel中，可以使用通配符的函数并不多，SEARCH就是其中一个。它可以使用通配符“？”和“*”，前者代表任意单个字符，后者代表任意多个字符。如图6-49所示，B1单元格公式为“=SEARCH（"[?????????]"，A1）”，意思是查找“包含着9个字符的一对中括号”所处的位置。




图6-49

与LEN、FIND函数一样，SEARCH也有一个兄弟SEARCHB函数，它是按字节数计算所要查找内容的位置。当充分了解这些函数的功能，灵活搭配使用，往往能产生意想不到的效果。

如图6-50所示，B1单元格公式为“=LEFT（A1，SEARCHB（"?"，A1）/2）”，公式中“SEARCHB（"?"，A1）”是查找任意单字节字符在文本中的字节位数。A1单元格中的第一个单字节就是“A”，而A2单元格第一个单字节就是“0”，A3单元格第一个单字节是“C”。由于文本中第一个单字节前面都是双字节的中文字符，因此“SEARCHB（"?"， A1）”得到的结果一定是一个奇数，除以2的整数部分一定是双字节中文字符的个数。尽管除以2后有小数部分，但LEFT函数的第二个参数是小数时，会自动取整计算。




图6-50





6.5 文本替换




文本替换是指，对文本中特定字符或特定位置的字符进行替换。





6.5.1 SUBSTITUTE替换具体字符




SUBSTITUTE函数可以将具体字符替换为另一指定字符。如图6-51所示，A列文本都有一个“元”字，如果C1单元格公式为“=A1*B1”，肯定无法计算，而C1单元格公式写成“=SUBSTITUTE（A1，"元"，""）*B1”，先将A1单元格的字符“元”，替换为“空字符”，也就是将字符“50元”换成了字符“50”，再与B1单元格的数值相乘，自然可以得到正确结果。注意，字符“50”可以直接参与计算，是因为它遇见了“乘号”，也就是前面曾经说过的“秀才遇见兵”。




图6-51

1．函数语法、参数

SUBSTITUTE函数实际上有4个参数，即SUBSTITUTE（原文本,待替换文本,替换文本,替换第几个待替换文本），当第4个参数省略时，默认将原文本中所有待替换文本全部用新文本替换。如图6-52所示，B1单元格公式为“=SUBSTITUTE（A1，0，1， 2）”，意思是将A1单元格文本中第“2”个“0”替换为“1”，如果把第4个参数“2”省略，结果将变成“2111年一季度”。




图6-52

2．应用实例

在图6-53中的A列是错误的日期格式，如果要参与计算，必须将其改为正确的日期格式。方法是将A列文本中的“.”替换为“-”。B1单元格的公式为“=YEAR （TODAY()）-YEAR（SUBSTITUTE（A1，"."，"-"））”。




图6-53

3．扩展应用

例1：计算某一字符在文本中出现的次数。如图6-54所示，B1单元格公式为“=LEN（A1）-LEN（SUBSTITUTE（A1，" "，""））”，其中“SUBSTITUTE（A1，" "，""）”是将文本中的“空格”，替换为“空”，相当于把文本中的“空格”全部删除。




图6-54

例2：有时，我们需要在某一具体字符的前后加入几个字符。如图6-55所示，现需要在A列文本中的字符“元”后面加上字符“/斤，”，B1单元格公式为“=SUBSTITUTE（A1，"元"，"元/斤，"）”。




图6-55





6.5.2 REPLACE替换具体位置上的文本




有时，我们并不确定需要被替换的具体文本，但我们知道它的位置，这时就可以使用REPLACE函数。

1．函数语法、参数

REPLACE函数也有4个参数，即REPLACE（原文本，待替换文本的起始位置，待替换文本的字符个数，替换文本）。如图6-52所示，想得到同样的结果，B1单元格的公式可以写成“=REPLACE（A1，3，2，11）”，意思是将A1单元格文本的第“3”个字符开始的“2”个字符用字符“11”替换。

2．应用实例

如图6-56所示的A列是以文本格式保存的数据，数据中的分隔符较为混乱，但位置固定，这时可以使用REPLACE函数进行替换，B1单元格公式为“=REPLACE （A1，5，1，"-"）”。




图6-56

3．扩展应用

例1：删除固定位置的字符。如图6-57所示，需要将A列数据中的分隔符删除， B1单元格公式为“=REPLACE（REPLACE（A1，5，1，""），7，1，""）”。




图6-57

例2：在固定位置插入字符。如图6-57所示的B列，需要转换为C列结果，我们曾用TEXT函数解决，参考图6-33。现使用REPLACE函数的公式为“=REPLACE （REPLACE（B1，5，，"-"），8，，"-"）”，注意函数的第三个参数留空，相当于第三个参数是“0”。

REPLACE有一个小兄弟REPLACEB函数，这是以字节数进行计算的替换函数。当数据是单双字节混排文本时，使用REPLACEB函数可能更为有用。如图6-58所示，B1单元格公式为“=REPLACEB（A1，SEARCHB（"?"，A1），，"-"）”。




图6-58





6.5.3 TRIM删除文本中多余空格




TRIM函数的作用是，将文本中连续多个空格只保留一个，删除其余空格，如图6-59所示。




图6-59





6.5.4 CLEAN删除文本中的非打印字符




若你的数据是从其他数据库导入的，往往会带有很多莫名其妙的字符，这些字符可以用CLEAN函数一扫而光。





6.6 文本拆分




从文本中提取部分文本是工作中常会遇到的问题。与大多数文本函数一样，其结果仍然是一个文本，如果提取出来的部分是“数字”，需要进一步参与计算，要注意“数字”转换为“数值”的问题。





6.6.1 LEFT从左侧开始截取部分文本




LEFT函数是从原文本的左侧第一个字符开始，按位数截取部分文本的函数。如图6-60所示，B1单元格公式为“=LEFT（A1，5）”。




图6-60

函数语法、参数

LEFT函数有两个参数，即LEFT（原文本，需要截取的字符个数）。当第二个参数省略时，默认提取原文本第一个字符。当第二个参数大于原文本字符数时，自动提取全部文本。当第二个参数是小数时，自动截取整数进行计算，参考图6-50。

LEFT函数有一个兄弟LEFTB函数，它的用法与LEFT函数基本相同，但它不是按字符个数提取，而是按字节个数提取左侧文本的函数。





6.6.2 RIGHT从右侧开始截取部分文本




RIGHT函数是从原文本的右侧第一个字符开始，按位数截取部分文本的函数。如图6-60所示，C1单元格公式为“=RIGHT（A1，3）”。

RIGHT函数的语法、参数与LEFT函数相同。

RIGHT函数有一个兄弟RIGHTB函数，用法与LEFTB函数类似。





6.6.3 MID从任意位置开始截取部分文本




MID函数是从原文本的任意位置开始，按位数截取部分文本的函数。如图6-60所示，D1单元格公式为“=MID（A1，6，1）”。

1．函数语法、参数

MID函数的语法、参数与LEFT函数类似。在图6-60中，B1单元格公式可以写成“=MID（A1，1，5）”。

MID函数有一个兄弟MIDB函数，用法与LEFTB函数类似。

2．应用实例

如图6-61所示，需要将A列数据进行拆分。B2单元格公式为“=MID（TEXT （$A2，"00000"），COLUMN（A1），1）”，将B2单元格公式向下向右复制。




图6-61

3．扩展应用

如图6-62所示，B1单元格公式为“=MIDB（A1，SEARCHB（"?"，A1），99）”。公式中的第三个参数“99”是一个明显超过字符数的数字。在此例中第三个参数使用“99”，与使用“6”没什么区别，目的都是超过文本字符数，但“99”更为明显，更为保险。




图6-62





6.7 文本合并








6.7.1 &合并运算符




&是文本合并运算符，用法十分简单，只要在文本之间加上这个运算符，即可将文本连接成一个字符串，在图6-35中已经有合并运算符的使用。

1．应用实例

如图6-63所示，我们常需要做一张按月份变动的表头，每月只要在表格里填入数据即可。A1单元格公式为“="精英公司"&TEXT（TODAY()，"[DBNum1][$-804]m月份"）&"办公费用清单"”。




图6-63

2．扩展应用

例1：连接一特定字符，规避错误值。如图6-64所示，如果B1单元格还用公式“=MIDB（A1，SEARCHB（"?"，A1），99）”，那么在B3单元格就会出现错误值，因为A1单元格根本不存在单字节字符，所以公式中的“SEARCHB（"?"，A1）”部分会得到一个错误值。为规避这一错误值，可以把这部分改为“SEARCHB（"?"，A1&0）”，即B1单元格公式写成“=MIDB（A1，SEARCHB（"?"，A1&0），99）”。




图6-64

例2：连接空文本，将数值转换为文本。如图6-65所示的原始数据，现要在原始数据中提取一部分数据，如图6-66所示的H3单元格公式为“=VLOOKUP（G3&""，&nbsp;A3:E5，5，）”。式中用G3连接一个空文本，目的是让G3单元格的数值转换为文本，因为原始数据的月份一列是文本型数字，否则将查找不到需要的数据。




图6-65




图6-66

例3：连接空文本，规避空单元格引用出现“0”值。还是在图6-65的原始数据中提取电话费数据，如图6-67所示的H3单元格公式为“=VLOOKUP（G3&""， A3:E5，2，）&""”。公式中最后连接一个空文本是为了保证引用原数据的空单元格时，这里不会出现“0”，保持单元格为“空”，而如果引用的原数据本身是“0”值时，这里的结果仍为“0”。




图6-67





6.7.2 CONCATENATE合并文本




CONCATENATE函数是合并文本函数，语法是CONCATENATE（文本1,文本2,文本3,……,文本30），它可以有30个参数，也可以只有一个参数，意思就是将括号里面的文本全部连接在一起。如果参数是单元格引用，则只能是一个一个的单元格引用，不支持区域引用。

如图6-67所示，若要将G2:J2单元格内的文本连接在一起，使用CONCATENATE函数的公式为“=CONCATENATE（G2，H2，I2，J2）”。因此，这一函数在实际工作中较少使用，要想得到相同结果，使用合并运算符更方便快捷。





6.7.3 PHONETIC合并区域文本




CONCATENATE函数不能合并单元格区域的文本，这是文本合并函数的一大软肋，工作中常会觉得不方便，遗憾的是直至2010版，微软公司仍未提供这一功能。

一个偶然的机会，我们发现PHONETIC函数却可以合并单元格区域的文本。如图6-67所示，若要将G2:J2单元格内的文本连接在一起，可以使用PHONETIC函数，公式为“=PHONETIC（G2:J2）”。

但PHONETIC函数在合并区域文本时表现得十分有性格，它不仅会忽略区域内的数值、空单元格和公式，而且它的参数只能使用单元格引用。

如图6-68所示，要想得到每个月的累积故障设备编号，在C1单元格输入公式“=PHONETIC（$B$2:B2）”即可，下一步可根据这一公式的结果，继续计算某一具体设备的累积故障率。




图6-68





6.7.4 REPT文本重复任意次数




1．函数语法、参数

REPT函数有两个参数，即REPT（需要重复的文本,重复次数）。当第二个参数为“0”时，函数结果返回“空文本”。当第二个参数是小数时，此函数与LEFT函数一样会取整计算。

如图6-69所示，B1单元格公式为“=REPT（A1，3）”，C1单元格公式为“=REPT （A1，）”，D1单元格公式为“=REPT（A1，3.6）”。




图6-69

2．应用实例

例1：小数点对齐。如图6-70所示A列数据，要达到小数点对齐的目的，在B1单元格中可以输入公式“=A1&REPT（" "，3-LEN（A1）+FIND（"."，A1））”。公式在原数据后面按“3”位补足空格，以确保数据右对齐时小数点对齐。




图6-70

例2 ：利用REPT函数生成重复格式代码。如图6-71所示， B1单元格公式为“=TEXT（A1，0&REPT（"0-0"，6）&0）”。




图6-71

文本函数在Excel的所有函数中占有重要的位置。工作中经常会用到文本函数，尤其是LEFT、RIGHT、MID、FIND、SEARCH、TEXT等函数，使用十分频繁。文本函数常与统计、数学函数以及日期函数结合使用。文本函数的用法大多比较简单，重点要弄清楚函数返回值的类型和区分单双字节。





第7章 用函数计算日期和时间




在日期与时间的相关运算中，比较常见的运算方式有间隔运算（例如计算员工工龄）、日期与时间推算（例如计算本月最后一天的日期）、改变日期与时间格式（例如日期转化为星期）等。相对于其他函数而言，由于日期与时间类函数在绝大多数情况下仅用于处理日期与时间相关的问题，导致这类函数在使用中适用范围相对较小，也没有太多复杂的变化。

在运算中，日期时间类函数总是遵循Excel中日期时间格式的基本原则，本章将从认识日期时间格式开始，逐一介绍工作中常用的日期时间类函数及常见问题的处理方法。





7.1 认识日期与时间








7.1.1 日期与时间格式




日期与时间作为Excel中的一种常见数字格式，其本质上仅是数值的另一种表现形式。在Excel中，数字1表示1天，由于在微软操作系统中安装Excel默认采用1900日期系统，日期将从1900年开始累计，数字1等同于1900年1月1日，2等同于1900年1月2日。数字的小数部分，会被当作时间来处理，由于整数1表示1天，1/24则为一小时，例如将数值3.5设置为日期与时间的显示方式，将显示为“1900-1-312:00”。

例如在单元格内输入数字1后，将该单元格数字格式修改为日期格式，显示内容变为“1900-1-1”；在单元格中输入日期“2011年5月27日”后，将该单元格数字格式修改为常规格式，显示内容变为40690，表示该日期是自1900年1月1日开始的第40690天。

这种修改也可以作为我们验证某个单元格内的日期字符串是否为真正日期格式的手段。真正的日期格式是数值的另一个表现形式，它可以通过将单元格数字格式修改为常规格式的方法转变为数字；而非真正的日期格式是指文本格式的日期或超出Excel日期格式允许范围的非法日期。

通过一个简单的小例子，就可以看到Excel对真日期与假日期不同的对待方式，如图7-1所示，其中B2:B3区域单元格被设置为文本格式，而B4:B5单元格被设置为日期格式。输入如图所示的日期后，使用LEN函数计算单元格内的文本长度。由于B2单元格为文本格式的日期， Excel只将其作为普通的字符串对待，并没有认为它是一个日期，所以求得B2单元格内存在8个字符，B3单元格同理；而B4单元格内容是一个真正的日期，虽然看起来“1900-1-1”这个字符串应该由8个字符构成，但由于“1900-1-1”这个日期其本质上是数值1，只有1位数字，所以求得其文本长度为1。而B5、B6单元格虽然被设置为日期格式，但由于B5单元格输入的日期已经超出了Excel允许的范围， B6单元格输入的日期在现实中并不存在，所以Excel将它们作为文本处理。




图7-1

注意：由于在微软操作系统中安装Excel，默认采用1900日期系统，Excel只承认1900年1月1日～9999年12月31日之间的日期（包含这两个日期）为合法日期，超出这个范围将被Excel当作文本处理。相对于1900日期系统，还有一种较为常见的日期系统即1904日期系统，在1904日期系统中，是将1904年1月1日作为日期的开始。





7.1.2 日期与时间格式的常见转化




1．日期与时间格式间转化

日期格式

日期与时间仅是数值的不同表现形式，而由于日期与时间格式各自的特性，有时也会显现出它们所具备的与数值不同的特殊属性，例如日期格式的年、月、日、周等属性，时间格式的小时、分、秒等属性。在工作中，我们也会经常遇到利用日期与时间提取特殊信息的问题，可以通过使用单元格自定义格式功能得到很好的解决。

如图7-2所示，B2:B17区域内日期均为“2011-6-1”，将它们的单元格数字格式设置为自定义数字格式，并在自定义格式中使用C2:C17区域中的对应自定义数字格式代码后，日期呈现出了截然不同的表现样式。

字母“y”在自定义数字格式中表示“年”，例如B2单元格与B3单元格比较，前者使用“yy”表示两位数年份，后者使用“yyyy”表示4位数年份。




图7-2

字母“m”在自定义数字格式中表示“月”，从B2:B6区域内日期显示的对比中可以看出，单独的一个字母“m”表示当月份为1位数字（1～9）时显示为1位数字的月份，当月份为2位数字（10～12）时显示为2位数字的月份；“mm”表示无论月份为1位数字还是2位数字，都显示为2位数字的月份，数字位数不足时，将在左侧增加0补足至2位数字；“mmm”表示月份的简写，本例中日期月份为6月（June），其简写为“Jun”；“mmmm”表示完整的月份，即June；“mmmmm”表示月份的首字母，即“J”。

字母“d”在自定义格式中表示“日”，单独的一个“d”表示当“日”为1位数字时显示为1位数字日期，当“日”为2位数字时显示为2位数字日期；“dd”表示无论日期为1位数字还是2位数字，都显示为2位数字日期，数字位数不足时，将在左侧增加0补足至2位数字； “ddd”表示英文星期的简写，本例中日期为星期三（Wednesday），其简写为“Wed”； “dddd”表示完整的月份，即Wednesday。

字母“a”仅用来表示中文的星期，“aaa”与“aaaa”分别表示简写与完整显示，在本例数据中，如果使用“aaa”，星期将显示为“三”；如果使用“aaaa”，星期将显示为“星期三”。

B16、B17单元格分别使用了中文大、小写描述日期，其中“[DBNUM1]”表示使用中文小写表示单元格内的数字，如一、二、三等；“[DBNUM2]”表示使用中文大写表示单元格内的数字，如壹、贰、叁等。

这类通过改变单元格数字格式影响日期表现值的做法，其理论依据来自于Excel本身对日期格式的定义，但通过修改单元格数字格式的方法，也仅能改变其表现值，并没有真正得到其显示的内容。上例B2:B15区域中看似不同的每个值背后的真实值均为40695，表示2011年6月1日是自1900年1月1日开始累计的第40695天。这一点可以通过将B2:B15区域选择性粘贴为“值”的方法得到证明。

如果想将这些类似于“化妆”的表面变化应用到真实变化中，可以使用一个与自定义数字格式密不可分的函数TEXT处理。TEXT函数具备将数值“整容”为某种表现值的特殊能力。

TEXT函数需要两个参数，第一参数为待处理数据，并可以将自定义数字格式代码稍作处理，作为其第二参数，而这个自定义数字格式代码就是我们希望得到的表现值。这类似于通过化妆（设置单元格数字格式）后的样子决定整容（TEXT函数处理）方案。

如图7-3所示，B2:B5区域是通过C2:C5区域内对应公式计算后得到的结果。以A2单元格为例，如果将A2单元格设置为自定义数字格式，并使用“yyyy”作为其自定义数字格式代码，A2单元格将显示为2011，而使用TEXT函数与这个自定义格式代码相结合，能真正得到数字2011。如果使用选择性粘贴中粘贴为“值”的方式将B2单元格复制到其他位置，其值依然为2011。




图7-3

时间格式

与日期格式类似，时间格式的值也能通过设置单元格数字格式的方式得到自身的一些属性信息。如图7-4所示，B2:B6区域中内容均为时间“11:02:20”，将这些单元格的自定义数字格式修改为C2:C6区域中所示的自定义数字格式代码后，单元格显示出不同的内容。

其中字母“h”、“m”、“s”分别表示小时、分钟、秒，1个字母表示当对应信息为1位数字时显示1位，对应信息为两位数字时显示2位；2个相同字母表示无论对应信息是1位数字还是2位数字，都显示2位数字，位数不足2位时，用数字0在左侧补齐。“上午/下午”及“AM/PM”均用来显示当前时间所处的时段。

时间格式的值也可以使用TEXT函数根据自定义格式代码提取特定信息，方法与在日期格式值中使用TEXT函数提取信息类似。




图7-4

2．数字或文本字符串转化为日期与时间格式

在工作中，我们也经常会见到一些看似日期又非日期的“假日期”，例如从公司系统导出的报表中存在大量文本格式的日期，这种情况也会出现在表示时间的单元格中。这类文本格式的“假日期”、“假时间”会导致某些运算出错、数据透视表中无法组合等问题；有时也会因为数据记录不规范，导致数据很难处理，例如“2011年6月1日”被书写为“110601”。这类数据不会被Excel作为日期或时间来对待，所以必须通过一些特殊的方式转化为真正的“日期”与“时间”。

处理文本格式的日期与时间

如果需要将文本格式的日期与时间转化为真正的日期与时间，可以使用本书第2章文本格式转化为数值格式中介绍的方法进行处理，即通过四则运算将文本格式转化为数值类格式，例如乘以1。

如图7-5所示，A2:A5区域内为文本格式的日期与时间，在B2:B5区域内使用乘以1的方式对其进行运算。其中，A2、A4和A5单元格在乘以1后，分别转化为数值，40695代表2011年6月1日为自1900年1月1日期累计的第40695天，而0.517592593代表当时间为12:25:20时，这一天已经过去了0.517592593天。B5单元格的值同理。将这3个数值的数字格式分别更改为日期格式和时间格式后，得到真正的日期与时间。




图7-5

由于A3单元格内的日期是一个非法日期，乘以1后得到错误值“#VALUE！”，如果想将这个日期转化为“2011年7月13日”，可以使用本章后续内容中的DATE函数。而A5单元格内的时间虽然也是一个非法时间，但相较于日期，时间格式拥有固定的进位规则，致使其乘以1后得到的数值是65分钟进位为小时后的结果，即13:05:30的值。

在上个例子中，核心问题是将文本格式的日期与时间，通过乘以1的方式转化为数值，再将单元格数字格式修改为日期或时间格式。Excel中提供了两个分别能将文本格式的日期与时间转化为数值的函数，DATEVALUE和TIMEVALUE，使用这两个函数处理文本格式的日期与时间，能达到上例中乘以1的处理方式完全相同的效果，如图7-6所示。

函数语法：

DATEVALUE（文本格式的日期）

TIMEVALUE（文本格式的时间）




图7-6

处理不符合日期与时间格式的字符串

相对于将文本格式的日期与时间转化为真正的日期与时间而言，将不符合日期与时间格式的字符串转化为日期和时间，更加困难一些。转化文本格式的日期与时间时，虽然其数字格式不同，但数据的记录方式依然符合基本的日期与时间格式记录规范，而不符合日期与时间格式的字符串在处理时，还需要将它们的格式处理为符合Excel日期与时间记录规范的格式后，再转化为数值，例如图7-7中所示的处理方式。




图7-7

在本例中，A2、A3单元格内分别为数值和文本格式的字符串，如果将A2单元格直接修改为日期格式，将无法显示，因为将A2单元格更改为日期格式，将会表示自1900年1月1日开始累计的第20110601天，这已经超出了Excel中日期所允许存在的范围（Excel中的最大合法日期为9999年12月31日，即2958465）。

显然，Excel并不了解本例中我们对于日期格式规范的定义——前4位为年份、第5、6位为月份、第7、8位为日期——所以只能先将字符串转化为Excel认同的日期格式，再进行处理。B2单元格内公式为“=TEXT（A2，"#-00-00"）*1”，其中“=TEXT（A2，"#-00-00"）”表示将A2单元格内的数字格式化，在十位、百位数字之间与千位、万位数字之间增加短横线“-”，这样处理将得到字符串“2011-06-01”。这个结果看似日期，但依然为文本，所以再次使用乘以1的方式将其转化为数值。公式“=TEXT （A2，"#-00-00"）*1”运算的最终结果为40695，将其单元格格式设置为日期后得到真正的日期。

这种先规范格式，再运用四则运算将其转为数值的方式在时间的处理中依然适用。例如图7-8中所示的时间处理问题。A2:A5区域中内容为数值和文本格式的字符串，使用公式“=TEXT（A2，"#\:00\:00"）*1”可在B2:B5区域内得到它们描述时间的数值型态，再将B2:B5区域单元格格式修改为时间格式，得到正确的时间。




图7-8

公式中“=TEXT（A2，"#\:00\:00"）”是将字符串规范为时间格式，这里使用的自定义格式代码，与日期问题处理中的代码稍有不同。代码“#\:00\:00”表示在字符串的第2位、第3位数字之间与第4位、第5位数字之间增加冒号“:”，而冒号是小时与分钟、分钟与秒的分隔符，时间在表示为数值时，又一定是一个小数，所以正常情况下，冒号会显示在整数部分的最后，因此在本例中使用“\”将冒号进行强制显示，而不是作为运算符出现在字符串中。





7.1.3 日期与时间的简单运算




由于日期与时间格式具有独特的进制规律及本质为数值的特点，使我们能通过一些简单的加减及格式设置解决一些运算问题。

日期运算

如图7-9所示，A2:A4区域与B2:B4区域中分别记录了产品的入库及出库时间，需要在C2:C4区域计算对应的滞留天数，即出库时间与入库时间的间隔天数。




图7-9

在C2单元格输入公式“=B2-A2”后下拉填充，此时计算的结果为日期格式的值，所以5被表示为“1900-1-5”，只需要将C2:C4区域单元格数字格式更改为常规，即可得到正确显示的滞留天数。

除了使用减法计算日期间隔之外，也可以使用加法进行日期推算，例如图7-10中所示的根据采购日期与保质期计算过期时间的问题。在D2单元格内输入公式“=B2+C2”后下拉填充。




图7-10

在日期的运算中，直接加、减的运算方法比较适合处理与天数相关的问题，当需要计算两个日期的间隔月份或年份时，这种方法得到的值就显得不够精确，原因在于日期格式不固定的进位机制。例如计算两个日期的间隔月份，使用两个日期相减的方式将得到两个日期的间隔天数，但是将间隔天数换算成月份并不是一个很好处理的问题，因为在天数进位为月份的时候，有些月份包含31天，而有些月份包含30天、29天、28天，如果直接使用间隔天数除以30，会导致计算不够精确。特别是在一些工龄的计算中，会存在“少一天则少一年”的情况，所以必须精确计算入职时间与离职时间的间隔年份。在本章后续内容中将会介绍使用函数DATEDIF来处理这样的间隔运算问题。

时间运算

相比日期格式，时间格式拥有固定的进位规则，即1天包含24小时、1小时包含60分钟、1分钟包含60秒。正是固定的进位规则，使时间格式值的运算比日期运算易于处理。

例如图7-11中所示的计算，B1单元格与D1单元格分别记录了某系统的开机时间和关机时间，需要通过这两个时间分别以小时、分钟、秒为单位计算系统运行时间。




图7-11

从原理上看，如果将B1、D1单元格均设置为数值格式，13:30:28秒将显示为0.562824，表示13:30:28是0.562824天；17:28:30将显示为0.728125，表示17:28:30是0.728125天。如果将这两个时间直接相减将得到03:58:02，即时间间隔为3小时58分2秒。而03:58:02转换成数值格式后，将显示为0.165301，即3小时58分2秒为0.165301天，如果将这个数值乘以24，将得到小时，再乘以60将得到分钟，再次乘以60即可得到秒。

计算步骤非常简单，例如在以小时为单位计算系统运行时间中，在C5单元格输入公式“=（D1-B1）*24”后，将C5单元格格式改为数值格式即可。





7.2 使用函数获取已有日期与时间中的信息








7.2.1 YEAR、MONTH、DAY函数




YEAR函数、MONTH函数、DAY函数分别用来提取日期中的年、月、日信息，它们的用法也非常相似。

1．函数语法

YEAR（待计算值）

MONTH（待计算值）

DAY（待计算值）

参数“待计算值”应该为日期值或可被转化为日期值的其他类型值或公式。

2．应用实例

以YEAR函数为例，图7-12中列举了对不同数据使用YEAR函数求其年份的运算。其中A2单元格内为常见日期格式，通过YEAR函数提取年份得到其年份值2011；A3单元格内为文本格式的日期，但依然可以被YEAR函数运算得到正确结果；由于“2011年6月1日”转化为数值后将显示为40695，所以40695等同于“2011年6月1日”，如果对40695取年份可以顺利得到2011，而无需考虑40695的格式是文本还是数值，这正是B4、B5单元格能正确计算的原因；A6单元格的日期格式依然能被系统识别为日期，所以还可以被正确计算，但A7单元格内的日期使用“.”作为分隔符，已经不再被操作系统当作日期来处理，所以无法得到年份信息，返回“#VALUE!”错误；由于日期存在范围限制，最大的日期为“9999年12月31日”，这个日期在数值格式下将显示为2958465，即自1900年1月1日开始累积的第2958465天，所以A8单元格内数值2958465依然可以正确运算得到年份9999：而A9、A10单元格内的数字由于超出了日期允许的范围，无法得到年份值。返回错误“#NUM!”




图7-12

可见，YEAR函数可以从日期或可被转化为日期的其他类型值中提取年份信息，这一点与MONTH函数及DAY函数相同。

MONTH函数与DAY函数在使用中与YEAR函数极其类似，对参数的要求也完全相同，如图7-13所示，3个函数在分别提取A2单元格内日期的年份、月份、日期时，除函数名称不同以外，使用方法没有任何不同。

由于季度与月份相关，所以可以使用MONTH函数计算某个日期所在的季度，如图7-14所示公式。公式“INT（（MONTH（A2）-1）/3）”表示将待处理日期的月份减1，除以3后使用INT函数取其整数部分。对于第一季度而言，包含1月、2月、3月，这3个月份值减去1后，得到0、1、2，除以3后的整数部分均为0；而对于第二季度而言，包含4月、5月、6月，这3个月份值减去1后，得到3、4、5，除以3后的整数部分均为1，；第三季度和第四季度的月份值减掉1后除以3，其整数部分将分别为2、3。通过这样的处理，已经可以把日期所在的季度描述出来，一、二、三、四这四个季度的日期将分别被表示为0、1、2、3。




图7-13




图7-14

以“2011-8-30”为例，其月份值为8，减1后得到7，再除以3得到2.33333，将这个数字使用INT函数取整数部分得到2，表示第三季度。

再将0、1、2、3加1，使其与季度号匹配，最后使用TEXT函数把得到的数字转化为“第N季度”的格式。其中“[DBNUM1]第0季度”表示将某个数字转换成中文小写格式，并在这个数字前增加文字“第”、数字后增加文字“季度”。

3．扩展应用

在很多企业中，经常会使用英文月份名称，例如图7-15中A2:A7区域中，使用了英文简写的月份。这样书写的月份在后续处理中并不是很好处理，例如按季度汇总数据、计算当月天数等。比较直接的解决方法是将英文月份转化为数值后再进行处理，本例中仅介绍英文月份转化为数值的方法。

如果在单元格中直接输入英文月份（例如“JAN”），Excel仅会将这个字符串作为文本来对待，与月份丝毫没有关系。而在单元格中输入“JAN1”时，单元格内将会显示“1-JAN”，如果更改该单元格日期格式，使其完整显示年、月、日，可以看到字符串“JAN1”被自动添加了当前日期的年份，最终得到当前年份的1月1日。可见，“JAN1”是一个可以被Excel识别为日期的字符串，并被认为表示1月1日。

根据上述原理，可在B2单元格输入公式“=MONTH（A2&1）”后下拉填充。其中“A2&1”是在英文月份后添加数字1，使其符合日期格式规范，再使用MONTH函数提取其月份。

7.2.2 HOUR、MINUTE、SECOND函数




与获取日期中的信息类似，获取时间信息是指从表示时间的数值中提取小时、分钟、秒这3个信息，分别使用HOUR、MINUTE、SECOND函数。

1．函数语法

HOUR（待计算值）

MINUTE（待计算值）

SECOND（待计算值）

其中参数“待计算值”应该为时间值或可被转化为时间值的其他类型值或公式。

如图7-16所示，从时间中分别提取小时、分钟、秒。

时间函数也不能处理不符合时间格式记录规范的数据，如果将14:22:30书写为142230，就无法使用上述3个函数提取其信息。如果使用HOUR函数提取这个字符串中的信息，将得到结果0。“142230”作为整数只能表示整天，并不含有小数部分，而时间在转化为数值后必然是小数，作为整数的“142230”并不含有时间信息，所以无法直接使用HOUR函数计算。





7.3 DATE、TIME函数




上一节内容分别介绍了使用函数从已有日期与时间中提取特定信息，例如年、月、日、小时、分、秒等，这种做法类似于将已有的日期和时间拆分。本节将学习使用函数得到日期或时间值，即使用年、月、日或小时、分、秒组合成日期或时间。由于时间具备固定的进位机制，比较容易计算，本节将重点介绍通过年、月、日组合日期值的函数计算方法。





7.3.1 DATE函数组合日期




在处理日期相关的计算时，我们经常会遇到这样的问题：未来的某一天是否存在（例如：2011年2月29日）？3个月后的日期是什么？这个月一共有多少天？

这些问题都需要我们对一个未知的日期进行判断，最直接的方法是将这个日期组合出来再进行分析，而组合出日期所需要的3个基本元素分别是“年”、“月”、“日”， DATE函数就能够将3者结合在一起，生成一个日期格式的值，供我们使用。

在下面的章节中，我们将使用DATE函数，并综合前面学到的内容解决以下问题。

计算设备的下次点检日期。

判断某年是否为闰年。

计算某日期所在的月份共有多少天。

计算日期所在季度的最后一天。

1．函数语法

DATE（年，月，日）

年

“年”参数可以为一到四位数字，同时Microsoft Excel将根据不同的日期系统来解释“年”参数。在Windows系统中安装Microsoft Excel，将默认使用1900日期系统。在1900日期系统中，将1900年作为日期的开始，如果“年”是0到1899（包含）之间的数字，Microsoft Excel会将该值加上1900以后作为日期的“年”。如图7-17所示，DATE（111，8，30）其结果为“2011年8月30日”，其年份为1900与111的和；如果“年”是1900到9999（包含）之间的数值，则Microsoft Excel将直接使用这个数值作为“年”。DATE（2011，8，30）其结果同样为“2011年8月30日”；如果“年”小于0或者大于等于10000，则Microsoft Excel将返回错误值“#NUM！”。由于这个原因，导致我们无法使用DATE函数得到1900年到9999年以外的日期。

月

“月”参数是一个代表月份的数字，但该数字不仅可以大于12，甚至可以为0或负数，当该数字为大于12的数字，将按照12进制的方式累加到年份上。如图7-18所示，DATE（2010，20，30）将返回“2011年8月30日”；当该数字为负数时，则同样按照12进制的方式减少年份，并将减少的年份换算成月份，与该负数相加，最终得到一个1到12之间的月份。DATE（2012，-4，30）将返回“2011年8月30日”；如果“月”是一个1到12之间数字，则直接将该数字作为月份生成日期。

日

“日”参数是一个代表在月份中第几天的数字。当“日”大于该月的最大天数时，则会从下个月的第一天开始累加，并同时累加月份。如图7-19所示， DATE（2011，7， 61）将返回“2011年8月30日”；当“日”小于0时，则会从上个月的最后一天减掉对应的天数，并同时减少月份。DATE（2011，9，-1）将返回“2011年8月30日“；当“日”为0时，则会得到上个月最后一天的日期。DATE（2011，9，0）将返回2011年8月31日。




图7-19

2．应用实例

计算设备下次点检日期

如图7-20所示，B列和C列分别记录了设备的上一次点检日期及点检周期，通过这两列数据，在D列中计算对应设备下次点检的日期。

公式：

D2 =DATE(YEAR(B2)+MONTH(B2)+C2+DAY(B2))

其中“YEAR（B2）”用于提取B2单元格内日期中的“年”作为下次点检日期的“年”、“MONTH（B2）+C2”用于提取B2单元格内日期中的“月”，并将此月份与点检周期的月份相加，作为下次点检日期中的月份、“DAY（B2）”用于提取B2单元格内日期中的“日”，作为下次点检日期的日期。最后使用DATE函数将上述3次计算“年”、“月”、“日”组合为一个日期，如果月份相加后的结果大于12，根据DATE函数中“月”参数的累加特性，最终依然可以得到正确日期。

判断年份是否为闰年

如下图7-21所示，A列列出了一些年份，在B列计算该年份是平年还是闰年。

公式：

B2 =IF(DAY(DATE(A2,2,29))=29," 闰年 "," 平年 ")

其中“DATE（A2，2，29）”是将A列的年份作为年份，“2月29日”作为月份和日期，如果该年份为闰年，则会存在“2月29日”，否则，该年份将不存在“2月29日”这个日期，根据函数语法中“日”参数的累加特性，将得到该年份的3月1日。

公式“DAY（DATE（A2，2，29））=29”表示使用DAY函数取出DATE函数所生成日期中的“日”，如果为闰年，则会得到29日；否则由于日期将被转换为“3月1日”，从而得到1日。再用这个值与29进行比较，最终闰年将返回结果TRUE，否则将为FALSE。

计算某日期所在的月份共有多少天

如图7-22所示，A列列出了一些日期，在B列计算该日期所在月份共有多少天。

公式：

B2=DAY(DATE(YEAR(A2),MONTH(A2)+1,0))

其中公式“YEAR（A2）”表示提取A2单元格内日期中的“年”作为新日期的“年”，“MONTH（A2）+1”表示提取A2单元格内日期中的“月”，并加1作为新日期的“月”，数字0作为新日期的日。

再利用DATE函数组成公式“DATE（YEAR（A2），MONTH（A2）+1，0）”，表示使用原日期中的年份、原日期中月份加1的和、0分别作为新日期的“年”、“月”、“日”生成日期。使用这种方法进行处理，本例中“2011年8月30日”将变成“2011年9月0日”。在这个新日期中，由于“日”为零，根据“日”参数的特性，将会得到上个月最后一天的日期，即2011年8月31日（当月的最后一天），也就是希望得到的新日期。

最后使用DAY函数计算这个新日期的“日”，由于新日期的“日”是本月的最后一天，所以结果也就是本月共包含的天数。

计算某日期所在季度最后一天的日期

公式：

C2=DATE(YEARCA2),INT((MONTH(A2)-1)/3)*3+4,0

如图7-23所示数据，通过A2:A6区域内日期，计算日期所在季度最后一天的日期。公式中“INT（（MONTH（A2）-1）/3）”是将日期的月份值减1后除以3并取整数部分，对于1、2、3月份而言，运算将得到0；对于4、5、6月份而言，运算将得到1；对于7、8、9月份而言，运算将得到2；对于10、11、12月份而言，运算将得到3。即4个季度的月份被分别表示为0、1、2、3。

公式“INT（（MONTH（A2）-1）/3）*3+4”表示将上述数字乘以3后加4，经过这一次处理，1、2、3月份将得到4；4、5、6月份将得到7；7、8、9月份将得到10；10、11、12月份将得到13。即4个季度的月份被分别表示为4、7、10、13，也就是当前季度的下季度首月。

再使用DATE函数生成日期，将待处理日期的年份作为新日期的年份，当前日期所在季度的下个季度首月作为月份，0作为日期。根据DATE函数“日”参数的特性，新日期将被向前推算一天，得到当前日期所在季度的下个季度首月第一天的前一天，即当前日期所在季度的最后一天。

3． DATE函数常见错误

由于DATE函数运算的结果一定是日期，而日期在Excel中有其允许存在的范围，所以当DATE函数运算得到的结果超出这个范围时，将显示函数错误。例如公式“=DATE（10000，2，3）”与“=DATE（-1，2，3）”，就由于这个原因返回错误值“#NUM!”。

4．常组合的函数

由于DATE是一个生成日期的函数，只需要我们提供“年”、“月”、“日”3个参数，就能够组合出一个日期格式的数字，所以常被我们用来生成一个不确定的日期，或将不符合日期格式的“假日期”转换成日期格式，因此常与以下函数组合使用。

与YEAR、MONTH、DAY函数结合生成新日期

例1：计算自今天开始100天以后的日期。

公式：DATE（YEAR（TODAY()），MONTH（TODAY()），DAY（TODAY()）+100）

例2：计算自今天开始3个月以后的日期。

公式：DATE（YEAR（TODAY()），MONTH（TODAY()）+3，DAY（TODAY()））

例2：计算自今天开始5年以后的日期。

公式：DATE（YEAR（TODAY()）+5，MONTH（TODAY()），DAY（TODAY()））

如图7-24所示。

与文本函数结合将“假日期”转换为日期格式的数字

例4：提取18位身份证中的生日信息

公式：DATE（MID（A2，7，4），MID（A2，11，2），MID（A2，13，2））

如图7-25所示。

7.3.2 TIME函数组合时间




相较于通过某个条件获得日期值，时间值的生成则简单很多。因为时间具备固定的进位机制，在简单运算中，TIME函数显得可有可无，所以本节只对该函数做简单介绍，用于同DATE函数类比。

1．函数语法

TIME（小时、分钟、秒）

小时

为0（零）到32767之间的数值，任何大于23的数值将除以24，其余数将视为小时。如果使用小数，将会对该小数取整（提取整数部分，并非四舍五入）作为参数。

分钟

为0（零）到32767之间的数值，任何大于59的数值将除以60，其商的整数部分将视为小时，并与参数“小时”相加，其余数将视为分钟。如果使用小数，将会对该小数取整（提取整数部分，并非四舍五入）作为参数。

秒

为0（零）到32767之间的数值，任何大于59的数值将除以60，其商的整数部分将视为分钟，并与参数“分钟”相加，其余数将视为秒。如果使用小数，将会对该小数取整（提取整数部分，并非四舍五入）作为参数。

2．应用实例

如图7-26所示，通过A2:A5区域中的系统运行时间和B2:B5区域中的系统持续时间，在C2:C5区域计算系统结束运行的时间。由于是关于生成某个时间的问题，可使用TIME函数完成，在C2单元格输入公式“=TIME（HOUR（A2），MINUTE（A2）+B2，SECOND（A2））”后下拉填充。

以上公式的写法非常类似于示例图7-20中通过开始日期和间隔月份计算结束日期的运算方式，除函数表示的意义不同外，计算思路完全一致。本例中提取出开始时间的小时数作为新生成时间的小时，提取开始时间的分钟值并与持续时间的分钟数相加，作为新生成时间的分钟，提取出开始时间的秒数作为新生成时间的秒。

由于TIME函数参数的累计进位特性，A5单元格内时间15:45:00在与50分钟相加后，并没有得到非法时间15:95:00，而是将95除以60，其结果的整数部分进位至小时，余数部分作为分钟显示出来，得到正确结果16:35:00。

如果将时间视为一个具有特殊数字格式、规则进位方式描述的数值，则可以通过简单的加、减运算处理时间值的问题，但在运算前，需要先统一数值单位。例如上例中计算结束时间的公式可使用“=A2+B2/24/60”代替。其中A2为开始时间，作为时间，其数值形态为小于1的小数。B2中的持续时间为整数，而整数在日期与时间中表示天数，需要将这个整数转化为分钟数，再与A2单元格内的时间相加。1天（即整数1）为24小时，1小时为60分钟，所以将整数转化为分钟数，可使用除以24再除以60的方式，当然也可以直接除以24与60的乘积1440。

可见使用加、减的方式计算时间，比使用TIME函数创建公式计算更加直接，而最为关键的问题是，需要统一数值单位。由于时间值具备规则的进位机制，统一数值单位也并不困难，而这一点正是日期值所欠缺的（每年的天数不确定、每月的天数不确定），所以相较DATE函数，TIME函数更容易被加、减法替换。如图7-27所示，计算开始时间与结束时间之间的间隔时间，由于需要以分钟为单位，所以将结束时间与起始时间相减后乘以24，再乘以60。

使用加、减法计算与使用TIME函数计算看似得到的结果一致，但也具备一些细微的区别，例如小时数大于23时的处理。TIME函数不会将23小时以上的时间进位到天数，只会直接消去。而使用加、减计算的方式，由于是将时间作为数值来处理，所以时间一旦大于或等于24小时，则会向天数（整数位）进位。如图7-28所示，通过起始时间与间隔时间计算结束时间，C2与C3单元格是使用不同方法运算，并将单元格数字格式设置为“常规”格式后显示的效果。

由于这个区别发生在对天数的进位上，所以并不会在时间值中显现出来，仅会在日期值中看到区别。在上个示例图7-28中，如果将C2与C3单元格设置为时间格式，显示的结果均为01:05:30，因为时间格式仅是对数值小数部分的描述，而C2与C3单元格内运算结果的小数部分是完全相同的。当结果涉及到天数的显示，这种差别就会立刻表现出来，如图7-29所示，使用两种不同方式得到的结束时间通过TEXT函数再次加工为有天数显示的时间信息。

注：本例使用自定义格式代码“d天h小时m分钟s秒”，“d”仅是从日期中抽取“日”的信息，实际上是将计算结果4.045486111转化为“1900年1月4日”1:05:30后提取“日”和时间，所以这样的表示方法最多只能计算31天。





7.3.3 TODAY函数与NOW函数获得当前时间




除了使用DATE函数与TIME函数组合成一个特定日期或时间外，使用TODAY函数和NOW函数也可以得到当前日期或时间。由于仅需要提取当前系统日期和系统时间作为运算结果，这两个函数在使用时都不需要添加参数。

1．函数语法：

TODAY()

NOW()

TODAY函数用来返回当前系统日期，例如在单元格中输入公式“=TODAY()”，单元格将显示当前日期，而在Excel中，工作表函数本身具备自动重算的特点，所以TODAY()函数得到的日期会随系统时间日期的改变而改变。对于一些经常需要打印，每次打印都需要加入当天日期的表格，可以使用TODAY函数来显示当天日期。这个日期与手工输入的日期相比，能自动更新为当天时间。

函数的自动重算需要一个动作来触发，例如重新打开工作簿、工作表中单元格内容被编辑、按下“重算”按钮（【F9】键）等。例如在单元格中输入公式“=TODAY()”，单元格将显示当前日期，修改系统日期后回到工作表中，按下【F9】键后，日期将变更为新的系统日期。

2．应用实例

由于TODAY的这个“善变”特性，使它常被用于提取当前的年、月、日等信息，再用DATE函数与其他数值组合在一起生成新日期，如图7-30所示问题，通过出生日期计算当年的生日。在B2单元格输入公式“=DATE（YEAR（TODAY()），MONTH （A2），DAY（A2））”后下拉填充，表示以当前的年为年、A2单元格日期的月为月、A2单元格日期的日为日，生成一个新日期，这个日期就是今年生日的日期。

与TODAY函数相比，NOW函数比TODAY函数计算的结果更加精确。NOW函数能返回包含日期信息的系统时间。例如在单元格内输入公式“=NOW()”，会得到一个既包含日期又包含时间的值，其数字格式为“yyyy-m-d h:mm”。我们可以将这个自定义数字格式修改为“yyyy-m-d h:mm:ss.00”后，持续按住【F9】键，会看到运算的结果在飞速变化，证明NOW函数会根据系统时间不停重算。“yyyy-m-dh:mm:ss.00”表示将时间精确到百分之一秒。





7.4 计算时间间隔








7.4.1 DATEDIF函数




DATEDIF是一个专门用于计算两个日期之间间隔的函数，这个函数很好地解决了由于日期进位方式不固定产生的日期区间难以计算的问题。

1．函数语法

DATEDIF（开始日期，结束日期，类型值）

开始日期

日期或可被运算为日期的值或表达式，不能大于结束日期。

结束日期

日期或可被运算为日期的值或表达式，不能小于开始日期。

类型值

希望得到的结果类型，例如间隔年数、间隔月数、间隔天数等，对应关系见表7-1。

表7-1




2．应用实例

DATEDIF函数是处理如图7-31所示的计算员工工龄问题的不二利器，在通过A2:A4区域内日期与B2:B4区域内日期计算日期间隔年数的计算中，在C2单元格输入公式“=DATEDIF（A2，B2，"y"）”，后下拉填充即可得到正确结果。通过C2单元格与C3单元格运算结果的比较，可以发现，DATEDIF函数很好地满足了工龄运算中“少一天则少一年”的高精度需求，这是将两日期直接相减得到间隔天数，再用间隔天数除以365得到年数的方式无法比拟的。




图7-31

从上例中可以看出，DATEDIF函数在运算时，将2001-8-30至2002-8-30日视为1整年，2001-8-30至2002-8-29日视为不足1整年。而在工作中，我们也经常将2001-8-30至2002-8-29日视为1整年，2001-8-30至2002-8-28日才视为不足1整年。在这种情况中计算工龄时，可以将上个示例中公式修改为“=DATEDIF（A2， B2+1，"y"）”，即为结束日期增加1天作为修正。

通过变更DATEDIF的类型值，可以得到两个日期之间不同的间隔信息，如图7-32所示，使用6种不同的类型值后迥异的计算结果。

3．常见错误

DATEDIF函数的错误常发生于开始日期与结束日期这两个参数中，这两个参数要求使用日期值或可被计算为日期值的其他类型值或表达式，在参数无法被转化为日期时将返回#NUM!错误，例如公式“=DATEDIF（20010830，20110830，"y"）”中由于开始日期和结束日期均超出了Excel所允许的最大日期限制，计算将返回错误值。

另外，在使用DATEDIF时需要注意两个待处理日期的先后顺序，DATEDIF参数中的结束日期不能小于开始日期，即结束日期不能是开始日期之前的日期，否则将返回#NUM!错误，例如公式“=DATEDIF（“2011-8-30”，”2001-8-30”，"y"）”。





7.4.2 NETWORKDAYS函数




1．函数语法

NETWORKDAYS（开始日期，结束日期，节假日）

开始日期

日期或可被运算为日期的值或表达式，大于结束日期时运算结果为负值。

结束日期

日期或可被运算为日期的值或表达式，小于开始日期时运算结果为负值。

节假日

指定的区域或数组，其中包含开始日期与结束日期间的节假日，如果节假日刚好是周末，函数不会重复统计。该参数可省略，省略后结果为开始日期与结束日期间去除周六、周日的天数。

2．应用实例

如图7-33所示，计算2011-6-1到2011-6-30之间的工作日，其中A6:A8区域为端午节假期的日期，C2单元格公式为“=NETWORKDAYS（A2，B2，A6:A8）”。




图7-33

在这个例子中，如果不考虑端午节的假期，只计算6月份的工作日，可以不选择节假日，即省略第三参数，公式为“=NETWORKDAYS（A2，B2）”，其结果为22。节假日区域虽然列出了3天的假期，但是由于2011-6-4和2011-6-5分别为周六和周日，在计算工作中不会重复统计，所以公式“=NETWORKDAYS（A2， B2，A6:A8）”的结果21只比省略第三参数的公式“=NETWORKDAYS（A2，B2）”少1天。

在编辑节假日区域时需要注意，如果节假日的日期不是开始日期和结束日期之间的日期，NETWORKDAYS函数将忽略该假期。

注意：在2003版中，NETWORKDAYS函数需要加载“分析工具库”才能正常使用，否则将返回“#NAME?”错误，加载方法见本书第2章第4节“加载分析工具库”。





7.4.3 WORKDAY函数




在工作和生活中，我们常听到这样的话：“请您在10个工作日内提交申请材料”、“我们将在5个工作日内结清余款”，对于这类使用工作日计算的日期，我们如何确定若干工作日后的具体日期呢？

与NETWORKDAY函数对应，WORKDAY函数表示对开始日期进行推算，计算若干工作日后的日期是哪一天。

1．函数语法

WORKDAY（开始日期，工作日，节假日）

开始日期

日期值或可被运算为日期的其他类型值或表达式。

工作日

数值或可被运算为数值的表达式

节假日

指定的区域或数组，其中包含开始日期与结束日期间的节假日，如果节假日刚好是周末，函数不会重复统计。该参数可省略，省略后工作日中将包含除周六、周日外的所有日期。

2．应用实例

如图7-34所示，在C2单元格计算“2011年6月1日”之后10个工作日的日期。C2单元格公式为“=WORKDAY（A2，B2，A6:A8）”。




图7-34

在WORKDAY函数的运算中，不会把起始日期算作第一个工作日，而是自起始日期的后一天算起，例如在2011-6-1和2011-6-2均为工作日的情况下，2011-6-1之后1个工作日的日期为2011-6-2。

如果本例公式省略第三参数，即不考虑节假日，公式为“=WORKDAY（A2， B2）”，其结果为2011-6-15。

注意：在2003版中，WORKDAY函数需要加载“分析工具库”才能正常使用，否则将返回“#NAME?”错误，加载方法见本书第2章第4节“加载分析工具库”。





7.5 星期相关计算








7.5.1 WEEKNUM




WEEKNUM函数能计算某日期是该年份的第几周，而该年份1月1日所在的周，将被视为第一周。

1．函数语法

WEEKNUM（日期，类型值）

日期

待计算周数的日期值，可以是日期或可被转化为日期的其他类型值或表达式。

类型值

类型值用来对一周的始末定义，可以为1或2。当类型值为1时，周日将被视为一周的第1天，即一周自周日开始；当类型值为2时，周一将被视为一周的第1天，即一周自周一开始。如果省略该参数，类型值将默认为1。

2．应用实例

如图7-35所示，使用WEEKNUM函数计算日期“2012年6月7日”属于2012年的第几周。其中B2与B5单元格计算的日期相同，但由于类型值不同而得到不同的结果。由于对周的定义不同，这两个单元格的计算结果产生了一周的差异。




图7-35

在B2单元格中，公式“=WEEKNUM（A2，1）”表示将周日视为一周的第1天，计算2012-6-7处于2012年的第几周。而2012年1月1日为周日，如果将周日视为一周的开始，则2012年1月1日至2012年1月7日为2012年的第1周、2012年1月8日至2012年1月14日为2012年的第2周，用这种方式推算得到2012年6月7日包含在2012年的第23周中。

在B5单元格中，公式“=WEEKNUM（A5，2）”，表示将周一视为一周的第1天，计算2012-6-7处于2012年的第几周。而2012年1月1日为周日，则第1周只包含1天，即2012年1月1日，2012年1月2日至2012年1月8日为2012年的第2周、2012年1月9日至2012年1月15日为2012年的第3周，用这种方式推算得，到2012年6月7日包含在2012年的第24周中。

注意：在2003版中，WEEKNUM函数需要加载“分析工具库”才能正常使用，否则将返回“#NAME?”错误，加载方法见本书第2章第4节“加载分析工具库”。





7.5.2 WEEKDAY




WEEKDAY函数能计算某个日期是该日期所在周的第几天，所以同样会由于对周起始日期定义的不同，而产生不同的计算结果。通过计算一个日期是该日期所在周的第几天，可以解决诸如某日期最近的周日是哪一天等问题。

1．函数语法

WEEKDAY（日期，类型值）

日期

待计算周数的日期值，可以是日期或可被转化为日期的其他类型值或表达式。

类型值

类型值用来定义一周的始末和返回值的类型，可以为1、2、3。类型值为1时，周日视为一周的第1天，结果将返回1～7的整数，分别对应一周的第1天（周日）～第7天（周六）；类型值为2时，周一视为一周的第1天，结果将返回1～7的整数，分别对应一周的第1天（周一）～第7天（周日）；类型值为3时，周一视为一周的第1天，结果将返回0～6的整数，分别对应一周的第1天（周一）～第7天（周日）。如果省略该参数，类型值将默认为1。

2．应用实例

如图7-36所示，计算2012年6月7日是日期所在周的第几天。2012年6月7日为周四，在使用了不同的类型值后，WEEKDAY函数产生了不同的计算结果。

B2单元格公式为“=WEEKDAY（A2，1）”，类型值为1时将周日视为一周的第1天，结果将返回1～7的整数，分别对应一周的第1天（周日）～第7天（周六）。在这样的规则中，周四为一周的第5天，B2单元格返回计算结果5。

B5单元格公式为“=WEEKDAY（A5，2）”，类型值为2时，将周一视为一周的第1天，结果将返回1～7的整数，分别对应一周的第1天（周日）～第7天（周六）。在这样的规则中，周四为一周的第4天，B2单元格返回计算结果4。




图7-36

B8单元格公式为“=WEEKDAY（A5，3）”，类型值为3时，将周一视为一周的第1天，结果将返回0～6的整数，分别对应一周的第1天（周日）～第7天（周六）。周四为一周的第4天，而返回结果的计数是以0开始的，第4天对应整数3，多以B8单元格返回计算结果3。

3．扩展应用

计算某个日期最近的周日的日期

如图7-37所示，计算2012年6月7日前一个周日的日期和后一个周日的日期。




图7-37

B2单元格公式为“=A2-WEEKDAY（A2，2）”，由于2012年6月7日为周四，公式“WEEKDAY（A2，2）”的计算结果为4，再将A2单元格日期减去4天，得到该日期的上个周日的日期“2012年6月3日”。

与上一个问题的处理方法略有不同，B5单元格公式为“=A5-WEEKDAY（A5）+8”，2012年6月7日为周日，由于公式“WEEKDAY（A5）”省略了第二参数，其结果将使用1～7表示自周日开始的一周，因此公式“WEEKDAY（A5）”的结果为5。而公式“=A5-WEEKDAY（A5）”将返回2012年6月7日的上个周六的日期，再将这个日期加8，得到2012年6月7日下一个周日的日期。

之所以在计算下个周日日期时，没有使用公式“=A2-WEEKDAY（A2，2）+7”，是因为这个公式在待计算日期为周日时返回的结果不正确。例如在计算2012年6月3日下一个周日的日期时，使用公式“="2012-6-3"-WEEKDAY（"2012-6-3"，2）+7”，其中WEEKDAY（"2012-6-3"，2）结果为7（2012年6月3日本身为周日），公式“="2012-6-3"-WEEKDAY（"2012-6-3"，2）+7”的计算结果为2012年6月3日，并没有返回下个周日的日期。

计算两个日期间周六的数量

如图7-38所示，计算开始时间与结束时间之间包含的周六的数量。在C2单元格中输入公式“=INT（（WEEKDAY（A2-6，2）+B2-A2）/7）”后下拉填充。

本例问题为计算两个日期间周六的数量，如果需要计算两个日期间周五的数量，则可以使用公式“=INT（（WEEKDAY（A2-5，2）+B2-A2）/7）”，即将公式中的6改为5。实际上，这个数字可以分别修改为0～6，0表示计算两个日期间周日的数量，1表示计算两个日期间周一的数量、2表示计算两个日期间周二的数量，依此类推。




图7-38





第8章 用函数增强Excel功能




“技巧是外家功夫，函数是内功。”——徐江（Excel精英培训网版主）

这句话的含义是，如果光练技巧这个“外功”，却不会用函数“内功”，或者不会将函数“内功”灵活地应用入技巧“外功”，练到一定程度就会出现瓶颈。

本章将展示一些常用技巧，让读者了解在使用过程中，如果有函数在背后支持，会进入怎样的一个新境界。





8.1 函数与数据有效性








8.1.1 数据有效性简介




数据有效性中的序列

数据有效性中有一种功能，就好像窗体里的“组合框”一样，选取单元格后，单元格的右边会出现一个倒三角，再单击这个倒三角，会出现一个“下拉列框”，只要单击里面的任意一项，就可以把这一项内容“选”进单元格，如图8-1所示。




图8-1

这一设置的过程非常简单。

2003版：执行“数据”→“有效性”→“设置”→“序列”菜单命令。

2007版或2010版：执行“数据”→“数据工具”→“数据有效性”→“设置”→“序列”菜单命令。

然后将指定序列通过引用的方式输入在“来源”下面的框里即可，如图8-2所示。

在图8-2中，“来源”下面的方框右边，有个这样的图标“
 ”。是不是看着有点眼熟呢？我们在“插入函数”对话框里就已经见过无数次了。事实上，这个小图标在Excel里到处可见，一般情况下，看到这个小图标，就意味着这里可以有3种方式输入内容：




图8-2

通过拖放单元格区域的方式输入单元格地址。

直接在里面输入字符或公式。

引用自定义名称。

当然这只是一个大体规则，具体的还是要根据各项功能自身的属性，比如同样是这个图标，有些可以引用其他工作表，甚至其他工作簿，有些则不可以。

据此，序列来源的设置方法，除了上述的引用单元格地址的方法以外，还可以直接输入字符，并以半角逗号分隔。

如图8-3所示，序列来源内容比较少，只需要限定“男”和“女”两种即可，这时可以直接输入“男,女”。

至于将自定义名称或者公式作为来源，本章后面的小节将做详细介绍。

数据有效性中的其他功能




图8-3

当然，数据有效性并不止“序列”一项功能，还有其他功能，如图8-4所示。




图8-4

数据有效性里的这些功能，除了“自定义”和“序列”以外，其他设置方法都基本相同，而且界面十分直观，操作亦非常简单，相信只要花点时间亲自操作一下，就不难掌握，这里就不一一介绍了。

清除数据有效性设置

数据有效性中并没有一个“清除”或者“删除”的设置，而是通过将有效性设置成“任何值”，从而达到了“清除”的目的。





8.1.2 跨工作表设置数据有效性




在2003版和2007版里，数据有效性无法直接实现跨工作表的设置，如图8-5所示。




图8-5

虽然在2010版里，这个问题已经得到解决，但目前仍是2003版占主导地位的时代，难道就不能进行跨工作表的设置了吗？

这里就需要用到我们之前提及的使用自定义名称作为数据有效性序列来源的方法。

关于自定义名称的设置，详见第2章。

例如把如图8-6所示的A2:A14这个区域自定义一个名称“江苏省”。




图8-6

再进行数据有效性的设置时，序列来源里不再是直接引用，而是引用这个自定义名称，如图8-7所示。




图8-7

通过这种方式设置，这个“来源”无论是本工作表还是跨工作表，都不会出错了。现在，即使使用2010版，最好仍以这个方式来设置跨工作表的数据有效性，免得换个版本打开后就报错。

至于跨工作簿的操作方法，十分麻烦。需要先在设置有效性的工作簿里，在任意单元格区域，将这个不在同一工作簿上的数据源的内容引用过来，再进行有效性设置，现实操作的意义不是很大，这里就不做详细介绍了。





8.1.3 用函数设置多级有效性




有时候，序列来源的内容太多，选取起来也十分麻烦，因为数据有效性的下拉显示项数只有8项，所以，如果有个“分类”将会方便很多。这就是我们常说的“多级有效性”。

如图8-8所示，D2单元格设置的是“一级有效性”，即各省省名；D3单元格设置的“二级有效性”。当D2单元格的内容为“江苏省”时，D3单元格就会显示所有江苏省的城市。




图8-8

当D2单元格改为“浙江省”时，D3单元格设置的“二级有效性”就会显示所有浙江省的城市，如图8-9所示。




图8-9

同样地，当D2单元格改为“安徽省”时，D3单元格的设置也会显示所有安徽省的城市。

这种设置其实非常简单。首先需要设置3个自定义名称：将A2:A14单元格区域名称定义为“江苏省”；将B2:B12区域名称定义为“浙江省”；将C2:C18区域名称定义为“安徽省”。

这里的3个区域大小不一样，只好逐一手工设置。如果区域大小一致，可以用自定义名称中的“指定名称创建于首行”来批量设置，详见第2章有关自定义名称的内容。

接下来就是设置数据有效性了，如图8-10所示。




图8-10

这里我们注意到，序列来源不再是直接引用单元格或自定义名称，而是一个公式：

=INDIRECT($D$2)

这个公式的含义当然十分明了，就是间接引用一个区域，被引用的区域名称显示在D2单元格里。当D2单元格的内容是“江苏省”的时候，这个公式得出的结果就是自定义名称“江苏省”所包含的所有内容，其他类推。

通过INDIRECT这一巧妙的运用，我们顺利实现了“二级有效性”，再要制作“三级”、“四级”、甚至更多级有效性，只要重复以上操作过程即可。

但是，每个省下属的城市数量未必固定，如果数量发生变化，例如图8-10中，当D2的内容为“江苏省”时，D3单元格就会显示出所有江苏省的城市，即A2:A14这个区域的内容。假如我们在A15单元格里增加了一个“丹阳市”，D3的下拉列框中是否能自动将这个“丹阳市”加进去呢？

我们可以将自定义名称的引用区域扩大，如图8-11所示，引用了A2:A65536这么大的一个区域。




图8-11

如此一来，无论A列的内容如何增减，一定都会在二级有效性的下拉列表框里显示出来。但是这样设置，除了A列必须没有其他不相关的内容干扰以外，我们还要忍受大把的“空地”，这个“空地”的大小，取决于A列曾经被用过的最下面一个单元格的位置，如图8-12所示。




图8-12

如何才能不要出现“空地”的同时，引用的内容随数据的增减而增减呢？这就需要在自定义名称中设置公式，而不是单纯地直接引用。

一般情况下，OFFSET套用COUNTA公式可以引用一个动态的区域，例如：

=OFFSET($A$2,,,COUNTA($A:$A)-1) 

1




1 这个公式逗号间省略了参数0，完整的写法应是“=OFFSET($A$2,0,0,COUNTA($A:$A)-1,1)”。

这个公式就是引用了A列所有数据的内容，并且可以根据A列数据的增减，引用区域大小也随之增减。类似的公式下面还会多次出现，都是用来动态引用区域。但是需要注意的是，这里A列的数据必须都是连续的，中间不可以出现空单元格，否则引用的区域就会与实际内容不一致。

但是在这里，仅仅动态引用区域还不够，还必须与D2单元格里设置的一级有效性相关联，这里不能用上述数据有效性来源用的INDIRECT函数方法，只能在这个公式的基础上进行修改。

首先，OFFSET的第3个参数，即向右偏移的部分，不能固定不动，而是要用一个MATCH函数来实现可以根据D2单元格设置的一级有效性进行动态偏移，公式为：

MATCH($D$2,$1:$1,)-1

再来看原公式的COUNTA部分，原来的参数“A:A”，如果换成OFFSET的写法，可以写成：

OFFSET($A$1,,,65536)

同理，需要用上述的MATCH换掉原来的第3个参数“0”，才能实现根据D2单元格设置的一级有效性进行动态偏移：

OFFSET($A$1,,MATCH($D$2,$1:$1,)-1,65536)

如此一来，OFFSET的第4个参数，即向下扩展的行数就写成：

COUNTA(OFFSET($A$1,,MATCH($D$2,$1:$1,)-1,65536))-1

最后将所有的组合起来，合并成一个完整的公式：

=OFFSET($A$2,,MATCH($D$2,$1:$1,)-1,COUNTA(OFFSET($A$1,,MATCH($D$2,$1:$1,)-1,65536))-1)

这个公式看似复杂，但只要之前完全掌握了OFFSET、COUNTA、MATCH这3个函数的基本用法，也不难理解。

接下来将这个完整的公式放在自定义名称中，并且另外命名，如图8-13所示命名为“data”。




图8-13

最后数据有效性直接引用这个自定义名称，就可以实现完全的动态功能了，如图8-14所示。




图8-14

这种动态的引用，还可以利用“列表

12


 ”功能来实现，关于这一功能的使用，将在本章的8.4.1小节详细介绍。

2 “列表”是2003版的名称，同样功能在2007版中叫“表”，2010版中叫“表格”，这里为避免混乱，统一称为“列表”。





8.1.4 数据有效性中的“自定义”




在数据有效性的设置中，最后一个“自定义”的用法，与其他略有不同，数据结果直接以“公式”来限制，那这里的“公式”又是如何设置的呢？

我们先来复习一下逻辑函数IF：

=IF(条件,是则结果,否则结果)

这个公式的结果是根据条件来判断的，条件为逻辑真（“TRUE”）或大于0的数字时，第2个参数就会作为整个公式的结果；反之条件为逻辑假（“FALSE”）或等于0时，第3个参数就会作为结果。

但是，函数不可能对对象的属性进行设置，所以我们不可以设置这样一个公式：

=自定义函数名(条件,指定单元格可以输入,指定单元格无法输入)

但是，把数据有效性中的“自定义”与函数相结合，就可以达到这样的效果。

换句话说：在数据有效性的“自定义”中，输入类似IF函数第1个参数的公式，就可以得到根据这个条件限制单元格输入的效果。

我们以身份证号位数限制为例，假设需要在A列里输入身份证号，身份证号的位数只有15位和18位，这里的“条件”就是：

=OR(LEN(A1)=15,LEN(A1)=18)

选取A列后设置数据有效性，并将上述公式输入“公式”栏里，如图8-15所示。




图8-15

这里有一点非常需要注意：选取A列后，在名称框里显示的是“A1”，这时的公式全部引用的是A1单元格，假如设置时选取区域后，名称框显示的是A2单元格，则公式里的引用要全部改成A2单元格。造成这一现象的原因是相对引用在起作用。有关绝对引用和相对引用，请详见第2章。

接下来，在A列输入身份证号时，只要位数不是15位或18位的，都会出现提示，如图8-16所示。




图8-16

可以通过这种方式设置很多限制，再比如设置只有在B列某行为空的时候，A列同行才可以输入内容，可以使用公式：

=B1<>""

设置过程如图8-17所示。




图8-17

设置好以后，如果B3单元格里已有内容，这时再往A3单元格里输入内容，就会出现如图8-18所示的提示。




图8-18

可见这类设置的过程都一样，所不同的就在于公式的使用。





8.2 函数与条件格式








8.2.1 条件格式简介




2003版与2007、2010两个版本的条件格式虽然界面大不相同，但是基本用法还是一致的。

2003版：执行“格式”→“条件格式”→“单元格数值”命令，进而设置范围→设置格式。

2007版或2010版：执行“开始”→“样式”→“条件格式”→“新建规则”→“只为包含以下内容的单元格设置格式”→“单元格值”命令，进而设置范围→设置格式。

假设我们要让以下数值小于60的反白高亮显示，只要选取A1:H5这个区域，按上述过程设置即可，2003版的设置如图8-19所示。




图8-19

2007版或2010版的设置如图8-20所示。

两者界面虽然差异很大，但实质还是基本相同的，具体的设置过程，并无任何难度，只要动手操作一下，相信要掌握并非难事。

相比之下，2007版或2010版比2003版多出了不少新功能，但其所新增的，大部分都是些“花哨”效果，以满足视觉上的享受。这些操作亦是十分简单，如图8-21所示的就是一个分别用“×”、“！”、“√”显示数字的条件格式。




图8-20




图8-21

在条件格式的对话框里，我们再次看到了这个图标“
 ”，同样也就意味着，这里可以通过拖放单元格区域的方式输入单元格地址；或者是直接在里面输入字符或公式；或者是引用自定义名称。

例如图8-22所示的，就是将条件放在A7单元格里，如此一来，需要修改条件时，就不再需要打开“条件格式”对话框，而是直接改一下A7单元格里的数字就可以了。




图8-22

2003版和2007版的条件格式和数据有效性一样，也是不能实现跨工作表设置的，只有2010版可以实现，这时我们同样需要借助于自定义名称来解决这个问题。具体的操作过程和数据有效性类似，只要将事先设置好的自定义名称填在带这个图标“
 ”的框里即可。





8.2.2 用函数公式作为条件格式的条件




有时候，我们需要设置的条件用一个固定的数字无法解决，比如设置过期高亮显示，因为日期每天都在变化，所以就需要使用显示系统日期的函数TODAY，如图8-23所示。

但是，这样的效果看上去似乎有点“别扭”，条件格式设置时虽然选取的是A2:E11这么大一个区域，但能够高亮显示的只有A列的内容，如何如图8-24所示的那样，让其他列也跟着A列一起高亮显示呢？




图8-23




图8-24

我们还是来复习一下逻辑函数IF：

=IF(条件,是则结果,否则结果)

这个公式的结果是根据条件来判断的，条件为逻辑真（“TRUE”）或大于0的数字时，第2个参数就会作为结果；反之条件为逻辑假（“FALSE”）或等于0时，第3个参数就会作为结果。

但是，函数不可能对对象的属性进行设置，所以我们不可以设置这样一个公式：

=自定义函数名(条件,指定格式,应用原格式)

但是，把条件格式的“公式”与函数相结合，就可以达到这样的效果。

换句话说：在条件格式的“公式”中，输入类似IF函数第1个参数的公式，就可以得到根据这个条件设置单元格格式的效果。

就上图的例子，这里的条件是（注意绝对引用）：

=$A2<TODAY()

选取A2:E11后设置条件格式，并将上述公式输入公式里，如图8-25所示。




图8-25

至于这个公式为什么用的是“A2”单元格，对照一下名称框，应该很清楚其中的原因了吧，如果还是有点不明白，那就去复习一下第2章有关绝对引用和相对引用的内容。

在这个例子中，我们选取的用来做条件格式的区域是A2:E11。但在现实中，为了保证再增加的数据也适用条件格式，会直接选取A:E列设置条件格式。这时就会发现，很多空格也“被高亮”了，如图8-26所示。




图8-26

这就需要我们对公式进行进一步的“改造”，让其忽略空单元格。

=AND($A1<TODAY(),$A1<>"")

结果如图8-27所示。




图8-27

在限制输入方面，如果说数据有效性是属于“事先控制”的话，那么条件格式就是“事后弥补”。图8-15和图8-17两个例子都是在输入数据之前设置数据有效性，以保证以后的输入受限制。但如果数据已经输入完成，将同样的公式放在条件格式里高亮显示错误，要比数据有效性的“圈释错误”功能更加有效持久。





8.3 函数与高级筛选








8.3.1 高级筛选简介




2003版的“数据”→“筛选”→“自动筛选”菜单功能，可以帮助我们实现对大堆数据的筛选显示。但是，当筛选条件变得十分复杂时，往往自动筛选无法满足要求，虽然在2007版或2010版的自动筛选有了巨大的改进，但当面对千变万化的条件时，仍显得有点“力不从心”。这时，我们就需要借助“高级筛选”来完成。

2003版：执行“数据”→“筛选”→“高级筛选”命令。

2007版或2010版：执行“数据”→“排序和筛选”→“高级”菜单命令。

高级筛选光有数据源的列表区域还不够，还要有用来设置条件的条件区域，如图8-28所示。




图8-28

如果希望筛选的结果在其他地方显示，只要将“方式”改为“将筛选结果复制到其他位置”，在“复制到”指定位置即可，筛选结果如图8-29中A11:D13单元格区域所示。




图8-29





8.3.2 用函数作为高级筛选的列表区域




列表区域的数据有时是会动态变化的，例如图8-29所示的例子，目前员工一共是8人，随着时间的推移，员工人数会出现增加或减少。这时，我们可以用一个公式来引用这个不断变化的列表区域：

=OFFSET($A$1,,,COUNTA($A:$A),4)

无论人数增加还是减少，这个公式所引用的结果都会包含所有员工数。

将这个公式直接放到“列表区域”里，高级筛选一样会忠诚地执行着它应执行的结果，如图8-30所示。




图8-30

不过这样一来，似乎变得更加繁琐了，本来每次高级筛选只要用鼠标选取个区域，现在变成了每次高级筛选都要写个公式。当然，我们可以通过自定义名称简化一下，将这个公式自定义为一个名称，再在列表区域内引用这个自定义名称。但即使使用自定义名称，高级筛选的使用仍不是十分方便。

事实上，高级筛选的过程本来就是繁琐的，所以经常和“录制宏”一起使用。而本小节所描述的内容，正是作为将来进一步学习的基础，并非强制大家在工作中使用如此繁琐的操作。





8.3.3 用函数作为高级筛选的条件区域




回想一下，高级筛选是不是和数据库函数十分相像呢？数据库函数里的第1个参数和第3个参数，分别对应高级筛选的“列表区域”和“条件区域”，不仅如此，两者的使用都十分相像，高级筛选的条件区域可以是多条件的“与”和“或”，可以用比较运算符与通配符，也可以用公式。

例如第4章图4-7所示的例子，加上高级筛选的结果如图8-31所示。

可见，只要熟练掌握了数据库函数中条件参数的用法，也就同样掌握了高级筛选的条件区域用法。




图8-31





8.4 函数与数据透视表








8.4.1 数据透视表简介




数据透视表是极佳的多条件统计工具，而且操作极为方便快捷，所以被广泛运用，如图8-32所示的例子。




图8-32

如果要计算各部门和职级的工资小计，利用数据透视表功能，只需要几个动作就可以完成了。

2003版：执行“数据”→“数据透视表和数据透视图”命令，完成后将字段拖放至表中。

2007版或2010版：执行“插入”→“数据透视表”命令，完成后勾选字段。2003版数据透视表结果如图8-33所示。




图8-33

2007版或2010版数据透视表结果如图8-34所示。




图8-34

可见，不同版本除了“长相”略有不同外，基本功能都一致。

数据透视表还有非常好的“联动”功能，不过这个“联动”并不像函数那样，数据源一有变化，结果就马上跟着变，而是要按下“刷新”按钮。

2003版：执行“数据”→“刷新数据”命令，或者选取数据透视表后，在透视表工具栏上找到“刷新数据”按钮。

2007版或2010版：选取数据透视表，执行“数据透视表工具”→“选项”→“数据”→“刷新”命令。

如图8-35所示，D9单元格由原来的1000改成了5000，按下“刷新数据”按钮后，数据透视表的结果也发生了变化。




图8-35

虽说这种数据的“联动”给我们使用带来了很大的方便，但是这仅限于数据源区域内数据发生变化，并不适合数据源区域大小的变化。如图8-36所示，员工增加了一人，如果不重新设置数据透视表的数据源，而是仅仅按下“刷新”键，数据透视表并不会把增加那一人的数据更新上去。




图8-36

这时候，我们可以将数据源转化成“列表”：

2003版：选取数据源区域，执行“数据”→“列表”→“创建列表”命令。

2007版或2010版：选取数据源区域，执行“插入”→“表”/“表格”命令。

再以这个列表作为数据透视表的数据源，如此一来，数据透视表的内容不仅会随数据本身更新，也会随列表区域的大小更新。下面我们将图8-36中的A1:D9这个区域设置成列表，如图8-37所示。




图8-37

然后点住右下角的“
 ”向下拖放扩大列表区域，再刷新数据透视表后，数据透视表的数据源区域也随之扩大，如图8-38所示。

列表有其优点，特别是2007版和2010版，列表内的数据还自动设置成隔行着色。另外，除数据透视表以外，数据有效性如果需要设置单列的动态区域，也可以引用列表。




图8-38

但是有一些功能在列表的基础上无法使用，如分类汇总等，而且列表多少有点改变表格的“原形”，可能会产生一些视觉上的不适应。





8.4.2 用函数作为数据透视表的数据源




除了列表以外，函数也可以做到类似效果，这就需要先构建一个随数据源区域大小变化而变化的动态区域。这个公式我们之前不止一次地提及，就是：

=OFFSET($A$1,,,COUNTA($A:$A),4)

当然，这个公式可以直接贴入数据透视表步骤之2：“选定区域”旁边的框里，如图8-39所示。




图8-39

但是这样一来，系统会自动将这个公式进行计算，转成这个公式所引用的固定区域，结果是又变回了“静态”，完全没有达到我们想要的效果。

如此，我们就需要借助于自定义名称了。先将这个公式自定一个名称，并取名“data”，如图8-40所示。




图8-40

然后，将数据源区域改成对这个“data”的引用，如图8-41所示。




图8-41

完成以后的数据透视表，就可以随着数据源区域行数的增减而变化了。





8.5 函数与图表








8.5.1 图表简介




如图8-42所示，A1:B7是日销量的流水账，虽然数字已经完全能说明问题了，但显示结果不是那么直观，如果能以图表的方式显示，一眼就能看出销量每天的波动情况。




图8-42

Excel提供了操作非常方便的图表功能。

2003版：首先选取数据源

 ，然后执行“插入”→“图表”命令，再选择图表类型，完成操作。

3 “数据源”是2007版或2010版的名称，2003版称为“源数据”，不过“数据源”的称法似乎更容易被接受，所以此处统称“数据源”。

2007版或2010版：首先选取数据源，然后执行“插入”→“图表”命令，再选择图表类型。

2003版折线图的结果如图8-43所示。




图8-43

2007版或2010版折线图的结果如图8-44所示。




图8-44

可见，除高级版本“长相”更美以外，其他并无太大差别。

图表的制作方式千变万化，上述是一种快捷方式。在选取数据源时，我们也可以通过以上方式实现。

2003版：执行“插入”→“图表”命令，再选择图表类型，下一步可选取数据区域或自定义系列。

当然，图表完成后，并非就不能再修改，我们还可以通过以下方式来修改数据源：

2003版：选取图表，执行“图表”→“源数据”菜单命令。

2007版或2010版：选取图表，执行“图表工具”→“设计”→“数据”→“选择数据”菜单命令。





8.5.2 用函数制作动态图表




我们在设置数据区域或者系列时，同样见到了这个图标“
 ”。这同样意味着，这里可以通过拖放单元格区域的方式输入单元格地址；或者是直接在里面输入字符或公式；或者是引用自定义名称。

图表的数据源设置或修改分两个部分，一是“数据区域”部分，可以直接指定数据区域，而被指定的数据区域通常都是连续的矩阵，如图8-45所示。




图8-45

当然，在“数据区域”里输入一个OFFSET的动态引用公式也可以。不过，和数据透视表一样，系统会自动将这个公式进行计算，转成这个公式所引用的固定区域，结果是又变回了“静态”。更不幸的是，即使将动态区域设置成自定义名称，在这里系统仍会自动将这个自定义名称进行计算，转成这个自定义名称所引用的固定区域。至此我们得出的结论是，如果想用函数设置动态区域，不能在“数据区域”部分设置。

修改图表的数据源的另一部分是“系列”部分，通过设置不同系列的方式来指定数据源，系列的名称与值的引用如图8-46所示。

这里可以输入的种类就比较多了，可以直接是文本，可以引用自定义名称，还可以是常量数组，而且系列还可以根据自己的需要增减，如图8-47所示。




图8-46




图8-47

2007版或2010版的数据源编辑，不再分成两个部分，而是直接在一个界面上显示，虽然“长相”差别比较大，但实质内容还是一致的，如图8-48所示。




图8-48

根据“系列”部分的特性，当我们需要用函数对图表的数据源进行动态设置时，就可以通过定义自定义名称的方式实现。

我们仍以图8-42中所示的数据为例，系列值引用公式：

=OFFSET($B$2,,,COUNTA($A:$A)-1)

分类轴标志引用公式：

=OFFSET($A$2,,,COUNTA($A:$A)-1)

将这两个引用自定义为名称“data”和“item”，如图8-49所示。




图8-49

再将这个“data”和“item”分别作为图表的系列值和分类轴标志。2003版如图8-50所示。




图8-50

2007版或2010版修改数据系列如图8-51所示。




图8-51

2007版或2010版修改分类轴标签如图8-52所示。




图8-52

经过这一设置，这个图表的数据源无论行数如何变化，图表都会自动跟着增减，如图8-53所示。




图8-53

如此复杂的一个操作过程，我们也可以非常简化地用列表功能来解决，将图表的引用区域设置成列表，那么列表区域发生变化，图表的数据源也会跟着变化。

但是列表的表头位置是固定不变的，如果将上述问题升个级，不是数据源区域随着日期的增加而扩大，而是随之后移，即永远显示最后6天的记录，那只能用公式了。

系列值引用公式：

=OFFSET($B$2,COUNTA($A:$A)-7,,COUNTA($A:$A)-1)

分类轴标志引用公式：

=OFFSET($A$2, COUNTA($A:$A)-7,,COUNTA($A:$A)-1)

然后将这两个公式设置为自定义名称，并分别被图表的系列值和分类轴标志引用，操作过程与上述完全一致。

还有一些比较流行的，比如通过指定单元格的内容显示相应的图表内容等，无非就是使用引用函数与指定单元格关联，其他操作完全一样。可见，动态图表的千变万化，其实很大程度上是动态引用函数的千变万化。





8.6 函数与图片




图片虽然看似和函数没什么关系，但是我们仍然可以借助函数来管理图片，如图8-54所示，B列每个单元格里放置着一张图片，A列是对应图片的编号。




图8-54

当图片很多的时候，一张一张去寻找十分麻烦，这时我们可以通过输入图片编号直接找出图片的方式来实现快捷的查找，如图8-55所示，在D2单元格里输入图片编号“2”，对应的图片就直接在E2单元格里显示。




图8-55

我们先要设置一个引用公式，与D2单元格关联。这里可不能用VLOOKUP，因为VLOOKUP是一个查找公式，不能起到引用的效果。公式应设置成：

=OFFSET($B$1,MATCH($D$2,$A:$A,)-1,)

这个公式并不难理解，就是从B1单元格向下偏移，而偏移数正好是D2单元格的值在A列的位置减1。这个公式直接放在单元格里使用，只能得出一个“0”的结果，因为函数无法对对象属性进行设置，自然也就无法把图片引出来，只能引出图片所在单元格的值。

当然，要把图片和公式串在一起，还必须借助于自定义名称，如图8-56所示，将刚才的公式设置成一个自定义名称“data”的引用。




图8-56

下一步，就要在图片上进行操作了，操作的方法很多，这里我们介绍一种非常快捷方便的方法：先选取B列带图片的任意单元格复制，注意复制的是单元格而不是图片。然后，不同版本操作方法略有不同，如下所述。

2003版：选取E2单元格，按下【Shift】键，再执行“编辑”→“粘贴图片链接”命令

2007版：选取E2单元格，执行“开 始”→“剪贴板”→“粘贴”→“以图片格式”→“粘贴图片链接”命令。

2010版：选取E2单元格，执行“开始”→“剪贴板”→“粘贴”→“其他粘贴选项”→“链接的图片”命令。

如此一来，我们选取图片时，可以在编辑栏看到一个直接引用，将这个引用改成刚才自定义的名称：

=data

注意，这里一定要选取图片后再在编辑栏里修改，也即，必须保证名称框里显示的是图片名而不是单元格名时，才能进行修改，否则只是单元格引用了“data”这个名称，如图8-57所示。




图8-57

然后在D2单元格里输入不同的编号，相应的图片就会自动显示在E2单元格里。





8.7 附：各类功能与函数的关系




附各类功能与函数的关系，见表8-1。

表8-1




续表








第9章 公式简化与调试




有经验的Excel使用者都知道，编写公式的过程就是一个不断调试的过程。即使一个成功的公式也不可能放在任何环境下都能使用。一个复杂的公式往往是由若干个独立公式组合而成。在编写公式过程中，需要掌握一些基本的技巧。





9.1 函数返回错误值的处理




Excel有7种错误值，这7种错误值在编写公式过程中都可能会出现。有时候错误值并不一定影响计算结果，只是有碍观瞻而已，可有时错误值的出现，却会给后续的计算带来麻烦。这就需要我们了解每一种错误值出现的含义，在公式中尽量避免。





9.1.1 #NAME?——晕，这是什么名字




“#NAME?”表示使用了Excel不认识的“名字”。这一错误往往是写错了函数名称或使用了未经定义的名称，只要书写正确即可消除。





9.1.2 #DIV/0!——晕，竟然除以0




“#DIV/0!”表示公式中用“0”值作了除数。有时作为除数的“0”值并不显而易见，它可能是一个空单元格，也有可能是另一公式的结果。如图9-1所示，D2单元格的公式用空单元格C2作除数，而D4单元格的公式用“0”值作为除数。




图9-1

1．规避“#DIV/0!”

如果出现“#DIV/0!”影响观瞻，可以使用IF函数、信息函数ISERR和ISERROR规避错误值。如图9-1所示，D2单元格的公式可改写成“=IF(C2=0,"",(B2-C2)/C2)”或“=IF(ISERR((B2-C2)/C2),"",(B2-C2)/C2)”。

如果出现的“#DIV/0!”影响进一步的计算，可以将上述公式中的“""”，改写为特定的值。如将公式改写为“=IF(C2=0,100,(B2-C2)/C2)”。

2．利用“#DIV/0!”

在统计函数的参数中，如果有错误值，结果往往也是错误值，因此需要在公式中加以规避。但在一些查找函数中，我们却常常主动制造错误值，以方便定位。如图9-2所示，想得到上半年首次盈利额，辅助列C2公式为“=0/(B2>0)”，而单元格D2公式为“=OFFSET(B1,MATCH(0,C2:C7,),)”。其中公式的关键部分“MATCH(0,C2:C7,)”得到的结果是“4”，意思是查找首个“0”在C2:C7中的位置。查找引用函数在查找时忽略区域中的错误值，而在辅助列人为制造了错误值。公式中，“B2>0”得到的是一个逻辑值“FALSE”或“TRUE”，如果是逻辑值“FALSE”作分母，就会得到错误值“#DIV/0!”，而用“TRUE”作分母时，相当于除以“1”。

如果学习了数组公式，可以省略掉辅助列，直接用数组公式“=OFFSET(B1,MATCH(,0/(B2:B7>0),),)”求出“首次盈利额”。




图9-2





9.1.3 #N/A——晕，无效数据




“#N/A”表示公式引用了一个含有“#N/A”的单元格，或查找引用函数没有发现匹配值。在查找引用函数公式中，有时仅仅是数据格式不一致，即可造成此类错误。





9.1.4 #NULL!——晕，竟然是空的




“#NULL!”表示交叉引用为空，如公式“=SUM(B1:B7 A8:C9)”，在参数“B1:B7”与“A8:C9)”之间有一个“空格”，这个“空格”就是两个范围的“交叉引用运算符”。但是这两个范围根本没有交叉，因此结果为“#NULL!”。





9.1.5 #NUM!——晕，数字错了




“#NUM!”表示公式或函数中使用了无效的数字值。如公式“=SQRT(-1)”就返回“#NUM!”。函数在什么情况下会出现错误值“#NUM!”，一般在函数的帮助文件中会有介绍。





9.1.6 #REF!——晕，引用的范围不存在




当公式引用一个无效的单元格时，就会出现“#REF!”错误。复制公式或删除公式所引用的单元格时，容易出现这类错误。如图9-2中，将B列直接删除或将C2单元格的公式复制到A列，都会造成“#REF!”错误。





9.1.7 #VALUE!——晕，用错值了




“#VALUE!”错误最为常见，大多是因为用错了参数类型。如公式“=MID(A6,3,-1)”试图截取文本的第“2”个字符，但MID函数的第三个参数不允许为负数，因此公式结果为“#VALUE!”。





9.2 公式常见的其他错误提示




Excel不是事后诸葛亮，尽管有7种错误值可以显现公式的一些错误，但更多的时候，Excel是在编写公式过程中就给出错误提示。





9.2.1 公式错误是否接受更正建议




输入公式时，如果括号不匹配，就会出现如图9-3所示的错误提示。平时要养成好习惯，成对输入括号，再退回到括号内编辑函数参数。如在图9-2的C2单元格输入公式“=0/(B2>0”，当回车结束输入时，就会弹出如图9-3所示的“更正建议”。意思是Excel发现你少输入了一个“括号”，并给出了它认为正确的公式，如果单击下面的“是（Y）”按钮，Excel会自动将公式修改为它认为正确的公式，如果单击“否（N）”按钮，Excel会锲而不舍地给出如图9-4所示的提示。如果继续单击“确定”按钮，或关闭此提示，公式停止运算，光标停留在公式编辑栏最有可能出错的位置上。




图9-3




图9-4

值得注意的是，在图9-3的提示中，Excel给出的“更正建议”一般情况下都是正确的，但也并非绝对，有时它给出的更正建议虽然符合括号匹配原则，但结果未必是我们想要的。

当我们编辑一个公式时，不同层次的括号有不同的颜色，相互匹配的一对括号使用同一颜色。当光标移动到某一括号上暂作停留时，Excel会主动用粗体把此括号和与它相匹配的括号显示约一秒钟。





9.2.2 循环引用不能计算公式




如无特殊计算需要，一般情况下，循环引用是我们不希望看到的。因此默认情况下，公式中出现循环引用，Excel都会立即给出警告提示。

1．循环引用提示

如果公式直接或间接地引用它自己所在的单元格，就会引起“循环引用”问题。如在A1单元格输入公式“=SUM(A1,A3)”，会弹出如图9-5所示的提示。如果单击“确定”按钮，会显示如图9-6所示的“循环引用”工具栏。




图9-5




图9-6

“循环引用”工具栏有一个“列表框”和3个“按钮”。在“定位循环引用”列表框，可以选择包含循环引用的单元格。“追踪从属单元格”按钮，是查找活动单元格被哪个单元格的公式所引用。“追踪引用单元格”按钮，是查找活动单元格中引用了哪个单元格。如果查找到“从属单元格”或“引用单元格”，会在两者之前显示一条蓝色箭头线，箭头一端是包含循环引用公式的单元格，圆点一端是被引用的单元格。工具栏上的“取消所有追踪箭头”按钮，可以取消这条蓝色箭头线。

每查找到一个循环引用公式，进行修改后，如果工具栏未消失，说明工作簿中仍存在有其他循环引用。在“定位循环引用”列表框中会显示下一个包含有循环引用的单元格。如果将所有循环引用公式修改正确，工具栏会自动消失。

2．利用循环引用

大多数情况下，我们不希望出现“循环引用”问题。但有时循环引用能解决一些看似很难解决的问题。如果要利用循环引用，必须要设置“迭代运算次数”。

例1：计算1～00的整数和。如图9-7所示，在A1输入公式“=A1+1”，在B1输入公式“=A1+B1”，结束公式输入时，如果提示“循环引用”，可以直接单击“取消”按钮。依次执行菜单栏的“工具”→“选项”→“重新计算”命令，将“迭代计算”选中，在“最多迭代次数”栏输入100。单击“确定”按钮，你会发现，A1单元格的值瞬间变为“100”，而B1单元格的值变为“5050”。




图9-7

这一过程瞬间完成，实际上却计算了100次。第一次A1加“1”变成“1”，B1也为“1”。第二次计算，A1又加“1”，变成“2”，B1为“1+A1”，结果是“3”。第三次计算，A1再加“1”变成“3”，B1为“3+A1”，结果是“6”。依次计算100次，B1单元格得到“1+2+3+4……+100”的值“5050”。

例2：文本倒排。Excel函数中没有提供将文本倒排的函数，除可以使用VBA函数进行文本倒排外，可以使用循环引用达到目的。如图9-8所示，在B1输入公式“=B1+1”，在C1输入公式“=MID(A1,B1,1)&C1”，由于A1中的文本共有12个字符，因此需将“最多迭代次数”设置为“12”。单击“确定”按钮后，C1得到原文本的倒排文本并连接一个“0”。




图9-8





9.2.3 为函数输入太少或太多参数




有些函数当输入的参数过多或过少时，会弹出错误信息。如输入公式“=MID(B1,3,2,1)”，会弹出如图9-9所示的“参数过多”提示。而输入公式“=MID(B1,3)”，又会弹出“参数过少”提示。




图9-9

注意，有时出现此类提示，可能仅仅是在输入函数参数时，多加了一个逗号或少写了一个逗号。如公式“=MID(B1,3,2,)”一样会弹出“参数过多”提示，而公式“=MID(B1,32)”会弹出“参数过少”提示。





9.2.4 是否更新链接




当打开一个工作簿时，可有会弹出如图9-10所示的提示信息，这是因为工作簿包含有对其他工作簿引用的公式。




图9-10

如果你确信被引用工作簿的数据，可以选择“更新”按钮，这时包含有外部引用的公式，会自动按被引用工作簿的新数据进行更新。如果不确信，可以选择“不更新”。如果想让这些外部引用公式全部转换为值，而其他公式不受影响，可以执行菜单栏的“编辑”→“链接”命令，调出如图9-11所示的“编辑链接”对话框。在此对话框中单击“断开链接”按钮，可以将全部外部链接转换为值，而工作簿中的其他公式则不受影响。




图9-11





9.2.5 输入的公式中包含错误




在编辑公式时，如果发生缺少紧跟函数名的左括号、参数超出限制、嵌套超过7层等语法错误时，会弹出如图9-12所示的错误提示。单击“确定”按钮后，公式停止计算，并将光标定位在出问题的疑似部位。




图9-12

如输入公式“=SUM(B1,B2,B3,B4,……,B100)”，就会出现“公式错误”提示。单击“确定”按钮后，光标点位在参数“B31”上。因为SUM函数的参数最多只允许有30个，参数“B31”就是疑似问题部位。将公式再加一对括号可以规避此类问题，如公式写成“=SUM((B1,B2,B3,B4,……,B100))”。





9.2.6 单元格填满“#”号




单元格里充满了“#”号有两种原因，一是由于单元格宽度不够容纳数字长度，另一种是公式返回了无效的日期或时间。如果单元格足够宽，都无法让“#”号消失，那么问题一定是后者。





9.3 没有提示的公式错误




公式中如果存在语法和逻辑错误，Excel都会有明确的提示。但我们编辑公式要的是正确结果。如果在编辑公式中忽略以下几个问题，即使写出符合语法和逻辑的公式，也未必能得到正确的结果。





9.3.1 正确公式复制后得不到正确结果




一个正确的公式，复制到其他单元格后，有可能得不到正确结果。这大多是由于忽略了公式中的相对与绝对引用。

如在C1单元格，公式“=A1+B1”计算是正确的。可将公式复制到A1单元格，就会出现“#REF!”错误。这是由于原公式使用了相对引用，复制到新位置后产生了“引用溢出”。

再如，一个如图9-13所示的“九九乘法表”，如果在B2单元格输入公式“=IF($B$1>A2,"",$B$1&"*"&A2&"="&$B$1*A2)”，向下复制是正确的，可向右复制就会得到错误值“#VALUE!”。将公式修改为“=IF(B$1>$A2,"",B$1&"*"&$A2&"="&B$1*$A2)”，才能得到如图所示的结果。




图9-13





9.3.2 按顺序书写的公式不按顺序计算




如希望计算25减2，即23的平方，如果公式写成“=25-2^2”，得到的结果肯定不是希望的结果。公式不一定按书写的顺序进行计算，这个概念有小学文化的人就懂，这是因为运算符有“优先级”。但并不是每个人都十分清楚Excel的所有运算符及其优先级。

Excel的运算符及优先级为：负号（-），百分比（%），指数（^），乘除（*，/），加减（+，-），文本连接（&），比较（=，<，>，<>）。

括号可以改变优先级。如以上公式可以写成“=(25-2)^2”。在书写公式时，如果对运算符的优先级不是很有把握，建议使用括号确保运算正常。

如对优先级十分清楚，可以省略一些不必要的括号。如图9-14所示，C2单元格的公式按常规可能会写成“="盈余"&(A2-B2)”。但由于“减号”的优先级排在“文本连接符”的前面，因此公式可以简化为“="盈余"&A2-B2”。不过后一个公式似乎让人更难理解。




图9-14





9.3.3 四舍五入后计算结果不正确




运用Excel进行统计计算的时候，免不了会遇到需要对数据进行四舍五入的情况。可不少初学者都会发现，经过“四舍五入”的数据，计算出来的结果似乎不正确。如图9-15所示，D1和D2单元格的公式都是对前面3个单元格的数据求和。它们的数据似乎都一样，可结果却不同。




图9-15

这是因为，A1、B1、C1单元格的数据是经过“设置单元格格式”——“数值”，将“数值”格式设置为“小数位数2位”。经过如此设置后，数据显示为“四舍五入”后保留两位小数的数据。但实际上单元格中真实数据并未改变，求和计算时，仍以真值计算。如图9-16所示，单元格显示数据与编辑栏真值不同。因此，通过设置“数值”格式并不能真正达到四舍五入的目的。




图9-16

如果想让函数按单元格所显示的数据进行计算，而不管其真值如何，也是可以做到的。如图9-17所示，执行菜单栏的“工具”→“选项”→“重新计算”命令，将“工作簿选项”选区下的“以显示精度为准”复选项勾选，单击“确定”按钮后，会弹出如图9-18所示的警告提示。如果在警告提示框单击“确定”按钮，工作簿中所有数据都将以单元格显示值为准进行计算。

建议尽量避免使用此设置，因为此设置是不可逆的。工作簿中所有单元格显示值变成了真值。如原值为5.36，显示值为5.4，如此设置后，数据彻底变成了5.4，即使再将“以显示精度为准”取消，也无法找回原值5.36，这可能会给你的工作带来重大损失。




图9-17




图9-18

真正的四舍五入，应该采用函数方法。如图9-19所示，D1单元格公式写成“=SUMPRODUCT(ROUND(A1:C1,2))”。公式首先使用ROUND函数，将区域中的数据四舍五入为两位小数，再进行求和计算。这里求和计算使用了SUMPRODUCT函数，而未用SUM函数，是为了避免使用数组公式。




图9-19





9.3.4 小数计算结果出错




我见过很多人使用Excel时，手边放一台计算器，随时检验Excel计算结果是否正确。除了以上“四舍五入”问题造成的“恐慌”外，还有就是不少人发现，Excel在小数计算时不是很准确。

如图9-20所示，想得到A列数据小数点后有几位小数，B1单元格的公式为“LEN(A1)-FIND(".",A1)”。我曾有过另一思路，C1单元格公式为“=MOD(A1,1)”，由此可得到小数部分，每一小数部分除了小数位外，都有一个“0”和一个“小数点”，如果用小数部分的字符数减去两个字符，应该就是小数位的字符数。基于此思考，D1单元格的公式为“=LEN(C1)-2”，结果十分令人震惊！在编辑栏将C1单元格的公式选中，按【F9】键，会发现C1的值实际上是“0.359999999999999”。




图9-20

出现以上计算不准确的情况是因为浮点运算的结果。作为计算机程序，Excel是以二进制存储数值的，某些10进制的非循环数在二进制下却是无穷循环数，常见示例是0.1，它的二进制数是000110011001100110011这样一串无穷循环数，由于Excel只能保留15位的数字精度，所以在运算时，这样的无穷循环数被截取为近似值进行计算，因此会有计算误差。这就好比分数1/3在十进制下无法精确表示一样。

作为初学者，我们没必要了解更多关于浮点运算的知识，只要知道出现了浮点运算，应该如何规避即可。如图9-20中，为规避浮点运算，可以避免使用统计和数学函数，而尽量使用文本函数。在必须使用数学和统计函数时，也可以使用ROUND等函数截取数据有效位进行计算。如图9-21所示，C1单元格公式如果为“=IF(A1-B1>0.1,">0.1",IF(A1-B1<0.1,"<0.1","=0.1"))”，由于浮点运算的误差，C列的计算结果不是大于0.1，就是小于0.1，没一个正确的。而D1单元格的公式为“=IF(ROUND(A1-B1,1)>0.1,">0.1",IF(ROUND(A1-B1,1)<0.1,"<0.1","=0.1"))”，D列的运算结果才是正确的。




图9-21





9.4 使用公式时的误区




许多初学者在学习和使用公式时，会有这样或那样的误区。比如英文不好的人一看到函数名称就头痛，以为英文不好就学不会函数。其实Excel函数尽管有300多个，但经常被用到的却不多。就像汉字有成千上万个，常用字却不多一样。最常用的Excel函数一般也就三四十个，如果在日常工作中经常使用，用过几次自然就记住了，无需刻意识记。





9.4.1 公式越完美越好




有人是完美主义者，总想编写一个“万能”公式。如有人不屑于编写这样的公式“=MIDB(A1,SEARCHB("?",A1),99)”，认为公式中的99无出处、不严谨，一定要将公式写成“=MIDB(A1,SEARCHB("?",A1),LENB(A1)-SEARCHB("?",A1)+1)”。从严谨的角度来讲，这样做是没什么错的。但具体工作也要具体分析，如果具体工作中，某些函数的参数比较固定的话，就没必要大费周章以求严谨。

任何一个“严谨”的公式，都不可能放之四海而皆准。每一个公式都有具体的适用环境，大可不必为公式设想过于复杂的运算环境。实际上，工作中我们常会根据具体适用环境简化公式。





9.4.2 公式越精简越好




也有人认为公式越精减、越短小越好，恨不得能写出一个字符的公式来。尤其是一些对函数相对比较了解的人，可能会以是否够精简来衡量一个公式的好坏，甚至衡量编写公式的人的水平。

正如完美主义一样，精简主义也是一把双刃剑。尽管精简公式对于锻炼初学者的思维有好处，但在工作中，公式好坏应以实用、够用为判断标准。





9.4.3 期望一劳永逸




如前所述，任何一个成功的公式都有其特定的运算环境。期待一个公式在任何情况下都不需做出更改是不现实的。编写好的公式，往往需要根据条件的改变，而作出相应的修改和调试。因此编写一个容易让人看懂并理解的公式很重要。如果公式的思路过于特殊，而日后又可能进行修改，应该留有适当的注释。比如利用批注或定义名称进行注释。





9.4.4 花时间编公式划不来




对于少量一次性工作，编写公式的价值可能难以体现。但对于批量性工作，甚至经常重复的工作，编写适当的公式至关重要。有些初学者，对于编写公式有畏难情绪，每当要用到公式时，总感觉写一个公式的时间已足够完成本次工作，因此懒于编写公式。殊不知，写一个公式的时间可能较长，但足可以应付多次重复工作，节省的时间更为可观。





9.5 公式的调试技巧




高手有可能不假思索就写出一个常用公式。但任何高手遇到新问题，都不可能一笔写出一个正确无误的公式。写公式的过程，不仅是一个思索过程，更像一个探索过程。就像写文章，常常要边想，边写，边改。因此必须了解基本的调试技巧。





9.5.1 错误指示器




经常可以看到单元格左上角有一绿色小三角，如图9-22所示，那就是“错误指示器”。选择出现“错误指示器”的单元格，在单元格的旁边会出现如图9-23所示的小方格，方格内有一感叹号，这就是“智能标记”。鼠标移动到“智能标记”上略作停留，会出现一条“错误类型”的提示。如果单击“智能标记”，它会为我们提供一些选项，如图9-24所示。




图9-22




图9-23




图9-24

“错误指示器”是否出现，可以通过依次执行菜单栏的“工具”→“选项”命令，在“错误检查”选项卡下，勾选“允许后台错误检查”复选项来指定。出现哪一类错误，“错误指示器”才会显示，可以通过“错误检查”选项卡下的“规则”区域进行指定。如图9-25所示，如果取消选中“公式引用空单元格（E）”复选项，则公式引用空单元格不再会出现“错误指示器”。“错误检查”选项，也可直接选择图9-24中的“错误检查选项”进入。

在“错误检查”选项，还可设置“错误指示器”的颜色，默认为“绿色”。如果将颜色更改为“红色”，那么出现错误的单元格左上角就会出现红色小三角。

在图9-24的选项中，可以选择“忽略错误”使“错误指示器”消失。单击图9-25中的“重新设置忽略错误”按钮，可以使消失的“错误指示器”重新显示。




图9-25





9.5.2 错误检查




可以通过执行菜单栏的“工具”→“错误检查”命令，来显示“错误检查”对话框。它是一个无模式对话框，类似“查找-替换”对话框，在对话框显示的同时，仍可访问工作表。

如图9-26所示，在“错误检查”对话框中，单击“选项”按钮，可以进入“错误检查”选项。单击“上一个”或“下一个”按钮，可以循环检查错误单元格。




图9-26





9.5.3 公式求值




依次执行菜单栏的“工具”→“公式审核”→“公式求值”命令，可以打开如图9-27所示的“公式求值”对话框。该对话框可按计算顺序逐步显示公式计算结果。类似在编辑栏用【F9】键显示各子公式计算结果一样。大多有经验的人，更喜欢用【F9】键，因为更灵活便捷。




图9-27

在图9-26所示的“错误检查”对话框，根据单元格的错误类型，可能出现“显示计算步骤”按钮，单击此按钮，也可启动“公式求值”对话框。





9.5.4 公式审核




执行菜单栏的“工具”→“公式审核”→“显示"公式审核"工具栏”命令，可以显示如图9-28所示的“公式审核”工具栏。单击工具按钮，可以打开“错误检查”对话框、“公式求值”对话框、“监视窗口”对话框、追踪引用单元格、追踪从属单元格、圈释无效数据等。




图9-28





9.5.5 监视窗口




在如图9-29所示的“监视窗口”中，可以添加监视单元格，窗口中会显示被监视单元格的“值”和“公式”等，便于及时了解被监视单元格的变化。双击对话框中的单元格条目，会自动激活工作表中的单元格。




图9-29





9.5.6 F9分步计算公式




尽管有以上多种公式错误检查工具，但对于熟练的Excel使用者来说，更多的还是用【F9】键分步计算检查公式。

在图9-30所示的表中，B1单元格的公式可能较为复杂。选择B1单元格，按【F2】键进入公式编辑状态，用键盘左右方向键移动光标，如图9-30所示。当光标每移动到一个括号上时，这个括号和其对应的另一括号就会粗体显示1秒钟。通过这一变化可以确定子公式的位置。选择公式的一部分子公式，按【F9】键，可以单独计算出这部分子公式的结果，如图9-31所示。




图9-30




图9-31

通过【F9】键分步计算公式，可以检查一个公式具体出错的部位。每执行一步【F9】键检查完毕，如果该子公式计算结果正确，可单击工具栏的“撤销”按钮，从公式结果返回公式状态，再选择另一部分子公式，执行【F9】键检查。当发现子公式错误后，单击“撤销”按钮，返回公式状态，然后进行修改。修改后，如果需要保存新公式，可直接按回车键。如果不需保存新公式，而要返回原公式，可单击“公式编辑栏”左侧的“撤销”按钮。





9.6 简化函数公式的方法




任何一个复杂的公式，都是由简单公式组合而成。一个公式也可能包含很多函数。但不管公式如何复杂，包含多少函数，都可以通过本节介绍的途径进行简化。简化公式的练习，不仅是函数的学习过程，更是了解和学习公式思路的过程。





9.6.1 辅助单元格




写一个复杂的公式，往往要通过辅助单元格一步一步完成。如图9-32所示，在E1单元格得到工作表名称，公式为“=RIGHT(A1,D1)”。得到这一结果，要经过4个单元格的辅助公式。A1单元格公式为“=CELL("filename")”，得到一个包括全部路径的文件名。B1单元格公式为“=FIND("]",A1)”，从A1单元格的文件名中，查找到标志性字符“]”的位置。 C1单元格公式为“=LEN(A1)”，得到文件名的字符数。D1单元格公式为“=C1-B1”，得到工作表名的字符数。

如果省去中间辅助单元格环节，E1单元格公式中的参数D1用“C1-B1”替换，得到公式“=RIGHT(A1, C1-B1)”。再进一步将公式中的参数C1用“LEN(A1)”替换， B1用“FIND("]",A1)”替换，得到公式“=RIGHT(A1, LEN(A1)- FIND("]",A1))”。最后将公式中的参数A1用“CELL("filename")”替换，得到最终公式“=RIGHT(CELL("filename"),LEN(CELL("filename"))-FIND("]",CELL("filename")))”，这时可以将所有辅助单元格删除。




图9-32

简化一个复杂公式的过程，正是以上编写复杂公式的逆过程。将一个复杂公式中的各子公式，用辅助单元格替换，不仅可以简单公式，而且可以了解公式的演算过程。





9.6.2 定义名称




定义名称是简化公式常用的方法，尤其是遇到公式超过1024个字符，或遇到某些函数限制的情况。比如上一节的最终公式，多次出现参数“CELL("filename")”，就可以把这部分参数定义成一个名称。

如图9-33所示，执行菜单栏的“插入”→“名称”→“定义”命令，打开“定义名称”对话框，在“名称”栏输入自己拟定的名称，如“name”。在“引用位置”栏输入公式“=CELL("filename")”。单击“确定”按钮关闭“定义名称”对话框。最后将公式改写成“=RIGHT(name,LEN(name)-FIND("]",name))”，这个公式不仅简短了许多，而且更容易理解。




图9-33





9.6.3 了解更多函数用法




了解更多函数无疑会对工作有更大帮助。比如要计算今天之后第8个月有多少天，按常规方法可能会写出这样一个公式，“=DAY(DATE(YEAR(TODAY()),MONT H(TODAY())+9,0))”。而如果了解另一个函数的用法，公式会更简单，如将公式写成“=DAY(EOMONTH(TODAY(),8))”。

如果在“插入函数”对话框无法找到“EOMONTH”函数，或者以上公式出现错误值，可能是未添加“加载宏”。可执行菜单栏的“工具”→“加载宏”命令，打开如图9-34所示的“加载宏”对话框，勾选“分析工具库”复选项，然后单击“确定”按钮。




图9-34





9.6.4 换个思路




简洁的公式往往来源于好的思路。即使是Excel高手，也经常分析别人的公式，以了解公式的独特思路。如图9-35所示，想知道A列与B列日期是否属于同一周，常规思路的公式可能写成“=IF(B1-A1=WEEKDAY(B1,2)-WEEKDAY(A1,2),"同一周","不在同一周")”。

换一个思路，“1900-1-0”是Excel默认的起始日期，这一天是周六。“1900-1-2”是周一，从这一天开始，每七天是一周。“1900-1-2”的日期序号是“2”，那么一个日期减去“2”再除以“7”，得到的整数部分如果相等，应该就在同一周。因此公式可写成“=IF(INT((A1-2)/7)=INT((B1-2)/7),"同一周","不在同一周")”。




图9-35

公式调试和错误值处理，是每个编写Excel公式的人都必须掌握的基本技能。公式简化则是研究和练习公式思路的最佳手段。学习函数是一个不断积累的过程，多看别人写的公式，利用【F9】键分步计算公式，了解公式运算过程和公式思路，是快速掌握Excel函数的有效途径。





第10章 宏表函数








10.1 认识宏表函数




Excel 5.0以前的版本中是没有VBA的，而是有一个专门的工作表叫作宏表，在宏表中可以编写一些程序语句，用于实现Excel的扩展功能。这些在宏表中用的函数就叫宏表函数。

以后的版本已由强大的VBA代替，但还保留对Excel 4.0宏表的支持。所不同的是，宏表函数不能直接在单元格中使用，而必须定义成名称。同时，宏表函数不属于易失性函数。

常用的宏表函数所述如下。

GET.CELL　　返回关于格式化、位置或单元格内容的信息。

GET.DOCUMENT　返回关于工作簿中工作表的信息。

GET.WORKBOOK　返回关于工作簿的信息。

EVALUATE　　对以文字表示的一个公式或表达式求值。

FILES　　返回指定目录的所有文件名。

DOCUMENTS　返回指定的已打开工作簿中按字母顺序排列的名字。

LINKS　　返回指定工作簿中的外部引用提及的所有工作簿的名字。

GET.FORMULA　返回出现在编辑栏中的单元格的内容。

本章着重介绍其中最常用的几个宏表函数。





10.2 用FILES获取指定目录的文件名




函数语法：

FILES(Directory_text)

参数Directory_text指定从哪一个目录中返回文件名，以及限定文件名的范围和后缀。接受通配符星号（*）代表文件名中任何长度的字符，问号（?）代表文件名中的单个字符。

如果Directory_text没有指定，FILES返回当前目录下的文件名。





10.2.1 获取文件名




先定义名称，如图10-1所示。




图10-1

把FILES定义成名称myfiles，并且指定列出E盘下的文件名。

定义好后，在单元中输入“=myfiles”确定，单元格中会显示第一个文件的文件名，如图10-2所示。




图10-2

在编辑栏中用【F9】键查看，可以看到实际上都列了出来，如图10-3所示。




图10-3

从这里可以看出所列出来的文件名组成了一个水平数组，通过以下公式可以计算出文件的数量：

=COUNTA(myfiles)

或者=COLUMNS(myfiles)





10.2.2 通过引用函数把这些文件名一一列到单元格里




通过引用函数，可以把这些文件名一一列到单元格里去，如图10-4所示。




图10-4

公式是：=INDEX(myfiles,ROW())。 A12显示错误值，是因为已经超过文件个数了。

假设把定义名称中的引用修改为“=FILES("E:\*.xl*")”，那么将只列出后缀xl开始的文件，如图10-5所示。




图10-5





10.2.3 解决宏表函数的非易失性问题




由于宏表函数都是非易失性函数，所以当相关路径中的文件有变化时，即使按【F9】键刷新，也不会更新，只有手动再次填充一次公式才可以更新。

当然强大的Excel是有办法弥补的。

在定义名称中，myfiles的引用修改为“=FILES("E:\*.xl*")&T(NOW())”，这时候按【F9】键刷新，结果就可以随文件的增删或修改而变化了，如图10-6所示。

NOW()是易失性函数，返回当前系统时间，外面套T()函数后，返回空字符。这样既带动整个定义名称公式在刷新时自动计算，又不会影响结果。

其他的宏表函数也可以做同样的处理。




图10-6





10.2.4 定义名称在宏表函数中的灵活应用




假设引用的路径不想每次都进入定义名称框内修改，而是放在单元格内，那么可以把这个单元格也定义成名称，供FILES使用，如图10-7所示。




图10-7

在图10-7所示的表中，B1单元格输入了路径“E:\”，并且利用名称框将其定义名称为“路径”，那么myfiles的定义名称就可以修改为：=FILES(路径&"*.xl*")&T(NOW())，如图10-8所示。




图10-8

这时输入“INDEX(myfiles,ROW())，”将列出E:\目录下的后缀以xl开始的文件名。

在B1单元格中修改路径为“D:\EXCEL课件\”，则会列出“D:\EXCEL课件\”目录下后缀以xl开始的文件名。

同样，这种把部分参数定义为名称的方法，也适用于其他宏表函数。





10.3 用GET.WORKBOOK获取工作簿的信息




函数语法：

GET.WORKBOOK(type_num, name_text)

type_num是指明要得到的工作簿信息类型的一个数值。这个参数一共有38个，包括返回工作簿中所有表的名字、工作簿中表的数量等。

name_text是打开的工作簿的名字，当省略时，默认为活动工作簿。





10.3.1 获取工作表名




在定义名称中，定义名称myworkbook为“=GET.WORKBOOK(1)”。和宏函数FILES相似的是，这个宏函数在单元格内生成一个以当前工作簿的工作表名为内容的水平数组，如图10-9所示。




图10-9

从图10-9可以看到，这里列出的每个工作表名，带有工作簿的文件名“[宏表函数.xls]”在内。显然，想简洁地显示的话，可以用替换函数SUBSTITUTE将文件名替换为空。





10.3.2 获取工作簿中工作表的数量




GET.WORKBOOK的第1参数还有一个常用的值是4，获取工作簿中表的数量。

定义名称myworkbook为“=GET.WORKBOOK(4)，”则公式“=myworkbook”返回的是当前工作簿的数量。

这个结果和内置函数的相似。在实际使用中，会出现INFO函数结果“出错”的情况，如图10-10所示。




图10-10

图10-10所示的工作簿是新建的，默认是3个工作表，没有做任何隐藏。

为什么“=INFO("numfile")”会出错呢？这个和加载宏是有关系的。

内置函数INFO("numfile")是计算已开启的工作簿中活动工作表的数目。加载宏其实对应有活动工作表，可以从【ALT+F11】快捷键进入的VBA编辑器的工程列表中看到相关的活动工作表。

执行“工具”→“加载宏”菜单命令，就可以看到几乎所有的加载宏都使用了，如图10-11所示。




图10-11

当把这些加载宏都去除后，就发现结果和myworkbook计算出来的是一样的了，如图10-12所示。

值得留意的是，INFO函数也是非易失性函数，在去除加载宏后，需要按【F9】键刷新工作表，才会显示正确的结果。




图10-12





10.3.3 GET.WORKBOOK的第2参数




以上GET.WORKBOOK的使用都是忽略了第2参数，即默认获取当前工作簿信息，如果加上文件名，即可获取其他工作簿的信息。

比如定义名称“otherbook=GET.WORKBOOK(4,"汉字数字对调.xls")，”即可获取“汉字数字对调.xls”这个文件的工作表数量。

但必须有个前提，就是“汉字数字对调.xls”这个文件必须在打开状态，所以第2参数不必注明工作簿文件的路径。如果文件关闭，公式将返回错误值“#N/A”。





10.4 用GET.DOCUMENT 获取关于工作簿中工作表的信息




函数语法：

GET.DOCUMENT(type_num, name_text)

Type_num是指明信息类型的数值。这个参数一共有88个，包括返回活动工作表和活动工作簿的文件名等。

Name_text是打开的工作簿的名字，当省略时，默认为活动工作簿。





10.4.1 获取活动工作表名称和活动工作簿名称




第1参数取1、76、88都可以获得工作表名，通常最好使用 GET. DOCUMENT(76) 和 GET. DOCUMENT(88) 来返回活动工作表和活动工作簿的文件名。

定义名称sheetname为“=GET.DOCUMENT(76)&T(NOW())”，可以获得当前工作表名称。

定义名称docuname为“= GET.DOCUMENT(88)&T(NOW())”，可以获得当前工作簿名称。

多的工作表名称的格式为“[Book.xls]Sheet1”，如果想只返回工作表名，则可以使用替换函数：

=SUBSTITUTE(sheetname,"["&docuname&"]",)

内置函数CELL("filename")也可以获取工作表名，返回的包括全部路径、工作簿名、工作表名。





10.4.2 获取文件路径




“=GET.DOCUMENT(2)&T(NOW())”可以获取工作簿保存时所在的路径，格式是“盘符:\路径”。





10.4.3 获取当前工作表的页码信息




Excel表格中的页码，只能设置在页眉和页脚中，有时候我们希望在单元格中显示页码信息，可以怎么做呢？先设置3个定义名称，如下所述。

横向当前页：=IF(ROW()<GET.DOCUMENT(64),1,MATCH(ROW(),GET. DOCUMENT(64))+1)

纵向当前页：=IF(COLUMN()<GET.DOCUMENT(65),1,MATCH(COLUMN(),GET. DOCUMENT(65))+1)

总页数：=GET.DOCUMENT(50)+NOW()*0

最后在当前工作表的任意单元格输入：

="第"&IF(当前横向页=1,当前纵向页,t当前横向页+当前纵向页)&"页/共"&totalpage&"页"

页码信息的显示格式是：第N页/共M页，如图10-13所示。




图10-13

公式说明：

GET.DOCUMENT(64)返回的是行数的数组，即分页符下面的行号。在Excel系统默认状态下，第一个页面中GET.DOCUMENT(64)返回的结果是48。

因此，可以通过当前行号和当前页面分页符下的行号做对比，判断是哪一个页面。

同时，还需要考虑Excel先列后行的打印顺序，来判断先计算纵向页码，再计算横向页码。





10.5 用GET.CELL获取单元格属性信息




GET.CELL是应用最广泛的宏表函数之一。

函数语法：

GET.CELL(type_num, reference)

Type_num是指明单元格中信息的类型。

Reference是提供信息的单元格或单元格范围。如果引用的是单元格范围，使用引用中第一个范围的左上角的单元格，如果引用被省略，默认为活动单元格。





10.5.1 获取单元格格式类型




当需要获取单元格格式类型的时候，就可以采用以下方法：

定义名称type=GET.CELL(7,A2)。输入公式=type就可以获取A2单元格的类型了，如图10-14所示。




图10-14

要注意的是，这里在B2输入公式“=type”后，需要下拉填充，因此在定义名称“=GET.CELL(7,A2)”中的A2被自动加上绝对符号时，需要把行号前的绝对符号删除。

GET.CELL(7,A2)返回的格式类型，和在单元格中单击鼠标右键调整的格式类型完全一致。读者可以通过单击鼠标右键进入格式来验证。

于此相类似的是内置函数CELL("FORMAT")，所不同的是，CELL("FORMAT")返回的是格式的代码，如图10-15所示。




图10-15





10.5.2 提取单元格颜色




在数据统计中，经常需要根据颜色汇总数据或者进行筛选，这时候可以通过GET.CELL来提取颜色代码，如图10-16所示。




图10-16

定义名称“color =GET.CELL(63,B2)”，在C2中输入“=color”，下拉填充。C列即形成辅助列获取B列的颜色代码。同样D列可以作为F列的辅助列。右侧的汇总即可根据此代码汇总了，如图10-17所示。




图10-17

在定义名称中，还可以定义"color =GET.CELL(38,B2)"，结果是一样的。

GET.CELL(38,B2)提取的是单元格的前景颜色，GET.CELL(63,B2)提取的是单元格的填充颜色。这里两个公式的结果是相同的。

所谓前景颜色，是在单元格格式设置中，图案选取的是非实心的图案，即点状图案或者条纹状图案等。

请看图10-18。




图10-18

在图10-18所示的界面中，单击“图案”标签，左上角的是实心图案。当选择实心图案的时候，两个公式返回的颜色代码是一致的。只有当选择其他图案，并且单元格的颜色和图案颜色不相同的时候，返回的颜色代码就会不同。

如图10-19所示。

在这里GET.CELL(38,)的前景色和GET.CELL(63,)的填充色就容易理解了。前景色对应的图案，可以看做是在背景色的上层。




图10-19





10.5.3 提取单元格里的公式




有时候希望在工作表中提取某些单元格的函数公式，这时候可以用GET.CELL(6,)来提取。如图10-20所示。




图10-20

定义名称“formula1=GET.CELL(6,A2)”，在B2单元格中输入“=formula1”，下拉填充，即可得到各个单元格的公式表达式。

除了GET.CELL外，GET.FORMULA也可以提取单元格的公式，如图10-21所示。




图10-21

定义名称“formula2=GET.formula(A2)”，输入“=formula2”，就可以返回单元格的表达式。

从图10-21可以看出，当公式表达式中有单元格的时候，公式的返回结果是R1C1格式。

而对于GET.CELL(6,A2)，只有当工作表设置成R1C1样式的时候，才返回R1C1格式。





10.6 用EVALUATE宏表函数计算表达式




当单元格中是计算的表达式的时候，可以用EVALUATE宏表函数来计算其结果。

公式语法：

EVALUATE(formula_text)

formula_text是一个要求值的、以文字形式表示的表达式，简单地说，就是没有等号的公式。

如图10-22所示。




图10-22

定义名称“mycal1=EVALUATE(A2)”，在B2中输入公式“=mycal1”，将计算出A列的表达式的结果。

表达式中如果有内置函数的，也可以计算出结果，如上图中的“power(3,2)+8”和“sum(1200,100)”可以计算出结果。但前提是内置函数中必须是常数，像“sum(A1,A2)”会返回数字0。

EVALUATE宏表函数还可以和SUBSTITUTE函数结合解决一些特别的问题。如图10-23所示，是需要计算用逗号间隔的数据。




图10-23

定义名称“mycal2 =EVALUATE(SUBSTITUTE(A2,",","+"))”。

“SUBSTITUTE(A2,",","+”)”在这里的作用，是把“1,2,3,4,5”转换成“1+2+3+4+5”这样的形式，然后用EVALUATE宏表函数来求值。

对于计算表达式中数据夹杂着注释的，处理起来就相对复杂点了。




图10-24

针对图10-24中的表达式，必须有不同的替换处理过程。

对于第一个计算表达式，定义名称：

mycal3 =EVALUATE(SUBSTITUTE(SUBSTITUTE(Sheet9!A2,"[","N("""),"]",""")+”))。

在这个公式里面，“SUBSTITUTE(SUBSTITUTE(Sheet9!A2,"[","N("""),"]",""")+")”的目的是将[仓库1]和[仓库2]都整理成N("仓库1")和N("仓库2")的形式，整个计算表达式就转换成N("仓库1")+120+N(“仓库2")+200。

套用EVALUATE后，公式“N("仓库1")和N("仓库2")”会计算成0。

对于第二个计算表达式，定义名称有所不同：

mycal4 =EVALUATE(SUBSTITUTE(SUBSTITUTE(Sheet9!A3,"[","ISTEXT(""["),"]","]"")*"))。

相同的道理，“SUBSTITUTE(SUBSTITUTE(Sheet9!A3,"[","ISTEXT(""["),"]","]"")*")”把表达式转换成：ISTEXT("[长]")*3*ISTEXT("[宽]")*4*ISTEXT("[高]")*5。

套用EVALUATE后，ISTEXT（）会计算成1。

由此看出，不同的表达式需要通过不同的方法转换，灵活处理。

EVALUATE宏表函数也在提醒我们，规范的数据录入格式，将会给统计汇总工作带来极大的方便，否则必须使用繁杂的公式进行转换。不仅增加了工作量，而且不利于数据的维护。

8.6 函数与图片



8.7 附：各类功能与函数的关系



第9章 公式简化与调试



9.1 函数返回错误值的处理





9.1.1 #NAME?——晕，这是什么名字









9.1.2 #DIV/0!——晕，竟然除以0









9.1.3 #N/A——晕，无效数据









9.1.4 #NULL!——晕，竟然是空的









9.1.5 #NUM!——晕，数字错了









9.1.6 #REF!——晕，引用的范围不存在









9.1.7 #VALUE!——晕，用错值了







9.2 公式常见的其他错误提示





9.2.1 公式错误是否接受更正建议









9.2.2 循环引用不能计算公式









9.2.3 为函数输入太少或太多参数









9.2.4 是否更新链接









9.2.5 输入的公式中包含错误









9.2.6 单元格填满“#”号







9.3 没有提示的公式错误





9.3.1 正确公式复制后得不到正确结果









9.3.2 按顺序书写的公式不按顺序计算









9.3.3 四舍五入后计算结果不正确









9.3.4 小数计算结果出错







9.4 使用公式时的误区





9.4.1 公式越完美越好









9.4.2 公式越精简越好









9.4.3 期望一劳永逸









9.4.4 花时间编公式划不来







9.5 公式的调试技巧





9.5.1 错误指示器









9.5.2 错误检查









9.5.3 公式求值









9.5.4 公式审核









9.5.5 监视窗口









9.5.6 F9分步计算公式







9.6 简化函数公式的方法





9.6.1 辅助单元格









9.6.2 定义名称









9.6.3 了解更多函数用法









9.6.4 换个思路







第10章 宏表函数



10.1 认识宏表函数



10.2 用FILES获取指定目录的文件名





10.2.1 获取文件名









10.2.2 通过引用函数把这些文件名一一列到单元格里









10.2.3 解决宏表函数的非易失性问题









10.2.4 定义名称在宏表函数中的灵活应用







10.3 用GET.WORKBOOK获取工作簿的信息





10.3.1 获取工作表名









10.3.2 获取工作簿中工作表的数量









10.3.3 GET.WORKBOOK的第2参数







10.4 用GET.DOCUMENT 获取关于工作簿中工作表的信息





10.4.1 获取活动工作表名称和活动工作簿名称









10.4.2 获取文件路径









10.4.3 获取当前工作表的页码信息







10.5 用GET.CELL获取单元格属性信息





10.5.1 获取单元格格式类型









10.5.2 提取单元格颜色









10.5.3 提取单元格里的公式







10.6 用EVALUATE宏表函数计算表达式

